



ROBERT YOUNG & ROGER BUSH

MSX

PROGRAMMA MIX



 ADDISON-WESLEY

**MSX
PROGRAMMA MIX**

MET COMPLETE BASIC LISTINGS

**MSX
PROGRAMMA MIX**

***MET COMPLETE
BASIC LISTINGS***

**ROGER BUSH
ROBERT YOUNG**



ADDISON-WESLEY EUROPE B.V.

CIP-GEGEVENS KONINKLIJKE BIBLIOTHEEK, DEN HAAG

Young, Robert

MSX programma mix : met complete BASIC listings / Robert Young, Roger Bush ; [vert. uit het Engels door Line-up Tekstprodukties]. – Amsterdam : Addison-Wesley Europe
Bew. van: Games Commodore 64 computers play. – Wokingham : Addison-Wesley, 1983.
ISBN 90-6789-060-X
SISO 365.3 UDC 681.3.06
Trefw.: MSX (computer) / programmeren (computer).
Omslagontwerp: BSE Advertising

Niets uit deze uitgave mag worden
verveelvoudigd en/of openbaar gemaakt door
middel van druk, fotokopie, microfilm,
geluidsband, elektronisch of op welke andere
wijze ook en evenmin in een retrieval system
worden opgeslagen zonder voorafgaande
schriftelijke toestemming van de uitgever.

De in dit boek aangeboden programma's zijn opgenomen
vanwege hun instructieve waarde. Alle programma's zijn
getest, uitsluitend voor de toepassing als in dit boek
beschreven. De uitgever kan niet verantwoordelijk worden
gesteld voor eventuele toch optredende fouten.

Colofon:

Productie, redactie & realisatie: LINE UP tekstprodukties, Groningen
Illustraties: C.F.M. de Valk

INHOUD

Voorwoord 1

ARCADE ACTIE 3

Springer 5
Afdaling in Chaos 11
Schiet 16

AVONTUUR EN SIMULATIE 21

Overleven 23
Het Romeinse Rijk 34
De slag om Engeland 46
Eigenheimerheerd 56

DAAG DE COMPUTER UIT 71

Boter, kaas en eieren in 3-D 73
Ontcijfer X-code 83
Othello 90
Schaakbord NIM 95
Hersensbreker 98

ZELF PROGRAMMA'S SCHRIJVEN 105

STAPSGEWIJS EEN PROGRAMMA ONTWIKKELEN 109

VOORWOORD

MSX is het antwoord op de vertwijfelde roep om een standaard voor betaalbare computers. De apparaten die zich daaraan conformeren, vallen vooral op door hun prestaties in BASIC. Inmiddels heeft de tweede generatie MSX computers zich aangediend, met nog meer mogelijkheden dan haar illustere voorgangers.

Zulke computers vragen om kwaliteitsoftware, liefst volwaardige programma's die een goede kans maken als klassiekers in de annalen te worden bijgeschreven.

Of de programma's in dit boek aan een dergelijke hoge eis voldoen, zal de tijd leren. Vast staat echter dat alleen mensen die hun vak verstaan, in staat zijn programma's te schrijven als NIM, X-code of Hersenbreker.

Misschien heeft u de indruk dat programmeren berust op ondoorgroendelijke vakkennis. Wat BASIC betreft is niets minder waar. Om u daarvan te overtuigen openen de auteurs ter afsluiting van dit boek hun gereedschapskist. Dat u daarvoor helemaal in de ban raakt van hun enthousiasme is niet denkbeeldig.

ARCADE ACTIE

Springer

Afdaling in Chaos


Schiet

SPRINGER

SPRINGER is een spannend spel waar veel behendigheid voor nodig is. De snel bewegende grafiek wordt bereikt met behulp van programmeerbare symbolen. De bedoeling is om een springend mannetje te besturen, dat zo snel mogelijk 21 ballonnen stuk moet prikken. Het mannetje springt omhoog van plateaus onderaan het scherm en kan naar boven, naar links en naar rechts springen. Ook kan hij, als hij in de lucht zweeft, een beetje naar weerskanten worden bewogen.

U bestuurt het mannetje met de cursorbesturingstoetsen NAAR LINKS en NAAR RECHTS. Als hij niet af en toe een zet naar boven krijgt met CURSOR OMHOOG, veert hij steeds minder op en komt tenslotte tot stilstand.

Het spel heeft negen niveaus, oplopend in moeilijkheid. Op elk niveau staan de plateaus op een andere plaats en hebben een andere breedte. Aan het begin van elk spelletje vraagt de computer welk niveau u wilt proberen. Drukt u op 0, dan neemt de computer een willekeurig niveau. Nadat het gewenste niveau is ingetypt, volgt een korte wachttijd, waarin de verschillende karakters opnieuw worden gedefinieerd. Het scherm wordt schoongemaakt en er verschijnen verschillende plateaus en een aantal ballonnen tegen een rose achtergrond. Het mannetje verschijnt aan de linkerkant van het scherm en springt van het plateau. Het spel kan dan beginnen.



Programmabeschrijving:

in regel 80 en 90 wordt de moeilijkheidsgraad opgeroepen. Regel 120 stuurt het programma naar de subroutine in regel 970–1040, waarin de karakters 100–106 worden omgeprogrammeerd tot de figuren die we nodig hebben. Regel 130 roept de routine in regel 1050 aan waar de basiswaarden van de variabelen worden bepaald.

De ON GOTO-instructie in regel 150 roept het gewenste plateau op, waarna het in regel 840–960 op het scherm wordt gePRINT. Als u het programma LIST, verschijnen de plateaus als een reeks h's. Na de herdefiniëring vormen deze h's op het scherm een dicht plateau. Regel 160 PRINT de ballonnen op willekeurige plaatsen op het scherm. De routine die het springende ventje op het scherm brengt, wordt aangeroepen in regel 170. Regel 180 POKEt een spatie (karakter 32) op de vorige positie van de springer. Dat geeft het effect van een continue beweging en zorgt ervoor dat er niet meer dan één springer tegelijk op het scherm komt.

De invoer van de cursorbesturingstoetsen wordt in regel 190 gelezen. Regel 200 geeft een pieptoon. De regels 210 en 220 controleren de bewegingen van de springer naar links en naar rechts wanneer hij in de lucht is. De instructie STICK(0) leest de invoer en vermindert of vermeerderd de waarde van X met 0.1. Dit betekent dat na 10 waardeveranderingen de springer van plaats verandert in horizontale richting (X-as). De verticale beweging van de springer wordt gecontroleerd in de regels 230–260. Of het mannetje een ballon heeft geraakt, wordt nagegaan in regel 270. Regel 280 POKEt de springer op het scherm en de variabele CN bepaalt welke gedaante van de springer moet verschijnen (springer naar links, naar rechts of omhoog).

Regel 290 zorgt ervoor dat de springer niet van het scherm valt en regel 300 gaat na of hij het plateau niet gemist heeft. De spronghoogte wordt berekend in regel 310, waarbij de variabele LMT (limit = grens) aangeeft aan hoe hoog het mannetje mag springen. Regel 320 wijst terug naar het begin van het hoofdprogramma, dat begint in regel 180.

Het aantal geraakte ballonnen wordt bijgehouden in de regels 330 en 340 door de variabele HIT. Deze regels POKEn ook de kapotte ballonnen op het scherm. Het karakter dat voor de gesprongen ballonnen wordt gebruikt, is de asterisk (*) van het toetsenbord. Die wordt zo op een heel effectieve manier gebruikt.

Door middel van de variabele LD in de regels 350–370 wordt de richting van de springer gecontroleerd. Waarde 1 betekent 'naar links', waarde 2 'naar boven' en waarde 3 'naar rechts'. De waarden voor LD worden vastgesteld in regel 580–600.

Zijn er 21 ballonnen doorgeprikt, dan zorgen de regels 380–540 ervoor dat de score op het scherm komt. Daarna volgt de vraag of u het spel nog eens wilt spelen.

De routine in de regels 550–630 wordt aangeroepen wanneer het mannetje van het plateau springt. Daarin wordt de variabele LD gebruikt om de richting van de sprong te bepalen. De dolle sprongen van de springer aan het begin van het spel worden geproduceerd door de regels 640–740. Strandt het spel voortijdig door een sprong naast een plateau, dan melden de regels 750–820 dat.

De plateaus die bij de verschillende niveaus horen, worden in de regels 830–960 door middel van PRINT–statements op het scherm gebracht.

SPRINGER is een leuk spel waarmee u zich uren kunt vermaken. Het programma werd geschreven door William Leader.

```
10 REM SPRINGER
20 KEY OFF
30 CLS
40 SCREEN 1
50 HF=5
60 TIME=0
70 COLOR 1,13,13
80 LOCATE 2,10
90 PRINT"WELK NIVEAU (0-9, 0=RND)?"
100 A$=INKEY$:IF A$="" THEN 100
110 R=VAL(A$):IF R<1 THEN R=INT(RND(-TIME)*9)+1
120 GOSUB 1170:GOSUB 970
130 LOCATE 2,11:GOSUB 1050
140 FOR T=1 TO 30:PRINT:NEXT
150 ON R GOTO 830,850,860,880,900,920,940,950,960
160 FOR L=1 TO 30:VPOKE 6144+INT(RND(1)*320),105:
    NEXT L
170 GOTO 640
180 VPOKE(6144+X+32*Y),32
190 P=STICK(0)
200 BEEP:GOSUB 1070
210 IF P=7 THEN X=X-.1
220 IF P=3 THEN X=X+.1
230 IF D=-1 THEN GOTO 350
240 IF D=1 THEN Y=Y+1:CN=101
250 IF D=1 AND VPEEK(6144+X+32*Y)=104 THEN GOTO
    550
260 IF D=-1 AND VPEEK(6144+X+32*Y)=104 THEN
    Y=Y+1:D=1:W=-1
270 IF VPEEK(6144+X+32*Y)=105 THEN GOTO 330
280 VPOKE (6144+X+32*Y),CN
290 IF X<1 OR X>31 THEN D=1:W=-1
300 IF Y>22 THEN 750
310 IF Y<LMT THEN FOR T=1 TO 100:NEXT T:D=1:W=-1
320 GOTO 180
330 VPOKE(6144+X+32*Y),106:HIT=HIT+1:IF HIT>20
    THEN GOTO 380
340 GOSUB 1090:GOTO 180
```

```

350 IF LD=1 THEN Y=Y-1:CN=101:GOTO 240
360 IF LD=2 THEN Y=Y-1:X=X-1:CN=103:GOTO 240
370 IF LD=3 THEN Y=Y-1:X=X+1:CN=102:GOTO 240
380 POKE 6144+X+32*Y,CN
390 GOSUB 1130
400 SC=TIME
410 FOR TD=1 TO 1000:NEXT TD
420 FOR TD=1 TO 2000:NEXT TD
430 CLS
440 LOCATE 4,10
450 PRINT"U SCOORDE";100-INT(SC)/100
460 FOR TD=1 TO 2000:NEXT TD
470 LOCATE 2,14
480 PRINT"WILT U NOG EEN SPEL?"
490 A$=INKEY$
500 IF A$<>"J" THEN GOTO 520
510 RESTORE:RUN
520 IF A$<>"N" THEN GOTO 490
530 CLS
540 END
550 VPOKE 6144+X+32*(Y-1),100:Y=Y-1
560 FOR T=1 TO 5:W=1
570 P=STICK(0)
580 IF P=1 THEN LD=1:LMT=LMT-1.5:IF LMT<1 THEN
    LMT=0:GOTO 630
590 IF P=7 THEN LD=2:GOTO 630
600 IF P=3 THEN LD=3:GOTO 630
610 NEXT T
620 LD=1:LMT=LMT+.5:IF LMT>21 THEN LMT=21
630 D=-1:LMT=LMT+1:GOTO 180
640 LOCATE 0,5:PRINT"hhhhhh"
650 FOR T=1 TO 1000:NEXT T
660 FOR T=0 TO 5:VPOKE 6144+T+32*4,100:FOR I=1 TO
    100:NEXT I:VPOKE 6144+T+32*4,32:NEXT T
670 POKE 6144+5+32*4,100:FOR I=1 TO 1000:NEXT I
680 FOR T=4 TO 0 STEP-1:VPOKE 6144+5+32*T,101:FOR
    I=1 TO 100:NEXT I:VPOKE 6144+5+32*T,32:NEXT T
690 FOR I=1 TO 100:NEXT I
700 FOR T=0 TO 4:VPOKE 6144+5+32*T,101:FOR I=1 TO
    100:NEXT I:VPOKE 6144+5+32*T,32:NEXT T
710 VPOKE 6144+5+32*4,100:FOR T=1 TO 1000:NEXT T:
    X=5:Y=4:CN=101:VPOKE 6144+X+32*Y,32:D=-1
720 VPOKE 6144+X+32*Y,101:LD=3
730 LOCATE 0,5:PRINT " "
740 GOTO 180
750 VPOKE 6144+X+32*Y,106:GOSUB 1110

```

```

760 COLOR 15,4,9
770 FOR TD=1 TO 1000:NEXT TD
780 CLS
790 LOCATE 0,10
800 PRINT "U FAALDE NA SLECHTS";HIT
810 PRINT
820 PRINT "BALLONNEN TE HEBBEN DOORGEPRIKT":GOTO 460
830 REM VERSCHILLENDE PLATEAUS PRINTEN
840 LOCATE 0,22:PRINT"hhhhhhhhhhhhhhhhhhhhhhhhhhhhhhhhhhhh
hh":GOTO 160
850 LOCATE 0,22:PRINT"hhhhhhh hhhhhhh hhhhhh hh
hh":GOTO 160
860 LOCATE 0,17:PRINT"hhhhhhhhh":LOCATE 9,22:
PRINT"hhhhhhhhhh"
870 LOCATE 20,17:PRINT"hhhhhhhhh":GOTO 160
880 LOCATE 0,22:PRINT"hhhhhhhhhh":LOCATE 10,17:
PRINT"hhhhhhhhhh"
890 LOCATE 20,22:PRINT"hhhhhhhhh":GOTO 160
900 LOCATE 0,16:PRINT"hhhhhhhhhh":LOCATE 11,19:
PRINT"hhhh"
910 LOCATE 17,22:PRINT"hhhhhhhhhh":GOTO 160
920 LOCATE 0,14:PRINT"hhhhhhh":LOCATE 7,22:PRINT"h
hhhh"
930 LOCATE 14,19:PRINT"hhhh":LOCATE 20,17:PRINT"hh
hhhhh":GOTO 160
940 LOCATE 0,22:PRINT"h h h h h h h h h h h h h h
":GOTO 160
950 LOCATE 0,17:PRINT"hhhhhhhhhh":LOCATE 11,22:
PRINT"hhhhhhhhhh":LOCATE 20,17:PRINT"hhhhhhh
h":GOTO 160
960 LOCATE 0,17:PRINT"hhhhhhhhhh":LOCATE 11,10:
PRINT"hhhhhhhhhh":LOCATE 20,22:PRINT"hhhhhhh
h":GOTO 160
970 REM HERDEFINIERING VAN KARAKTERS
980 FOR N=100 TO 106
990 FOR M=0 TO 7
1000 READ A
1010 VPOKE N*8+M,A
1020 NEXT M
1030 NEXT N
1040 RETURN
1050 SO=100:W=1
1060 RETURN
1070 BEEP

```

```
1080 RETURN
1090 PLAY"O3CEF"
1100 RETURN
1110 PLAY"O5CEF"
1120 RETURN
1130 FOR K=1 TO 96
1140 PLAY"O5 N=K;"
1150 NEXT K
1160 RETURN
1170 CLS:LOCATE 6,11:PRINT"GEKOZEN NIVEAU";R:RETURN
1180 DATA 24,24,60,102,153,24,24,60,153,90,62,36,
24,36,66,129
1190 DATA 12,12,248,44,58,72,132,129,96,96,62,104,
184,68,66,129
1200 DATA 0,0,0,255,255,0,0,0,24,60,126,126,126,
126,60,24
1210 DATA 24,24,36,36,60,60,195,129
```

AFDALING IN CHAOS

AFDALING IN CHAOS is een bewegend grafisch spel, waarbij u uw ruimteschip laat afdalen in een lange, draaiende schacht naar het binnenste van de planeet Chaos. Om de missie tot een goed einde te brengen, moet u over een goed concentratie- en reactievermogen beschikken.

De besturing van het ruimteschip ligt in uw handen. De cursorbesturingstoetsen NAAR LINKS en NAAR RECHTS zorgen voor de verplaatsing van het schip. De afdaling kunt u vertragen door CURSOR OMHOOG in te drukken.

Nadat de titel op het scherm is verschenen, wordt het gewist en verschijnt het ruimteschip in de vorm van een diamant bovenaan het scherm. Het bevindt zich dan aan het begin van een lange schacht waarin het moet afdalen. Tijdens de tocht naar het einde van de schacht moet u botsingen met de wanden zien te vermijden. Het schip heeft vier 'levens'. Als u die heeft verloren, krijgt u te zien tot hoever u in de tunnel bent afgedaald.

Naarmate u dichterbij de kern van de planeet komt, wordt het spel moeilijker omdat de tunnel dan steeds nauwer wordt.



Programmabeschrijving:

het voornaamste deel van het programma, de hoofdroutine, zit tussen de FOR-NEXT loop (regel 80–240). Dat is de lus die het programma steeds doorloopt terwijl de planeet en de tunnel op het scherm worden gebracht.

De initialisatie-routine (regel 690–740) wordt in regel 20 aangeroepen. Regel 30 wijst naar de routine die de titelpagina van het spel verzorgt. In regel 40–60 wordt het scherm zwart gemaakt en gewist en wordt de stringvariabele S\$ (een string is een rij karakters) geïnitieerd. Deze string bestaat uit een aantal lege karakters (spaties) om de schacht te kunnen maken.

In regel 130 wordt de waarde van Q bepaald (die nodig is om S\$ over het scherm te TABuleren). De variabele X krijgt de waarde 259 (X wordt gebruikt om het ruimteschip op het scherm te kunnen laten bewegen).

In regel 80 begint de spellus en worden 36 grafische karakters (GRAPH –P) op het scherm gePRINT. Daarna wordt nagegaan (regel 110) hoe breed de schacht is en als de breedte niet meer dan twee karakters is, wordt de eindroutine aangeroepen. In regel 120–160 wordt een willekeurig getal gekozen, dat er, samen met variabele Q, toe dient om de string over het scherm te TABuleren. Deze veranderende waarde zorgt voor draaiingen en bochten in de schacht. Regel 140 en 150 zijn beveiligingen om te voorkomen dat de schacht van het scherm verdwijnt.

Het ruimteschip (karakter 4) wordt op het scherm gePOKEt met de X uit regel 280. Regel 180 roept de routine in regel 1100–1160 op, die de invoer van de cursorbesturingstoetsen leest. Wanneer CURSOR OMHOOG is ingedrukt, wordt een korte wachttijd ingelast om het spel te vertragen. Afhankelijk van de invoer wordt X met 1 verhoogd of verlaagd.

In regel 200 wordt nagegaan of u tegen de wand van de schacht bent aangekomen. Bij botsingen wordt variabele H in regel 310, die het aantal botsingen telt, verhoogd. De botsing wordt ook hoorbaar gemaakt door een 'bleep' (r. 320). Tevens wordt bekeken tegen welke zijde van de schacht u bent gebotst en wordt ervoor gezorgd dat u weer in het midden terechtkomt, zodat u het spel kunt voortzetten (r. 200).

In regel 230 wordt de variabele M verhoogd. Deze variabele geeft bij het einde van het spel aan hoever u in de schacht bent afgedaald.

Nadat de FOR-NEXT loop 200 maal doorlopen is, wordt in regel 250 een karakter van de variabele S\$ afgehaald, waardoor de schacht smaller wordt. Daarna gaan we weer terug naar regel 80 voor een volgende loop.

```

10 REM AFDALING IN CHAOS
20 GOSUB 690
30 GOSUB 750
40 COLOR 9,1
50 CLS
60 S$=CHR$(30)+"      "
70 REM 8 SPATIES
80 FOR RR=1 TO 200
90 PRINT"      "
100 REM 36*GRAPH+P
110 IF S$=CHR$(30)+"  " THEN GOTO 520
120 Z=RND(1)
130 Q=Q+(Z<.5)-(Z>=.5)
140 IF Q>23 THEN Q=23
150 IF Q<4 THEN Q=4
160 PRINT TAB(Q+3);S$
170 IF CSRLIN>21 THEN GOSUB 270
180 GOSUB 1110
190 IF CSRLIN<21 AND VPEEK(X+40)=219 THEN GOTO
    200 ELSE GOTO 210
200 IF VPEEK(X-2)=219 THEN X=X+2 ELSE X=X-2
210 IF CSRLIN>21 AND VPEEK(X+40)=219 THEN 300
220 FOR TD=1 TO 25:NEXT TD
230 M=M+10
240 NEXT RR
250 S$=LEFT$(S$, (LEN(S$)-1))
260 GOTO 80
270 VPOKE R-40,32
280 VPOKE X,4
290 RETURN
300 REM BOTSING GEHAD?
310 H=H+1
320 BEEP
330 FOR TD=1 TO 1000:NEXT TD
340 IF H=4 THEN GOTO 380
350 IF VPEEK(X-2)=219 THEN X=X+2:GOTO 370
360 X=X 2
370 GOTO 230
380 REM EINDE SPEL
390 CLS
400 LOCATE 0,10
410 PRINT "JE BENT";M/1000;"KM DOORGEDRONGEN IN"
420 PRINT
430 PRINT "DE TUNNELS VAN CHAOS"

```



```

440 PLAY "L405C"
450 FOR TD=1 TO 1000:NEXT TD
460 LOCATE 8,16
470 PRINT "NOG EEN SPELLETJE?"
480 A$=INKEY$
490 IF A$="J" THEN RUN
500 IF A$="N" THEN END
510 GOTO 480
520 REM DE BODEM BEREIKT
530 FOR X=1 TO 3
540 PLAY "O4 CEG","O5 CEG","O6 CEG"
550 NEXT X
560 FOR Z=1 TO 14
570 COLOR Z
580 FOR TD=1 TO 100:NEXT TD
590 LOCATE 7,10:PRINT"OLEE!"
600 NEXT Z
610 SS=SS+1:IF SS>3 THEN SS=0:GOTO 630
620 GOTO 560
630 CLS
640 LOCATE 0,5
650 PRINT TAB(12)"GOED GEDAAN!":PRINT
660 PRINT TAB(1)"JE BENT DOORGEDRONGEN TOT DE
  BODEM":PRINT
670 PRINT TAB(8)"VAN DE TUNNELS VAN CHAOS":PRINT
  :PRINT
680 GOTO 410
690 REM INITIALISATIE
700 CLS
710 KEY OFF
720 SCREEN 0
730 Q=10:X=259
740 RETURN
750 REM TITELPAGINA
760 P=RND(-TIME)
770 CLS
780 COLOR 8,1
790 FOR R=9 TO 25
800 READ B
810 IF R=17 OR R=20 THEN NEXT R
820 FOR Y=0 TO 13
830 LOCATE R,Y
840 PRINT" "
850 REM GRAPH SHIFT + |

```

```

860 LOCATE R,Y+1
870 PRINT CHR$(B)
880 FOR TD=1 TO 5:NEXT TD
890 NEXT Y
900 NEXT R
910 Y=13
920 FOR QW=1 TO 12
930 R=INT(RND(1)*17)+9
940 IF R=17 OR R=20 THEN 930
950 GOSUB 1200
960 FOR Y=13 TO 8 STEP -1
970 LOCATE R,Y+1
980 PRINT CHR$(32)
990 LOCATE R,Y
1000 PRINT CHR$(V)
1010 FOR TD=1 TO 15:NEXT TD
1020 NEXT Y
1030 FOR Y=8 TO 14
1040 LOCATE R,Y-1:PRINT" "
1050 REM GRAPH SHIFT + |
1060 LOCATE R,Y:PRINT CHR$(V)
1070 FOR TD=1 TO 15:NEXT TD
1080 NEXT Y
1090 NEXT QW
1100 RETURN
1110 REM STUURROUTINE
1120 D=STICK(0)
1130 R=X
1140 IF D=7 THEN X=X-1
1150 IF D=3 THEN X=X+1
1160 IF D=1 THEN FOR TD=1 TO 100:NEXT TD
1170 RETURN
1180 DATA 65,70,68,65,76,73,78,71,32
1190 DATA 73,78,32,67,72,65,79,83
1200 V=VPEEK((R+14*40)+2)
1210 RETURN

```

SCHIET


SCHIET gebruikt opnieuw gedefinieerde karakters om een geweldig spel van het type Space Invaders te maken. De buitenaardse indringer beweegt van links naar rechts over het scherm. Steeds als hij de rand van het scherm bereikt, valt hij een stukje naar beneden tot het volgende niveau om dan aan de andere kant van het scherm weer te verschijnen. U kunt maximaal tien punten verzamelen. Als het vreemde vliegtuig op uw niveau is gekomen, hebt u een punt verspeeld en verschijnt het vliegtuig weer bovenaan.

Gebruik de cursorbesturingstoetsen om de afweerbasis naar links en naar rechts te bewegen. Met de spatiebalk schiet u projectielen af.

Voor een flitsende start van het programma wordt een ingenieuze aftelroutine gebruikt. Deze routine maakt gebruik van vijf stringvariabelen en van de DATA in de regels 1060-1150.

De variabelen X en Y worden gebruikt als coördinaten voor de beeldschermpositie van de afweerbasis en de variabelen SX en SY om het vreemde toestel te verplaatsen.

SCHIET werd geschreven door Daniel Quinn.



```

10 REM SCHIET
20 DIM S1(90)
30 SCREEN 1
40 GOSUB 940
50 GOSUB 530
60 CLS:ST=0:W=6144
70 Y=751:EI=32
80 FOR X=ST TO EI
90 VPOKE W+Y,100:VPOKE W+X,101
100 D=STICK(0)
110 IF D=7 AND Y>736 THEN VPOKE W+Y,32:GOSUB 200
120 IF D=3 AND Y<767 THEN VPOKE W+Y,32:GOSUB 210
130 A$=INKEY$
140 IF A$=" " THEN VPOKE W+X,32:GOSUB 340
150 IF W+X>6880 THEN GOTO 440
160 VPOKE W+X,32
170 NEXT X
180 ST=ST+32:EI=EI+32
190 GOTO 80
200 Y=Y-1:VPOKE W+Y,100:RETURN
210 Y=Y+1:VPOKE W+Y,100:RETURN
220 VPOKE W+X,46:VPOKE W+X-1,46:VPOKE W+X+1,46
230 VPOKE W+X-33,46:VPOKE W+X+33,46
240 FOR TD=1 TO 10:NEXT TD
250 VPOKE W+X,32:VPOKE W+X-1,32:VPOKE W+X+1,32
260 VPOKE W+X-33,32:VPOKE W+X+33,32
270 FOR TD=1 TO 10:NEXT TD
280 VPOKE W+X+1,46:VPOKE W+X-2,46:VPOKE W+X+2,46
290 VPOKE W+X-34,46:VPOKE W+X+34,46
300 FOR TD=1 TO 10:NEXT TD
310 VPOKE W+X+1,32:VPOKE W+X-2,32:VPOKE W+X+2,32
320 VPOKE W+X-33,32:VPOKE W+X+33,32
330 GOTO 440
340 FOR S=1 TO 23:VS=32
350 VPOKE W+X,32
360 Z=W+Y-(VS*S)
370 VPOKE Z,46
380 X=X+1
390 VPOKE W+X,101
400 IF W+X=Z THEN GOTO 220
410 VPOKE Z,32
420 NEXT S
430 RETURN
440 FOR TD=1 TO 300:NEXT TD

```

```

450 LEV=LEV+1:IF W+X<6880 THEN SC=SC+1
460 IF LEV=10 THEN 490
470 LUS=LUS-5
480 GOTO 60
490 CLS
500 LOCATE 4,10
510 PRINT"JE SCOORDE:";SC;"PUNTEN"
520 END
530 REM INITIALISATIE
540 CLS
550 COLOR 1,3,4
560 CLS
570 LOCATE 0,7
580 K$="ffffff"
590 L$="f      f"
600 M$="      f"
610 O$="f      "
620 P$="  f  "
630 FOR A=9 TO 0 STEP -1
640 FOR Z=1 TO 10
650 READ F
660 IF F=1 THEN PRINT "          ";K$
670 IF F=2 THEN PRINT "          ";L$
680 IF F=3 THEN PRINT "          ";M$
690 IF F=4 THEN PRINT "          ";O$
700 IF F=5 THEN PRINT "          ";P$
710 NEXT Z
720 FOR TD=1 TO 300:NEXT TD
730 CLS
740 FOR TD=1 TO 30:NEXT TD
750 LOCATE 0,7
760 NEXT A
770 FOR TD=1 TO 100:NEXT TD
780 LOCATE 0,6
790 PRINT "  fff  ff f  f  f  ffff  fffff"
800 PRINT "  f    f  f  f  f f      f"
810 PRINT "  f    f  f  f  f f      f"
820 PRINT "  f  f  ffff  f fff  f"
830 PRINT "    f f  f  f  f f      f"
840 PRINT "    f f  f  f  f f      f"
850 PRINT "  fff  ff f  f  f  ffff  f"
860 PRINT:PRINT:PRINT"          DOOR DANIEL QUINN"
870 FOR A=1 TO 30:IF A/2=INT(A/2) THEN C=2:B=0:
      GOTO 890

```

```

880 C=0:B=1:LUS=50
890 COLOR 2,C,B
900 FOR TD=1 TO 10:NEXT TD:NEXT A
910 COLOR 15,4,4
920 CLS
930 RETURN
940 REM HERDEFINIERING VAN KARAKTERS
950 FOR N=100 TO 102
960 FOR M=0 TO 7
970 READ B
980 VPOKE N*8+M,B
990 NEXT M
1000 NEXT N
1010 RETURN
1020 DATA 6,12,24,60,60,126,255,126
1030 DATA 3,134,204,252,254,48,96,192
1040 DATA 255,255,255,255,255,255,255,255
1050 REM DATA VOOR HET AFTELLEN
1060 DATA 1,2,2,2,2,1,3,3,3,3
1070 DATA 1,2,2,2,2,1,2,2,2,1
1080 DATA 1,3,3,3,3,3,3,3,3,3
1090 DATA 1,4,4,4,4,1,2,2,2,1
1100 DATA 1,4,4,4,4,1,3,3,3,1
1110 DATA 2,2,2,2,2,1,3,3,3,3
1120 DATA 1,3,3,3,3,1,3,3,3,1
1130 DATA 1,3,3,3,3,1,4,4,4,1
1140 DATA 5,5,5,5,5,5,5,5,5,5
1150 DATA 1,2,2,2,2,2,2,2,2,1

```


AVONTUUR EN SIMULATIE

Overleven

Het Romeinse Rijk

Slag om Engeland


Eigenheimerheerd

OVERLEVEN

Dit avonturenprogramma, geschreven door Doug Rolfe, is anders dan anders.

U bent passagier op een vlucht naar een ander melkwegstelsel. Het ruimteschip is aangevallen en geplunderd door ruimtepiraten. De bemanning is gedood en een paar passagiers die meevlogen, zijn ontvoerd. U bent de enige die nog aan boord is. Om te overleven moet u in de controlekamer zien te komen, het ruimteschip weer op gang brengen en een veilige plaats zien te bereiken. Wees alstublieft voorzichtig, want een ruimteschip op hol heeft vele gevaren in petto voor iemand die er onbekend mee is.

OVERLEVEN is niet vertaald uit het Engels. De oorzaak daarvan ligt bij de verschillende attributen die in het spel een rol spelen. Die worden namelijk door het programma aan hun Engelse namen herkend. De schrijfwijze van de woorden is daarbij doorslaggevend. Vertaling in het Nederlands zou er daardoor op neerkomen dat het complete programma moet worden herschreven. U kunt dat zelf proberen, maar dan hebt u de oorspronkelijk structuur nodig. Het belangrijkste argument om dit avontuur niet in het Nederlands te vertalen is dat een vertaalde versie pas na zeer veel experimenteren even sterke speelmogelijkheden heeft als het programma waarover u nu beschikt.



Voor alle duidelijkheid volgt hieronder wel de vertaling van alle in het programma voorkomende Engelse termen.

nothing	niets
the hull	het omhulsel
a closed door	een dichte deur
an open door	een open deur
an airlock	een luchtsluis
a hand-print controlled door	een deur die beveiligd is door een handaftaster (sensor)
a locked locker	een gesloten slot
an open locker	een open slot
autopilot instructiontape	instructietape voor automatische piloot
space suit	ruimtepak
doorkey	deursleutel
laser cutter	laser-snijapparaat
backpack	rugzak
lockerkey	sleutel van een slot
sonic screwdriver	sonische schroevendraaier
logic probe	logische sonde
pliers	buigtang
warhead	kernkop
detonator	ontsteker
nuclear charge	kernlading
helmet	helm
gloves	handschoenen
boots	laarzen
laser pistol	laserpistool
coursetape	instructietape
missile casing	verpakking van een raket
knife	mes
scalpel	ontleedmes
handtranciever	handontvanger
powerpack	batterijenset
grappling hook	enterhaak
binoculars	verrekijker
body	(in dit geval) lijk
laserrifle	lasergeweer
riflecharge	geweerlading
cable	kabel
food pack	voedselpakket

Werkwoorden die in het programma zijn gebruikt:

go	ga
get	neem
convert	zet om

continue	ga door
unlock	ontsluit
open	open
don't	doe... niet
wear	draag (kleding)
use	gebruik
fire	vuur, schiet
ignore	negeer
assemble	stel samen
detect	vind (zoekend)
disable	stel buiten werking
detonate	ontsteek

Verder komt u tegen:

noseward	voorwaarts
tailward	achterwaarts
spinward	rechtsom
antispinward	linksom

Bevelen moeten als twee woorden worden ingevoerd: eerst een werkwoord, zoals GET of GO, gevolgd door een zelfstandig naamwoord of een richting, zoals BOOTS of NOSEWARD. U hebt vier mogelijkheden wat betreft de richting: NOSEWARD en TAILWARD (bedoeld wordt richting voorschip en richting achterschip) en SPINWARD en ANTISPINWARD. Deze twee laatste richtingen hebben betrekking op de richting waarin het ruimteschip rond de as draait om een kunstmatige zwaartekracht op te wekken. SPINWARD is tegen de draairichting van het schip in en ANTISPINWARD is met de draairichting van het schip mee.

Het scherm geeft alle informatie die u in de loop van het spel nodig heeft. Er wordt steeds verteld wat u kunt zien in de vier richtingen en daarbij wordt u geïnformeerd over eventuele in de buurt liggende voorwerpen. Dan wordt naar uw bevel gevraagd.

We willen het spel niet te gemakkelijk maken, dus geven we niet meer informatie. Toch zijn er nog twee dingen die u moet weten: let erop dat deuren en kasten die op slot zitten, van het slot gedaan moeten worden (UNLOCK). Deuren die dicht zitten, moeten geopend worden (OPEN). De OPEN- en UNLOCK-routines zijn verschillend. Als u een deur wilt OPENEN die op slot is, kan het voorkomen dat er wordt meegedeeld dat er geen deur is. Het programma zoekt naar een dichte deur en niet naar een deur die op slot is. Is er geen dichte deur, dan neemt het programma uiteraard aan dat u een fout heeft gemaakt. Natuurlijk kunt u pas een deur of kast van het slot doen als u de juiste sleutel heeft.

Tenslotte moet u voorzichtig zijn met wat u oppakt. Sommige voorwerpen kunt u beter niet bij u hebben en wat u eenmaal heeft gepakt, kunt u niet meer wegleggen. Veel succes!

```

10 REM OVERLEVEN
20 CLS
30 GOSUB 2310
40 WL$(0)="A WALL":WL$(1)="NOTHING":WL$(2)="THE
   HULL":WL$(3)="A CLOSED DOOR"
50 WL$(4)="A LOCKED DOOR":WL$(5)="AN OPEN DOOR":
   WL$(6)="AN AIRLOCK"
60 WL$(7)="A HAND-PRINT CONTROLLED DOOR":WL$(8)=
   "A LOCKED LOCKER"
70 WL$(9)="AN OPEN LOCKER":DR$(1)="NOSEWARD":
   DR$(2)="TAILWARD"
80 DR$(3)="ANTISPINWARD":DR$(4)="SPINWARD"
90 DIM O$(5,15),D$(5,15),OB$(35)
100 CLS:LOCATE 8,11:PRINT"PLEASE WAIT..."
110 N=14:N2=51:DIM V$(N),N$(N2):X=1:Y=1
120 FOR XX=1 TO N:READ V$(XX):NEXT XX
130 FOR XX=1 TO N2:READ N$(XX):NEXT XX
140 FOR XX=1 TO 5:FOR YY=1 TO 15:READ D$(XX,YY):
   NEXT YY:NEXT XX
150 FOR XX=1 TO 5:FOR YY=1 TO 15:READ O$(XX,YY):
   NEXT YY:NEXT XX
160 FOR XX=0 TO 34:READ OB$(XX):NEXT XX
170 CLS
180 PRINT
190 GOSUB 1090
200 PRINT
210 PRINT"YOUR COMMAND PLEASE";:INPUT A$
220 GOSUB 2450
230 PRINT:PRINT
240 F$=LEFT$(A$,3)
250 IF F$="INV" THEN FOR XX=0 TO NP:PRINT PO$(XX):
   NEXT XX:GOTO 200
260 IF LEN(A$)<4 THEN PRINT"I DON'T UNDERSTAND":
   GOTO 200
270 FOR P=1 TO LEN(A$):P$=MID$(A$,P,1):IF P$=" "
   THEN GOTO 290
280 NEXT P:PRINT:PRINT"PLEASE ENTER A TWO WORD
   SENTENCE":GOTO 200
290 S$=MID$(A$,P+1,3)
300 FOR XX=1 TO N:IF V$(XX)=F$ THEN F=1:V=XX
310 NEXT XX
320 IF F=0 THEN PRINT"NO RECOGNISABLE VERB":GOTO
   200
330 F=0

```

```

340 FOR XX=1 TO N2:IF N$(XX)=S$ THEN F=1:NO=XX
350 NEXT XX
360 IF F=0 THEN PRINT"NO RECOGNISABLE NOUN":GOTO
    200
370 F=0
380 ON V GOTO 410,510,1170,750,1020,1380,1380,
    1970
390 ON (V-8) GOTO 1810,1810,1810,1730,1850,1970
400 PRINT"YOU CAN'T DO THAT":GOTO 200
410 IF MID$(D$(X,Y),NO,1)="1" THEN GOTO 450
420 IF MID$(D$(X,Y),NO,1)="5" THEN GOTO 450
430 IF MID$(D$(X,Y),NO,1)="9" THEN GOTO 450
440 GOSUB 2390:PRINT"THUMP !!!!":GOTO 180
450 ON NO GOTO 470,480,490,500
460 PRINT"THAT DOESN'T MAKE SENSE":GOTO 200
470 Y=Y+1:GOTO 180
480 Y=Y-1:GOTO 180
490 X=X+1:GOTO 180
500 X=X-1:GOTO 180
510 IF K$="NOTHING" THEN PRINT "THERE IS NOTHING
    HERE":GOTO 180
520 FOR P=1 TO LEN(K$):PP$=MID$(S$,P,S):IF PP$<>
    "0" THEN GOTO 540
530 NEXT P:PRINT"I CAN'T SEE ANY SUCH OBJECT":
    GOTO 180
540 IF NP=7 THEN PRINT "YOU CAN'T CARRY ANY
    MORE":GOTO 200
550 IF K$="BODY" THEN PRINT " THE BODY IS TOO
    HEAVY TO CARRY":GOTO 180
560 IF K$="COMPUTER" THEN PRINT"YOU CANT'T, IT'S
    FIXED TO THE WALL":GOTO 180
570 IF K$="AUTOPILOT INSTRUCTION TAPE" THEN FT=1
580 IF K$="SPACE SUIT" THEN FS=1
590 IF K$="DOOR KEY" THEN DK=1
600 IF K$="LASER CUTTER" OR K$="POWER PACK" THEN
    LF=LF+1
610 IF K$="BACK PACK" THEN FS(5)=1
620 IF K$="LOCKER KEY" THEN LK=1
630 IF K$="SONIC SCREWDRIVER" THEN T1=1
640 IF K$="LOGIC PROBE" THEN T2=1
650 IF K$="PLIERS" THEN T3=1
660 IF K$="WARHEAD" THEN W1=1:EX=1
670 IF K$="DETONATOR" THEN W2=1:EX=1
680 IF K$="NUCLEAR CHARGE" THEN W3=1:EX=1

```



```

690 IF K$="HELMET" THEN FS(1)=1
700 IF K$="SPACE SUIT" THEN FS(2)=1
710 IF K$="GLOVES" THEN FS(3)=1
720 IF K$="BOOTS" THEN FS(4)=1
730 NP=NP+1:PO$(NP)=K$:O$(X,Y)="0"
740 PRINT"YOU HAVE";RR$;" ";K$:GOTO 180
750 IF NO<>14 THEN GOTO 830
760 PRINT"WHERE";INPUT A$:A$=LEFT$(A$,3)
770 FOR XX=1 TO 4:IF LEFT$(DR$(XX),3)=A$ THEN
GOTO 790
780 NEXT XX:PRINT"THAT DOESN'T MAKE SENSE":GOTO
180
790 IF MID$(D$(X,Y),XX,1)<>"4" THEN PRINT"THERE
IS NO LOCKED DOOR";DR$(XX):GOTO 180
800 IF DK<>1 THEN PRINT"YOU DON'T HAVE THE KEY":
GOTO 180
810 PRINT"DOOR UNLOCKED":D$(X,Y)=LEFT$(D$(X,Y),XX
-1)+"3"+RIGHT$(D$(Y,X),4-XX)
820 NW$="3":ON XX GOTO 980,1000,1010
830 IF NO<>15 THEN GOTO 910
840 PRINT"WHERE";:INPUT A$:A$=LEFT$(A$,3)
850 FOR XX=1 TO 4:IF LEFT$(DR$(XX),3)=A$ THEN
GOTO 870
860 NEXT XX:PRINT"THAT DOESN'T MAKE SENSE":GOTO
180
870 IF MID$(D$(X,Y),XX,1)<>"8" THEN PRINT"NO
LOCKED LOCKER ";DR$(XX):GOTO 180
880 IF LK<>1 THEN PRINT"YOU DON'T HAVE THE KEY!!":
GOTO 180
890 D$(X,Y)=LEFT$(D$(X,Y),XX-1)+"9"+RIGHT$(D$(X,
Y),4-XX)
900 NW$="9":ON XX GOTO 980,990,1000,1010
910 PRINT"WHERE";:INPUT A$:A$=LEFT$(A$,3)
920 FOR XX=1 TO 4:IF LEFT$(DR$(XX),3)=A$ THEN
GOTO 940
930 NEXT XX:PRINT"THAT DOESN'T MAKE SENSE":GOTO
180
940 IF MID$(D$(X,Y),XX,1)<>"6" THEN PRINT"THERE
IS NO AIRLOCK ";DR$(XX):GOTO 180
950 IF RF=1 AND FS=2 AND X<>3 THEN PRINT"YOU MADE
IT!!":END
960 IF FS<>2 THEN PRINT"YOU EXPLODED IN THE
VACUUM!!":END
970 PRINT"YOU HAD NOWHERE TO GO AND YOU FLOATED

```

```

    TO YOUR DEATH!!":END
980 D$(X,Y+1)=LEFT$(D$(X,Y+1),1)+NW$+RIGHT$(D$(X,
    Y+1),2):GOTO 180
990 D$(X,Y-1)=NW$+RIGHT$(D$(X,Y-1),3):GOTO 180
1000 D$(X+1,Y)=LEFT$(D$(X+1,Y),3)+NW$:GOTO 180
1010 D$(X-1,Y)=LEFT$(D$(X-1,Y),2)+NW$+RIGHT$(D$(X
    -1,Y),1)+NW$:GOTO 180
1020 PRINT"WHERE";:INPUT A$:A$=LEFT$(A$,3)
1030 FOR XX=1 TO 4:IF LEFT$(DR$(XX),3)=A$ THEN
    GOTO 1050
1040 NEXT XX:PRINT"THAT DOESN'T MAKE SENSE":GOTO
    180
1050 IF MID$(D$(X,Y),XX,1)<>"3" THEN PRINT"NO
    CLOSED DOOR ";DR$(XX):GOTO 180
1060 PRINT"DOOR OPEN":D$(X,Y)=LEFT$(D$(X,Y),XX
    -1)+"5"+RIGHT$(D$(X,Y),4-XX)
1070 NW$="5":ON XX GOTO 980,990,1000,1010
1080 GOTO 180
1090 FOR TD=1 TO 1000:NEXT TD:CLS:FOR DR=1 TO 4:
    PRINT
1100 PRINT DR$(DR);" YOU SEE ";WL$(VAL(MID$(D$(X,
    Y),DR,1)))
1110 NEXT DR
1120 IF VAL(O$(X,Y))<>0 THEN K$=OB$(VAL(O$(X,Y)))
    :GOTO 1150
1130 IF O$(X,Y)<>"0" THEN K$="NOTHING":RR$="":GOTO
    1150
1140 K$=OB$(ASC(O$(X,Y))-47)
1150 PRINT:PRINT"YOU SEE ";K$;" HERE"
1160 RETURN
1170 IF O$(X,Y)<>"P" THEN PRINT"THERE IS NO
    COMPUTER HERE":GOTO 180
1180 CLS:PRINT"DO YOU WISH TO SET A COURSE (Y/N)"
1190 A$=INKEY$:IF A$="" THEN GOTO 1190
1200 IF A$="Y" THEN GOTO 1230
1210 PRINT"I HAVE NO OHER FUNCTION,"
1220 PRINT"DISCONNECTION PROCEEDS":GOTO 180
1230 IF FT=1 THEN GOTO 1250
1240 PRINT"IMPOSSIBLE WIHOUT":PRINT"AUTO PILOT
    COURSE TAPE":GOTO 180
1250 PRINT"COURSE SET:":PRINT"DO YOU WISH TO"
1260 PRINT"PROCEED TO DESTINATION (Y/N)?"
1270 A$=INKEY$:IF A$="" THEN GOTO 1270
1280 IF A$="N" THEN GOTO 180

```

```

1290 IF A$<>"Y" THEN GOTO 1270
1300 CLS:PRINT"HYPERDRIVE IN OPERATION..."
1310 FOR XX=1 TO 2000:NEXT XX:CLS:RF=1:PRINT:
PRINT"PRESENT STATUS:"
1320 PRINT"ORBITING 4TH PLANET OF":PRINT"STAR
'CANOPUS'."
1330 PRINT"SENTRY CRAFT OF THIS":PRINT"PLANET
REQUEST DOCKING"
1340 PRINT:PRINT"DO YOU WISH TO COMPLY? (Y/N)":
PRINT"ADVISABLE"
1350 A$=INKEY$:IF A$="" THEN 1350
1360 IF A$="Y" THEN RF=1:PRINT"DOCKED: PROCEED
TO AIRLOCK '2'":GOTO 180
1370 PRINT:PRINT"YOU WHERE SHOT OUT OF THE SKY!!
!!!!":END
1380 IF F2=2 THEN PRINT"YOU ALREADY HAVE A COMPLETE
SUIT":GOTO 200
1390 IF NO=34 THEN GOTO 1450
1400 IF NO=11 OR NO=12 THEN GOTO 1500
1410 IF NO=35 THEN GOTO 1550
1420 IF NO=36 THEN GOTO 1600
1430 IF NO=37 OR NO=35 THEN GOTO 1660
1440 PRINT"THAT DOESN'T MAKE SENSE!!":GOTO 180
1450 IF FS(1)<>1 THEN PRINT"YOU DON'T HAVE A
HELMET":GOTO 1710
1460 PRINT"YOU ARE NOW WEARING THE HELMET":FOR
XX=1 TO NP
1470 IF PO$(XX)="HELMET" THEN PO$(XX)=PO$(NP):
NP=NP-1:FS(1)=0
1480 NEXT XX
1490 F3=F3+1:GOTO 1710
1500 IF FS(2)<>1 THEN PRINT"YOU DON'T HAVE A
SPACE SUIT":GOTO 1710
1510 PRINT"YOU ARE NOW WEARING THE SPACE SUIT":
FOR XX=1 TO NP
1520 IF PO$(XX)="SPACE SUIT" THEN PO$(XX)=PO$(NP
):NP=NP-1:FS(2)=0
1530 NEXT XX
1540 F3=F3+1:GOTO 1710
1550 IF FS(3)<>1 THEN PRINT"YOU DON'T HAVE
GLOVES":GOTO 1710
1560 PRINT"YOU ARE NOW WEARING THE GLOVES":FOR
XX=1 TO NP
1570 IF PO$(XX)="GLOVES" THEN PO$(XX)=PO$(NP):

```

```

      NP=NP-1:FS(3)=0
1580 NEXT XX
1590 F3=F3+1:GOTO 1710
1600 IF FS(4)<>1 THEN PRINT"YOU DON'T HAVE BOOTS":
      GOTO 1710
1610 PRINT"YOU ARE NOW WEARING THE BOOTS":FOR
      XX=1 TO NP
1620 IF PO$(XX)="BOOTS" THEN PO$(XX)=PO$(NP):
      NP=NP-1:FS(4)=0
1630 NEXT XX
1640 F3=F3+1:GOTO 1710
1650 PRINT"YOU ARE WEARING THE BOOTS":FS2=FS2+1:
      GOTO 180
1660 IF FS(5)<>1 THEN PRINT"YOU DON'T HAVE A BACK
      PACK":GOTO 1710
1670 PRINT"YOU ARE NOW WEARING THE BACK PACK":FOR
      XX=1 TO NP
1680 IF PO$(XX)="BACK PACK" THEN PO$(XX)=PO$(NP):
      NP=NP-1:FS(5)=0
1690 NEXT XX
1700 F3=F3+1
1710 IF F3=5 THEN F2=2:NP=NP+1:PO$(NP)="FULL SPACE
      SUIT"
1720 GOTO 180
1730 IF NO<>(16 AND 27) THEN PRINT"YOU CAN'T":GOTO
      180
1740 IF LF<>2 THEN PRINT"YOU CAN'T":GOTO 180
1750 FOR XX=1 TO NP:IF PO$(XX)="LASER CUTTER" THEN
      POS$(XX)="CHARGED CUTTER":LF=3
1760 NEXT XX
1770 FOR XX=1 TO NP:IF PO$(XX)="POWER PACK" THEN
      POS$(XX)=PO$(NP):NP=NP-1
1780 NEXT XX
1790 PRINT"YOU HAVE A FULLY CHARGED LASER CUTTER
      BE CAREFUL!!"
1800 GOTO 180
1810 IF LF=0 THEN PRINT"YOU DON'T HAVE ONE":GOTO
      180
1820 IF LF<>3 THEN PRINT"NOTHING HAPPENED!":GOTO
      180
1830 PRINT"YOU HAVE BURNED A HOLE RIGHT THROUGH
      THE HULL "
1840 PRINT:PRINT"YOU HAVE DIED IN THE VACUUM!":END
1850 IF EX<>1 THEN PRINT"YOU HAVE NOTHING TO

```

```

    DETONATE":GOTO 180
1860 IF NO=19 AND W1=1 THEN PRINT"THE WARHEAD HAS
    DESTROYED THE SHIP"
1870 FOR TD=1 TO 1000:NEXT TD
1880 PRINT:PRINT:PRINT"AND YOU ALONG WITH IT":END
1890 IF NO=18 AND W2=1 THEN PRINT"THE DETONATOR HAS
    BLOWN UP"
1900 FOR TD=1 TO 1000:NEXT TD
1910 PRINT"IT HAS KILLED YOU":END
1920 IF W3=1 AND(NO=16 OR NO=27) THEN GOTO 1940
1930 PRINT"IMPOSSIBLE!!":GOTO 180
1940 PRINT"ONLY A FOOL WOULD DETONATE A NUCLEAR
    WEAPON INSIDE A SPACESHIP"
1950 FOR TD=1 TO 2500:NEXT TD
1960 PRINT:PRINT:PRINT"BUT YOU DID!!!":END
1970 IF NO=14 OR NO=21 THEN GOTO 1990
1980 PRINT"YOU CAN'T":GOTO 180
1990 IF T1<>1 OR T2<>1 OR T3<>1 THEN PRINT"YOU
    DON'T HAVE THE TOOLS":GOTO 180
2000 IF X<>3 OR Y<>12 THEN PRINT"HOW?!!":GOTO 180
2010 PRINT"YOU HAVE SUCCESFULLY DISMANTLED THE
    DOOR"
2020 D$(3,12)="3011":D$(3,13)="1311":GOTO 180
2030 DATA T255O5G,E,O6G,O5E,G,E,O6G,O5E,D,O5D,C,
    D,E,D,C
2040 DATA "GO ",GET,CON,UNL,OPE,DON,WEA,LIG,USE,
    FIR,IGN,ASS,DET,DIS
2050 DATA NOS,TAI,ANT,SPI,NUC,CHA,RIF,PIS,KEY,TAP,
    SPA,SUI,COM,DOO,LOC,CUT
2060 DATA AIR,DET,WAR,TRA,HAN,SCA,LOG,PRO,AUT,INS,
    LAS
2070 DATA MIS,CAS,SCA,KNI,POW,PAC,HEL,GLO,BOO,BAC,
    GRA,HOO,UNI,BIN
2080 DATA BOD,CAB,FOO,WAT,VIT,SON,SCR,PLI,LOG,PRO
2090 DATA 1212,1182,1182,0182,0046,1092,1182,0182,
    1012,1192,1192,8112
2100 DATA 0812,0002,0002
2110 DATA 0211,0088,0088,0088,1014,1109,1108,1108,
    0101,1009,0109,0011
2120 DATA 1011,0110,0000
2130 DATA 0611,1018,1118,4118,0411,1010,4110,3410,
    1310,1100,0110,7011
2140 DATA 1711,1111,2100
2150 DATA 0211,1001,1101,0101,0011,1001,0101,0011,

```

```


      8001,0800,0001,0011
2160 DATA 1001,0101,0000
2170 DATA 1221,1120,1120,1120,1121,1120,1120,1121,
      1120,1120,1120,8121
2180 DATA 0820,0020,0020
2190 DATA 0,0,N,M,B,3,D,C,0,@,0,0,8,0,0
2200 DATA 0,A,V,K,0,0,0,0,2,E,F,0,I,T,0
2210 DATA 0,G,1,0,Y,5,J,6,0,R,0,0,Q,L,P
2220 DATA ;,0,0,0,=,9,H,W,X,4,<,0,Z,S,0
2230 DATA 0,0,7,0,0,0,0,0,0,0,>,0,0,U,0,0
2240 DATA NOTHING,TORCH,DOOR KEY,LOCKER KEY,LOGIC
      PROBE
2250 DATA SONIC SCREWDRIVER,PLIERS,LASER CUTTER,
      LASER PISTOL,SCANNER
2260 DATA NUCLEAR CHARGE,AUTOPILOT COURSE TAPE,
      WARHEAD
2270 DATA DETONATOR,MISSILE CASING,SCALPEL,KNIFE,
      HAND TRANCEIVER,ANTENNA
2280 DATA POWER PACK,SPACE SUIT,HELMET,GLOVES,
      BOOTS,BACK PACK,COMPUTER
2290 DATA GRAPPLING HOOK,UNIFORM,BINOCULARS,BODY,
      LASER RIFLE
2300 DATA RIFLE CHARGE,CABLE,FOOD PACK,WATER PACK,
      VITAMIN PACK,CABLE
2310 REM TITEL
2320 SCREEN 0:P=RND(-TIME)
2330 LOCATE 13,11:PRINT"SURVIVAL"
2340 FOR LUS=1 TO 15
2350 READ M$
2360 PLAY M$,M$,M$
2370 NEXT LUS
2380 RETURN
2390 Z=INT(RND(1)*60+36)
2400 FOR LUS=Z TO Z-10 STEP-1
2410 PLAY "N=LUS;"
2420 NEXT LUS
2430 RETURN
2440 REM GELUID
2450 FOR Z=60 TO 70
2460 PLAY "N=Z;"
2470 NEXT Z
2480 RETURN

```

HET ROMEINSE RIJK

Dit programma biedt de mogelijkheid om voor een poosje de machtigste man van de klassieke wereld te zijn: de Keizer van het Romeinse Rijk! Om succes te hebben, moet u alle beschikbare bronnen aanboren om een sterk en welvarend rijk op te bouwen. U moet een evenwicht zoeken tussen handel en militaire kracht, terwijl u de bevolking moet voorzien van voldoende voedsel.

Elke spelronde komt overeen met een jaar. Aan het begin van elk jaar ontvangt u een rapport van de Senaat, waarin de toestand van het Rijk voor het komende jaar beschreven staat. Dit rapport is erg belangrijk en moet zorgvuldig worden bestudeerd. Het geeft informatie over de bevolkingsgrootte, de omvang van leger, marine en handelsvloot en de hoeveelheid arbeid die voor de scheepsbouw beschikbaar is. Ook wordt verteld hoeveel handesgildes zich bezighouden met de productie van goederen voor de export en hoeveel arbeid er nog niet toegewezen is en die u in het komende jaar moet inzetten. Het rapport laat verder zien hoeveel goudreserves er zijn, hoeveel voedsel er opgeslagen ligt en de benodigde hoeveelheid voedsel om de bevolking te voeden.



De eerste taak aan het begin van elke ronde is om de nog niet toegewezen arbeid in te zetten. Dit is zeer belangrijk, vooral in de eerste jaren, als het Rijk nog maar klein is en hard moet vechten. Ook is het belangrijk om de juiste verhoudingen tussen de verschillende sectoren (leger, marine en handwerkslieden) te handhaven. De omvang van het leger bepaalt of u wel of niet in staat bent oorlogen te winnen of opstanden in de provincies neer te slaan. Wat de waarde van de exporthandel zal zijn hangt af van de uitkomst van deze veldslagen.

U moet ervoor zorgen dat u voldoende bemanningen heeft voor de schepen die u gaat bouwen. Elk scheepsbouwcohort bouwt 1 schip per jaar. Het is dus niet verstandig om 40 cohorten te hebben, als u maar 25 bemanningen heeft. Het is natuurlijk ook verspilling wanneer de bemanningen moeten wachten op de scheepsbouwers omdat er onvoldoende schepen zijn. Daarom moet u voldoende arbeid inzetten voor de marine en de handelsvloot om de schepen die worden gebouwd, te kunnen bemannen.

Een opmerking: u kunt arbeid aftrekken van elke sector door een negatief getal in te voeren. U kunt ook naar hartelust arbeid verschuiven totdat u alle arbeidsreserve verdeeld heeft. Dit kan meteen al in de eerste ronde nodig zijn om een onevenwichtige beginsituatie te corrigeren. Als u geen arbeid meer heeft om in te zetten (waarde 0), gaat het programma over naar de volgende fase.

Elke gilde van handwerkslieden produceert voldoende exportgoederen om een schip te vullen. Met deze handelsgoederen verdient u uw inkomen. De plaatselijke handelsschepen zorgen voor het voedsel voor uw volk.

Nadat alle arbeid is ingezet, krijgt u de gelegenheid om hulptroepen (Auxilia) te huren om het leger uit te breiden. Auxilia zijn duurder dan gewone legioenen en ze gaan niet altijd even rustig weg wanneer ze ontslagen worden.

De volgende routine behandelt de scheepsbouw. Zoals gezegd, zorg ervoor dat u bemanningen hebt voor al uw schepen. U kunt naar eigen inzicht de schepen verdelen over de marine en de handelsvloot. Er wordt gevraagd hoeveel schepen u over zee wilt sturen en hoeveel oorlogsschepen nodig zijn als escorte. Uw handelsschepen lopen altijd kans door piraten aangevallen te worden, dus is het raadzaam oorlogsschepen als escorte mee te sturen. Elk oorlogsschip dat u niet meestuurt, wordt geacht in de kustwateren te patrouilleren om de plaatselijke handelsschepen te beschermen. Alle oorlogsschepen met de overzeese handelsschepen meesturen, betekent dan ook dat de kans groot is dat plaatselijke handelsschepen verloren gaan.

Nadat u al deze beslissingen heeft genomen, wordt u geïnformeerd over mogelijke oorlogen en opstanden in het Rijk. Aan het eind van dit deel wordt u meegedeeld hoe de strijdkrachten het hebben gedaan en hoe groot de verliezen zijn. Als u een veldslag verliest, kunnen de verliezen hoog oplopen! De computer geeft een volledige lijst van verliezen en veranderingen in de handelswaarde na elk gevecht.

In de volgende routine wordt verteld hoe groot de schade is die de piraten aan de handelslieden hebben toegebracht. Deze verliezen kunnen worden vermindert door de handelsvloot te beschermen met een sterke vloot oorlogsschepen.

Hierna krijgt u een compleet verslag van de financiële transacties in dat jaar, inclusief de handelsinkomsten en de uitgaven voor het leger. En wat voor u als keizer natuurlijk van het meeste belang is; de laatste regel geeft u een overzicht van de goudvoorraad.

Nu moet de bevolking voedsel hebben. Het programma vertelt u hoeveel voedsel de plaatselijke handelsschepen hebben teruggebracht en of dit genoeg was. Is dat niet het geval, dan moet u meer graan gaan kopen tegen enorme woekerprijzen.

De laatste routine in elke ronde geeft de mate van populariteit aan. U krijgt een score in punten, die is gebaseerd op de toestand van uw Rijk en uw poging om uit te breiden. Deze score kan erg op en neer gaan tijdens het spel. Pas wel op dat u het niet al te slecht doet als keizer, dat zou kunnen betekenen dat u vroegtijdig een einde aan uw keizerschap maakt.

```
10 REM HET ROMEINSE RIJK
20 GOSUB 150:REM INITIALISATIE
30 GOSUB 3240:REM START
40 GOSUB 320:REM WERKVERDELING
50 GOSUB 650:REM HUURLINGEN INHUREN
60 GOSUB 970:REM SCHEPEN BOUWEN
70 GOSUB 1180:REM HANDEL BEPALEN
80 GOSUB 1340:REM OORLOGEN
90 GOSUB 1590:REM WIN- OF VERLIESPROCEDURE
100 GOSUB 2190:REM MUITERIJ
110 GOSUB 2330:REM HANDEL DRIJVEN
120 GOSUB 2480:REM VOEDSEL VOOR DE INWONERS
130 GOSUB 2640:REM POLITIEK GEDEELTE
140 GOTO 2790:REM EINDROUTINE OF OVERWINNING
150 KEY OFF
160 CLS:COLOR 15,1:TR=50000!:TN=-55:PO=100000!
170 P=RND(-TIME)
180 MS=INT(20*RND(1))+111
190 WS=INT(10*RND(1))+30
200 FS=INT(10*RND(1))+10
210 VL=100:FR=100:FL=1
220 LA=INT(10*RND(1))+6
230 MA=INT(6*RND(1))+5
240 LE=INT(10*RND(1))+11
250 LN=WS:LM=MS
260 LS=INT(30*RND(1))+16
```

```

270 UL=250-LS-LE-LM-LN-LA
280 IF UL<0 THEN GOTO 180
290 SCREEN 0
300 M1$="T10003ABA":M2$="T8004CDC":
    M3$="T25505EC"
310 RETURN
320 IF FL=0 THEN CLS:X=3:Y=10:GOSUB 3180:
    PRINT"TOETSING BENODIGDE WERKKRACHTEN"
330 FOR Z=1 TO 500:NEXT Z:FL=0
340 IF UL<>0 THEN GOSUB 2980
350 IF UL=0 THEN X=3:Y=13:GOSUB 3180:PRINT"
    ARBEIDSRESERVE VERDEELD"
360 IF UL=0 THEN GOSUB 3140
370 IF UL=0 THEN RETURN
380 CLS:PRINT:PRINT"ARBEIDSRESERVE =";UL
390 PRINT:PRINT"WAAR WILT U DIE INZETTEN?"
400 PRINT:PRINT"(1) = LEGER":PRINT"(2) = MARINE"
410 PRINT"(3) = HANDELSVLOOT":PRINT"(4) =
    HANDWERKSLIEDEN"
420 PRINT"(5) = SCHEEPSBOUWERS"
430 INPUT A
440 IF A<1 OR A>5 THEN GOTO 380
450 IF A=1 THEN GOTO 500
460 IF A=2 THEN GOTO 530
470 IF A=3 THEN GOTO 560
480 IF A=4 THEN GOTO 590
490 IF A=5 THEN GOTO 620
500 PRINT:PRINT"EENHEDEN VOOR HET LEGER"
510 INPUT B:IF B>UL THEN GOTO 500
520 LA=LA+B:UL=UL-B:GOTO 320
530 PRINT:PRINT"EENHEDEN VOOR MARINE"
540 INPUT B:IF B>UL THEN GOTO 540
550 LN=LN+B:UL=UL-B:GOTO 320
560 PRINT:PRINT"BEMANNING HANDELSVLOOT"
570 INPUT B:IF B>UL THEN GOTO 570
580 LM=LM+B:UL=UL-B:GOTO 320
590 PRINT:PRINT"PRODUKTIEGROEPEN HANDWERK"
600 INPUT B:IF B>UL THEN GOTO 600
610 LE=LE+B:UL=UL-B:GOTO 320
620 PRINT:PRINT"PRODUKTIEGROEPEN SCHEEPSBOUW"
630 INPUT B:IF B>UL THEN GOTO 630
640 LS=LS+B:UL=UL-B:GOTO 320
650 CLS
660 X=0:Y=6:GOSUB 3180

```

```

670 PRINT:PRINT"TE WERVEN EENHEDEN HUURLINGEN =";
    MA
680 PRINT:PRINT"AANTAL EENHEDEN HUURLINGEN ONTSLAAN =
    ";MH
690 PRINT:PRINT"WAT WILT U .....":GOSUB 3160
700 PRINT"(1) = WERF HUURLINGEN"
710 PRINT"(2) = ONTSLA HUURLINGEN"
720 PRINT"(3) = GEEN VAN BEIDE"
730 INPUT B:IF B<1 OR B>3 THEN GOTO 730
740 IF B=1 THEN GOTO 770
750 IF B=2 THEN GOTO 800
760 IF B=3 THEN RETURN
770 PRINT:PRINT"HOEVEEL EENHEDEN WILT U HUREN?"
780 INPUT B:IF B>MA OR B<0 THEN GOTO 770
790 MH=MH+B:MA=MA-B:GOTO 650
800 PRINT "HOEVEEL EENHEDEN WILT U ONTSLAAN?:
810 INPUT B:IF B>MH OR B<0 THEN GOTO 810
820 MH=MH-B:MA=MA+B:MI=B*20:X=INT(100*RND(1))+1
830 COLOR 1,6:PRINT:PRINT"DE ONTSLAGEN HUURLINGEN
    DREIGEN OP TE TREKKEN TEGEN ROME!!"
840 GOSUB 3130:IF MI<X THEN GOTO 940
850 PRINT"DE HUURLINGEN MUITEN!":GOSUB 2940:GOSUB
    3130
860 U=INT(10*RND(1)+1)*1000:V=INT(20*RND(1)*1000):
    W=INT(10*RND(1)+2)
870 PRINT:PRINT"ZE HEBBEN UW SCHAT VAN";U;"GOUDEN
    STAVEN GESTOLEN":GOSUB 3140
880 PRINT:PRINT"EN REGEN";V;"BURGERS AAN HET
    ZWAARD":GOSUB 3140
890 PRINT:PRINT"ARBEIDERS GEDOOD:-":GOSUB 3140
900 PRINT"HANDWERKSLIEDEN=";INT(W/3)
910 PRINT"SCHEEPSBOUWERS =";INT(W/3)
920 TR=TR-U:PO=PO-V:LS=LS-INT(W/3):LE=LE-INT(W/3)
930 GOSUB 3200:GOTO 650
940 PRINT:PRINT"DE HUURLINGEN HEBBEN BESLOTEN"
950 PRINT"OM VREEDZAAM NAAR HUIS TERUG"
960 PRINT"TE KEREN":GOSUB 3130
970 CLS:US=LS
980 IF US=0 THEN PRINT:PRINT"U HEBT GEEN SCHEPEN
    MEER TE BOUWEN":GOSUB 3140:RETURN
990 PRINT:PRINT"SCHEPEN TE BOUWEN =";US
1000 N=LN-WS:MW=LM-MS
1010 PRINT:PRINT"MARINEBEMANNINGEN ZONDER SCHIP
    ";NW

```

```

1020 PRINT:PRINT"KOOPVAARDIJBEMANNINGEN ZONDER
      SCHIP";MW
1030 PRINT:PRINT"WAT WILT U BOUWEN ....."
1040 PRINT"(1) = OORLOGSSCHEPEN":PRINT"(2) =
      KOOPVAARDIJSCHPEPEN"
1050 PRINT"(3) = GEEN"
1060 INPUT B
1070 IF B<0 OR B>3 THEN GOTO 1060
1080 IF B=1 THEN GOTO 1110
1090 IF B=2 THEN GOTO 1140
1100 IF B=3 THEN GOTO 1170
1110 PRINT:PRINT"HOEVEEL OORLOGSSCHEPEN WILT U
      BOUWEN";
1120 INPUT B:IF B<0 OR B>3 THEN GOTO 1120
1130 US=US-B:WS=WS+B:GOTO 980
1140 PRINT:PRINT"HOEVEEL HANDELSSCHEPEN WILT U
      BOUWEN";
1150 INPUT B:IF B<0 OR B>3 THEN GOTO 1150
1160 US=US-B:MS=MS+B:GOTO 980
1170 RETURN
1180 IF LN>WS THEN WA=WS
1190 IF WS>=LN THEN WA=LN
1200 IF MS>LM THEN MU=LM
1210 IF LM>=MS THEN MU=MS
1220 CLS:GOSUB 3150
1230 X=0:Y=6:GOSUB 3180
1240 PRINT"AANTAL BESCHIKBARE OORLOGSSCHEPEN";WA
1250 PRINT"AANTAL BESCHIKBARE HANDELSSCHEPEN";MU
1260 PRINT:PRINT"U KUNT";LE;" HANDELSSCHEPEN NAAR
      VERRE GEBIEDEN STUREN"
1270 PRINT:PRINT"HOEVEEL WILT U ER STUREN";
1280 INPUT B:IF B<0 OR B>LE THEN GOTO 1280
1290 MO=B:ML=MU-B
1300 PRINT"HOEVEEL OORLOGSSCHEPEN MOETEN HEN
      ESCORTEREN";
1310 INPUT B:IF B<0 OR B>WA THEN GOTO 1310
1320 WO=B:WL=WA-B
1330 RETURN
1340 CLS
1350 LOCATE 2,2:PRINT"LET OP OORLOGEN EN
      OPSTANDEN"
1360 GOSUB 3140
1370 PP=INT(1000*RND(-TIME)+1)+(VF*3)
1380 IF PP>950 THEN X=2:Y=6:GOSUB 3180:VF=VF+5

```

```

1390 IF PP>950 THEN PRINT"ER ZIJN GEEN OORLOGEN
      OF OPSTANDEN":GOSUB 3150:GOTO 1550
1400 IF PP<175 THEN A$="OORLOG":B$="MET
      CARTHAGO":GOTO 1500
1410 IF PP<350 THEN A$="OPSTAND":B$="IN
      GALLIE":GOTO 1510
1420 IF PP<450 THEN A$="OPSTAND":B$="IN
      BRITTANIE":GOTO 1510
1430 IF PP<525 THEN A$="OORLOG":B$="MET
      MACEDONIE":GOTO 1520
1440 IF PP<625 THEN A$="OORLOG":B$="MET
      DE PARTHEN":GOTO 1520
1450 IF PP<675 THEN A$="OORLOG":B$="MET
      DE GRIEKEN":GOTO 1530
1460 IF PP<750 THEN A$="OORLOG":B$="MET
      DE PICTEN":GOTO 1530
1470 IF PP<875 THEN A$="OPSTAND":B$="IN
      JUDEA":GOTO 1530
1480 IF PP<925 THEN A$="OPSTAND":B$="IN
      SPANJE":GOTO 1540
1490 A$="OPSTAND":B$="OP CRETA":GOTO 1540
1500 G=INT(15*RND(1))+5:H=1.8:GOTO 1560
1510 G=INT(12*RND(1))+3:H=1.6:GOTO 1560
1520 G=INT(10*RND(1))+2:H=1.4:GOTO 1560
1530 G=INT(8*RND(1))+1:H=1.2:GOTO 1560
1540 G=INT(4*RND(1))+1:H=1:GOTO 1560
1550 G=INT(3*RND(1))+1:GOSUB 3200:PX=100:
      FX=100:RETURN
1560 X=0:Y=10:GOSUB 3180
1570 PRINT A$;" BREEKT UIT ";B$:GOSUB 2940
1580 GOSUB 3200:RETURN
1590 IF PP>950 THEN GOTO 2060
1600 ZB=INT(100*RND(-TIME))+1
1610 IF TN<=-50 THEN ZD=2:GOTO 1650
1620 IF TN<-50 THEN ZD=1.5:GOTO 1650
1630 IF TN<-40 THEN ZD=1:GOTO 1650
1640 IF TN<-35 THEN ZD=.75:GOTO 1650
1650 ZC=INT((LA+MH+(WL/3))*ZD/H)
1660 IF ZC>100 THEN ZC=99
1670 CLS
1680 X=0:Y=8:GOSUB 3180:PRINT"UW KANS OM TE
      WINNEN IS ";ZC;"%"
1690 GOSUB 2940:GOSUB 3140
1700 IF ZB<=ZC THEN GOTO 1830
1710 PRINT:PRINT"**** ROME IS VERSLAGEN ****":

```

```

GOSUB 2960
1720 WW=INT(10*RND(1)+1*H):IF WW>WL THEN WW=WL
1730 MW=INT(20*RND(1)+7*H):IF MW>ML THEN MW=ML
1740 MX=INT(5*RND(1)+2*H):IF MX>MO THEN MX=MO
1750 WX=INT(3*RND(1)+1*H):IF WX>WO THEN WX=WO
1760 AW=INT(6*RND(1)+1*H):IF AW>LA THEN AW=LA
1770 SW=INT(8*RND(1)+1*H):IF SW>MH THEN SW=MH
1780 FX=INT(20*RND(1)+79)
1790 PX=INT(5*RND(1)+95)
1800 SX=INT(3*RND(1)*H):IF SX>LS THEN SX=LS
1810 EX=INT(3*RND(1)*H):IF EX>LE THEN EX=LE
1820 VF=VF-INT(G*2):GOTO 1940
1830 PRINT:PRINT"*** ROME IS OVERWINNAAR ***":
      GOSUB 2960
1840 WW=INT(5*RND(1)+1*H):IF WW>WL THEN WW=WL
1850 MW=INT(10*RND(1)+3*H):IF MW>ML THEN MW=ML
1860 MX=1:IF MO=0 THEN MX=0
1870 WX=0
1880 AW=INT(2*RND(1)+1*H):IF AW>LA THEN AW=LA
1890 SW=INT(3*RND(1)+1*H):IF SW>MH THEN SW=MH
1900 FX=100
1910 PX=100
1920 SX=0
1930 EX=0:VF=VF+INT(G*1.5)
1940 PRINT:PRINT"VERLIEZEN"
1950 PRINT WW;"KUSTBEWAKING"
1960 PRINT WX;"OORLOGSSCHEPEN OVERZEE"
1970 PRINT:PRINT MW;"LOKALE HANDELSSCHEPEN"
1980 PRINT MX;"OVERZEESE HANDELSSCHEPEN"
1990 PRINT:PRINT AW;"LEGIOENEN"
2000 PRINT SW;"HUURLINGEENHEDEN"
2010 PRINT:PRINT 100-FX;"% VAN VOEDSELRESERVE"
2020 PRINT 100-PX;"% VAN DE RIJKSBEVOLKING"
2030 PRINT:PRINT SX;"SCHEEPSBOUWERSGROEPEN"
2040 PRINT EX;"HANDWERKSLIEDEN"
2050 PRINT:PRINT:GOSUB 3200
2060 CLS:X=0:Y=10:GOSUB 3180
2070 IF ZB<=ZC THEN PRINT "ROME'S HANDELSKOERS
      STEEG MET";G;"%"
2080 IF ZB<=ZC THEN VL=VL+G
2090 IF ZB>ZC THEN PRINT "ROME'S HANDELSKOERS
      ZAKTE MET";G;"%"
2100 IF ZB>ZC THEN VL=VL-G

```

```

2110 PRINT:PRINT"DE HANDELSKOERS IS NU";VL;"%"
2120 GOSUB 3200
2130 WS=WS-WW-WX:LN=LN-WW-WX
2140 MS=MS-MW-MX:LM=LM-MW-MX
2150 LA=LA-AW:MH=MH-SW
2160 FS=INT(FS*FX/100)
2170 PO=INT(PO*PX/100)
2180 LS=LS-SX:LE=LE-EX:RETURN
2190 CLS:X=0:Y=6
2200 PRINT"PIRATEN VALLEN UW SCHIP AAN":GOSUB
    2920
2210 RL=INT(50*RND(1)+50)
2220 RP=INT((RL-WL)/10)
2230 IF RP<0 THEN RP=0
2240 IF RP>ML THEN RP=ML
2250 ML=ML-RP:MS=MS-RP:LM=LM-RP
2260 PRINT:PRINT RP;"KUSTVAARDERS VERLOREN"
2270 RO=INT(50*RND(1)+50)
2280 RQ=INT((RO-WO)/10)
2290 IF RQ<0 THEN RQ=0
2300 IF RQ>MO THEN RQ=MO:MO=MO-RQ:MS=MS-RQ:
    LM=LM-RQ
2310 PRINT:PRINT RQ;"OVERZEESE HANDELSCHEPEN
    VERLOREN"
2320 GOSUB 3140:RETURN
2330 CLS:X=2:Y=2:GOSUB 3180
2340 COLOR 1,10:PRINT"FINANCIELE WINST":PRINT
2350 PRINT INT(ML*VL);"=LOKALE HANDELSWINST"
2360 PRINT INT(MO*VL*20);"=OVERZEESE
    HANDELSWINST"
2370 TR=TR+INT((ML*VL)+(MO*VL*20))
2380 PRINT TR;"= GOUD IN SCHATKAMERS"
2390 GOSUB 3150:X=2:Y=9:GOSUB 3180
2400 PRINT"UITGAVEN":PRINT
2410 PRINT MH*2500;"VOOR";MH;"HUURLINGENTROEPEN"
2420 PRINT LA*500;"VOOR";LA;"LEGIONAIRS"
2430 PRINT LN*500;"VOOR";LN;"MARINEBEMANNINGEN":
    PRINT:PRINT
2440 TR=TR-(LA*500)-(LN*500)-(MH*2500)
2450 PRINT" UITEINDELIJKE BALANS":PRINT
2460 PRINT TR;"GOUDSTUKKEN"
2470 GOSUB 3200:RETURN
2480 FV=INT(40*RND(1))+85
2490 CLS:X=0:Y=2:GOSUB 3180

```



```

2500 PRINT "VOEDSEL VOOR HET VOLK"
2510 PRINT "UW";ML;"KUSTVAARDERS HEBBEN"
2520 PRINT INT(ML*FV/100);"VOEDSELEENHEDEN
GEBRACHT"
2530 PRINT "PLUS UW VOORRAAD VAN"; FS
2540 PRINT "EENHEDEN, GEEFT U";FS+INT(ML*FV/100
);"EENHEDEN"
2550 FS=FS+INT(ML*FV/100)
2560 FS=FS-FR:IF FS<0 THEN GOTO 2580
2570 PRINT:PRINT"ER IS RUIM VOLDOENDE VOEDSEL":
GOSUB 3200:RETURN
2580 PRINT:PRINT"U LAAT HET VOLK VERHONGEREN"
2590 PRINT:PRINT"HET HEEFT";ABS(FS);"VOEDSEL-
EENHEDEN NODIG"
2600 GOSUB 3130:TR=TR+(2000*FS)
2610 PRINT"BETAAL";-2000*FS;"GOUDSTUKKEN VOOR
IMPORT":PRINT"VAN GRAAN TEGEN WOEKERPRIJS"
2620 VF=VF-(2*FS):FS=0
2630 GOSUB 3200:RETURN
2640 CLS:X=3:Y=2:GOSUB 3180:PRINT"OPINIEPELINGEN
":X=0:Y=5:GOSUB 3180
2650 IF VF<0 THEN GOTO 2680
2660 PRINT"UW REGERING IS NOG STEEDS POPULAIR"
2670 GOSUB 3200:RETURN
2680 PRINT"DE RIJKSONDERZOEKSDIENST RAPPORTEERT:"
2690 PRINT"DAT UW REGERING NIET POPULAIR IS"
2700 PRINT"U HEEFT";ABS(VF);"PUNTEN NODIG"
2710 PRINT"OM VOLDOENDE POPULARITEIT TE VERWERVEN
":IF TR<1000 THEN RETURN
2720 PRINT:PRINT"HOEVEEL WILT U BESTEDEN OM ROME"
2730 PRINT"EN HAAR BEVOLKING TE SUSSEN";
2740 INPUT A
2750 IF A=0 THEN RETURN
2760 A=INT(A/1000)
2770 VF=VF+A
2780 RETURN
2790 PO=INT(PO*1.07):FR=INT(PO/1000):WW=0:MW=0:
WX=0:MX=0:AW=0:SW=0:FX=0
2800 MA=MA+INT(6*RND(1)+1):PX=0:SX=0:EX=0
2810 TN=TN+1:UL=INT(PO/2500)
2820 SC=INT(((LA+LN+LM+LE+LS)*100)+(TR/2)+(PO/50)/
(100-TN))
2830 CLS:X=4:Y=10:GOSUB 3180:PRINT"UW SCORE IS =";
SC:GOSUB 3200

```



```

2840 IF TN>-36 OR SC<0 THEN GOTO 2860
2850 X=0:Y=10:GOSUB 3180:GOTO 40
2860 FOR Z=1 TO 1500:NEXT Z:CLS
2870 X=3:Y=10:GOSUB 3180
2880 PRINT"HET SPEL IS UIT !!!"
2890 X=3:Y=14:GOSUB 3180
2900 PRINT"UW SCORE IS";SC;"PUNTEN"
2910 END
2920 PLAY M1$
2930 RETURN
2940 PLAY M2$
2950 RETURN
2960 PLAY M3$
2970 RETURN
2980 CLS:PRINT"JAARVERSLAG SENAAT";ABS(TN);"
      V.C."
2990 PRINT:PRINT"BEVOLKING          ";PO
3000 PRINT"LEGEREENHEDEN           ";LA
3010 PRINT"HUURLINGENTROEPEN      ";MH
3020 PRINT"OORLOGSSCHEPEN         ";WS
3030 PRINT"MARINE-EENHEDEN        ";LN
3040 PRINT"HANDELSSCHEPEN         ";MS
3050 PRINT"SCHEEPSBEMANNINGEN     ";LM
3060 PRINT"SCHEEPSBOUWERS         ";LS
3070 PRINT"HANDWERKSLIEDEN        ";LE
3080 PRINT"ARBEIDSRESERVE         ";UL
3090 PRINT"GOUDVOORRAAD          ";TR
3100 PRINT"VOEDSELVOORRAAD       ";FS
3110 PRINT"VOEDSELBEHOEFTE       ";FR
3120 GOSUB 3200:RETURN
3130 FOR Z=1 TO 1000:NEXT Z
3140 FOR Z=1 TO 500:NEXT Z
3150 FOR Z=1 TO 475:NEXT Z
3160 FOR Z=1 TO 25:NEXT Z
3170 RETURN
3180 LOCATE X,Y
3190 RETURN
3200 X=5:Y=21:GOSUB 3180:PRINT"GA VERDER NA
      TOETSDRUK":GOSUB 2920
3210 A$=INKEY$
3220 IF A$="" THEN 3210
3230 RETURN
3240 COLOR 15,6

```

```
3250 XX$="          HET ROMEINSE RIJK"  
3260 GX=35  
3270 FOR XG=1 TO 35  
3280 G2$=MID$(XX$,GX,XG)  
3290 LOCATE 2,10:PRINT G2$;  
3300 GOSUB 3370  
3310 GX=GX-1  
3320 NEXT  
3330 REM GELUID UIT  
3340 FOR Z=1 TO 1000:NEXT  
3350 CLS:COLOR 1,3  
3360 RETURN  
3370 REM GELUIDSROUTINE  
3380 FOR Z=1 TO 50:NEXT Z  
3390 RETURN
```

DE SLAG OM ENGELAND

Dit programma laat de eerste twee weken van september 1940 herleven. U voert het bevel over de 11e Eenheid Royal Air Force en uw opdracht is om Spitfires en Hurricanes in te zetten om te voorkomen dat de Luftwaffe de door haar gekozen doelen bombardeert.

Aan het begin van elke dag krijgt u een inlichtingenrapport dat een schatting geeft van het totale aantal vliegtuigen dat de vijand ter beschikking staat en van de staat van paraatheid van uw vliegvelden. Deze factoren veranderen voortdurend in de loop van het spel. Vliegtuigen gaan verloren in gevechten, de vervangingsnelheid van de vliegtuigen verandert als er vliegtuigfabrieken worden gebombardeerd en de inzetbaarheid wordt aangetast als uw vliegvelden en radarposten worden aangevallen.

Nadat u het inlichtingenrapport hebt gelezen, selecteert de Luftwaffe haar doelen voor de acties van die dag. Het scherm wordt schoongemaakt en u krijgt de gelegenheid om vliegtuigen in te zetten om voor een zo goed mogelijke verdediging te zorgen.



Het is belangrijk zoveel mogelijk vliegtuigen in actie te krijgen. Het aantal gevechtsvliegtuigen dat erin slaagt de binnendringende bommenwerpers te onderscheppen, is minder naarmate de vliegtuigen slechter zijn ingezet en het weer slechter is. U krijgt elke ronde een weerrapport waarin de bewolkingdichtheid staat aangegeven. Op dagen met zware bewolking is het goed inzetten van de vliegtuigen extra belangrijk.

Als de Luftwaffe succesvolle aanvallen uitvoert op de radarposten of op de vliegvelden, wordt het niveau van inzetbaarheid drastisch verminderd. Dat is van grote invloed op het aantal gevechtsvliegtuigen dat beschikbaar is en het aantal vliegtuigen dat een goede kans heeft de binnendringende bommenwerpers te onderscheppen. Acties tegen vliegtuigfabrieken verminderen het aantal toestellen, dat voor vervanging wordt gebouwd.

Hoewel de kans klein is, kunnen een zorgvuldige planning en een dosis geluk u de overwinning bezorgen.

Programmabeschrijving:

het spel wordt bestuurd vanuit de hoofdlus in regel 40–110. De opbouw in programmamodulen zorgt ervoor dat het programma beter uitgevoerd wordt en het ook gemakkelijker is om fouten op te sporen, mochten die zich voordoen.

In regel 20 wordt de initialiseringsroutine aangeroepen. Er worden stringvariabelen opgezet die de namen kunnen bevatten van alle toestellen en van verschillende andere woorden die in de loop van het spel worden gebruikt. Deze stringvariabelen (een string is een rij karakters) worden vervolgens gebruikt in PRINT-instructies. Dit vermijdt een heleboel herhalingen bij het intypen en maakt de instructies gemakkelijker te gebruiken. De regels 2150–2180 geven de beginsterkte van alle toestellen en de beginwaarden van alle variabelen die worden gebruikt om de inzetbaarheid van de Engelse gevechtsvliegtuigen (BS) en de Duitse toestellen (GS) te berekenen. De variabele DD wordt gebruikt om het aantal dagen vast te stellen dat in de loop van het spel is gespeeld. Regel 2320 wordt gebruikt om aan de hand van X en Y de positie van de PRINT-instructies op het scherm te bepalen.

In regel 2220–2260 zien we een aantal FOR–NEXT lussen die worden gebruikt om vertragingen in het programma aan te brengen. Er kunnen lussen van verschillende lengten worden gemaakt door GOSUB-instructie te gebruiken en verschillende regelnummers aan te roepen. GOSUB 2230 geeft bijvoorbeeld een vertraging van 1000 en GOSUB 2240 een vertraging van 500. In regel 30 wordt de titelroutine aangeroepen en als die voorbij is, komen we bij de lus van het hoofdprogramma.

Regel 40 roept de routine aan die het inlichtingenrapport op het scherm brengt. In regel 150–220 wordt een schatting gemaakt van de inzetbaarheid van de Luftwaffe-toestellen; die wordt vervolgens afgebeeld. Dit getal is ruwweg gebaseerd op het werkelijke aantal bommenwerpers en gevechtsvliegtuigen, dat toentertijd in Frankrijk beschikbaar was. De regels 230–260 doen hetzelfde voor de Engelse toestellen. Ook hier wordt uitgegaan van het werkelijke aantal Spitfires en Hurricanes dat destijds de 11e Eenheid ter beschikking stond.

De routine vanaf regel 300 berekent elke ronde de doelwitten van de Luftwaffe. We verklappen hier niet hoe dit wordt gedaan, maar het resultaat komt aardig overeen met de historische feiten. In regel 380 wordt de weersfactor berekend, die bepaalt hoeveel bewolking er zal zijn. Zware bewolking betekent een aanzienlijke vermindering van het aantal gevechtsvliegtuigen dat in actie zal komen.

De regels 400–790 stellen u in staat het gebied te kiezen dat u gaat verdedigen. Deze routine vergelijkt het aanvalsplan van de Luftwaffe met uw keuze en berekent welk percentage van uw toestellen de bommenwerpers zal onderscheppen.

De gevechtsroutines staan in de regels 800–1290. Elk type toestel wordt een kanspercentage toebedacht om een vijandelijk toestel neer te schieten. De weersfactor, de inzetbaarheidsgraad en de onderscheppingsgraad (IR) worden dan gebruikt om te berekenen hoeveel toestellen van elk type meedoen aan de gevechten. De regels 2900–3100 gebruiken vervolgens het kanspercentage om de overwinningen te berekenen, terwijl elk type toestel in de slag gaat. Het resultaat wordt per toestel op het scherm gebracht.

De routine die de resultaten van de bombardementen van de Luftwaffe nagaat, wordt aangeropen in regel 90. Vervolgens krijgt de Luftwaffe op grond hiervan overwinningspunten. Regel 110 roept de regels aan die zorgen voor vergeldingsaanvallen van het RAF-bommenwerperscommando en geeft overwinningspunten aan de RAF.

Regel 1950 gaat na of het spel uit is. Zo ja, dan wordt het programma naar de eindroutine gestuurd. Mocht u het spel in eerste instantie te moeilijk vinden, dan kunt u uw kansen vergroten door 10 toe te voegen aan de waarden in regel 1980–2020.

Bent u nog geen 15 dagen onderweg, dan gaan we terug naar het begin van de hoofdlus om de volgende dag in september 1940 mee te maken.

```
10 REM SLAG OM ENGELAND
20 GOSUB 2090:REM INITIALISATIE
30 GOSUB 2360:REM START
40 GOSUB 120
50 GOSUB 290
60 GOSUB 400
70 GOSUB 800
80 GOSUB 1300
90 GOSUB 1480
100 GOSUB 1790
```

```

110 GOTO 1840
120 CLS
130 PRINT TAB(7);"R.A.F. INLICHTINGDIENST"
140 PRINT TAB(14);"SEPTEMBER";DD:PRINT
150 PRINT"SCHATTING VAN BRUIKBARE VLIEGTUIGEN":
PRINT
160 P=RND(-TIME)
170 PRINT B9$;"=";INT(B9*GS*(40*RND(1)+80)/100)
180 PRINT B0$;"=";INT(B0*GS*(40*RND(1)+80)/100)
190 PRINT HE$;"=";INT(H1*GS*(40*RND(1)+80)/100)
200 PRINT DO$;"=";INT(D7*GS*(40*RND(1)+80)/100)
210 PRINT JU$;"=";INT(J8*GS*(40*RND(1)+80)/100)
220 PRINT ST$;"=";INT(J7*GS*(40*RND(1)+80)/100):
PRINT
230 PRINT TAB(3);"11e LUCHTMACHTGROEP BESCHIKBAAR":
PRINT
240 PRINT SP$;"=";INT(SP*BS/100)
250 PRINT HU$;"=";INT(HU*BS/100)
260 PRINT BF$;"=";BF
270 PRINT:PRINT"BRUIKBAAARHEID R.A.F.-BASES ="
BS;"%"
280 GOSUB 2280:RETURN
290 CLS
300 LOCATE 4,4:PRINT "DE "LU$;" BERAADT ZICH OP
AANVALLEN"
310 GC= INT(2000*RND(1)+1)+(LL*2)+CF
320 IF GC<300 THEN GT=1:GC$="RADARSTATIONS":
GOTO 380
330 IF GC<1200 THEN GT=3:GC$="LUCHTMACHTBASES":
GOTO 380
340 IF GC<1500 THEN GT=4:GC$="VLIEGTUIGFABRIEKEN":
GOTO 380
350 IF GC<1800 THEN GT=2:GC$="ENGELSE KANAALHAVENS
":GOTO 380
360 IF GC<2001 THEN GT=5:GC$="KLEINE INDUSTRIE-
STEDEN":GOTO 380
370 GT=6:GC$="LONDEN"
380 WF=150-INT(100*RND(1)+1):IF WF>100 THEN WF=
100
390 GOSUB 2230:RETURN
400 CLS:PRINT TAB(7)"11e GROEP'S INZETKEUZE":
PRINT:GOSUB 2240
410 PRINT "WEERBERICHT:-";100-WF;"% BEWOLKT
420 PRINT LU$;" VERLIEZEN =" ;LL:PRINT

```

```

430 PRINT "DE ";LU$;" HEEFT AANVALLEN AFGELAST
    OP:";
440 IF (LL*2)+CF<301 THEN PRINT "NIETS"
450 IF (LL*2)+CF>300 THEN PRINT "RADARSTATIONS"
460 IF (LL*2)+CF>1199 THEN PRINT "11E GROEP'S
    LUCHTMACHTBASES"
470 IF (LL*2)+CF>1799 THEN PRINT "ENGELSE
    KANAALHAVENS"
480 IF (LL*2)+CF>2000 THEN PRINT "KLEINE
    INDUSTRIESTEDEN"
490 X=0:Y=11:GOSUB 2320
500 PRINT"WAAR WILT U DE 11e GROEP'S VERDEDIG-":
    PRINT"ING CONCENTREREN?":PRINT
510 PRINT"1=RADARSTATIONS":PRINT"2=KANAALHAVENS"
520 PRINT"3=LUCHTMACHTBASES":PRINT"4=
    VLIEGTUIGFABRIEKEN"
530 PRINT"5=KLEINE STEDEN":PRINT"6=LONDEN"
540 INPUT D
550 IF D<1 OR D>6 THEN GOTO 540
560 GOSUB 2240
570 CLS:X=1:Y=3:GOSUB 2320
580 PRINT"U BESLUIT TE VERDEDIGEN...:"
590 IF D=1 THEN PRINT"DE RADARSTATIONS"
600 IF D=2 THEN PRINT"DE KANAALHAVENS"
610 IF D=3 THEN PRINT"DE LUCHTMACHTBASES"
620 IF D=4 THEN PRINT"DE VLIEGTUIGFABRIEKEN"
630 IF D=5 THEN PRINT"DE KLEINE STEDEN"
640 IF D=6 THEN PRINT"LONDEN"
650 GOSUB 2250
660 PRINT:PRINT"DE ";LU$;" CONCENTREERT ZIJN"
670 PRINT "AANVAL OP ";GC$:PRINT
680 IF GT=D THEN IR=20
690 IF GT-D=1 OR GT-D=-1 THEN IR=40
700 IF GT-D=2 OR GT-D=-2 THEN IR=60
710 IF GT-D=3 OR GT-D=-3 THEN IR=80
720 IF GT-D>3 OR GT-D<-3 THEN IR=100
730 PRINT IR"% VAN UW VLIEGTUIGEN ONDERSCHEPT":
    PRINT:GOSUB 2240
740 IF IR=100 THEN PRINT "11e GROEP IS
    UITSTEKEND UITGERUST!!"
750 IF IR=80 THEN PRINT "11e GROEP IS GOED
    UITGERUST!!"
760 IF IR=60 THEN PRINT "11e GROEP VERWACHT
    OVERMACHT"

```

```

770 IF IR=40 THEN PRINT "11e GROEP IS SLECHT
    UITGERUST"
780 IF IR=20 THEN PRINT "11e GROEP IS TOTAAL
    INGEMAAKT"
790 GOSUB 2220:RETURN
800 ZY=INT(SP*BS*WF/1000000#*IR)+1
810 KF=12:AZ$=SP$:BT=1
820 GOSUB 1080
830 ZY=INT(HU*BS*WF/1000000#*IR)+1
840 KF=7:AZ$=HU$:BT=1
850 GOSUB 1080
860 ZY=INT(BF*GT*WF/600)+1
870 KF=3:AZ$=BF$:BT=1
880 GOSUB 1080
890 ZY=INT(B9*GS*WF/100*.65)+1
900 KF=14:AZ$=B9$:BT=0
910 GOSUB 1080
920 ZY=INT(B0*GS*WF/100*.75)+1
930 KF=5:AZ$=B0$:BT=0
940 GOSUB 1080
950 ZY=INT(H1*GS*WF/100*.4)+1
960 KF=2:AZ$=HE$:BT=0
970 GOSUB 1080
980 ZY=INT(D7*GS*WF/100*.4)+1
990 AZ$=DO$:BT=0
1000 GOSUB 1080
1010 ZY=INT(J7*GS*WF/100*.3)+1
1020 AZ$=ST$:BT=0
1030 GOSUB 1080
1040 ZY=INT(J8*GS*WF/100*.4)+1
1050 KF=3:AZ$=JU$:BT=0
1060 GOSUB 1080
1070 RETURN
1080 CLS
1090 PRINT ZY;AZ$;" : AANVAL":GOSUB 2240
1100 FOR ZZ=1 TO ZY
1110 IF INT(100*RND(1))+1=<KF THEN KI=1
1120 IF KI=0 THEN PRINT AZ$;ZZ;"....."
1130 IF KI=1 AND BT=1 THEN GOSUB 1180:PRINT AZ$;
    ZZ;BZ$:GOSUB 2330:PH=PH+1
1140 IF KI=1 AND BT=0 THEN GOSUB 1260:PRINT AZ$;
    ZZ;BZ$:GOSUB 2330:PH=PH+1
1150 KI=0:GOSUB 2260:NEXT ZZ
1160 CLS:PRINT AZ$;PH;"VLIEGTUIGEN UITGESCHAKELD"

```



```

1170 PH=0:GOSUB 2220:RETURN
1180 SD=INT(100*RND(1))+1
1190 IF SD<20 THEN D7=D7-1:BZ$=DO$:L1=L1+1:
      GOTO 1250
1200 IF SD<44 THEN H1=H1-1:BZ$=HE$:L2=L2+1:
      GOTO 1250
1210 IF SD<61 THEN J8=J8-1:BZ$=JU$:L3=L3+1:
      GOTO 1250
1220 IF SD<71 THEN J7=J7-1:BZ$=ST$:L4=L4+1:
      GOTO 1250
1230 IF SD<84 THEN B9=B9-1:BZ$=B9$:L5=L5+1:
      GOTO 1250
1240 B0=B0-1:BZ$=B0$:L6=L6+1
1250 LL=LL+1:L0=L0+1:RETURN
1260 SD=INT(100*RND(1))+1
1270 IF SD<76 THEN HU=HU-1:BZ$=HU$:L7=L7+1:
      GOTO 1290
1280 SP=SP-1:BZ$=SP$:L8=L8+1
1290 L9=L9+1:RETURN
1300 CLS
1310 PRINT"VERLIEZEN SEPTEMBER";DD
1320 PRINT"R.A.F."
1330 PRINT SP$;"=";L8
1340 PRINT HU$;"=";L7
1350 PRINT"TOTAAL"=";L9
1360 PRINT:PRINT LU$:PRINT
1370 PRINT HE$;"=";L1
1380 PRINT DO$;"=";L2
1390 PRINT JU$;"=";L3
1400 PRINT ST$;"=";L4
1410 PRINT B9$;"=";L5
1420 PRINT B0$;"=";L6
1430 PRINT "TOTAAL"=";L0
1440 GOSUB 2280
1450 L0=0:L1=0:L2=0:L3=0:L4=0
1460 L5=0:L6=0:L7=0:L8=0:L9=0
1470 RETURN
1480 CLS
1490 PRINT "DE LUFTWAFFE-AANVAL":PRINT
1500 PRINT GC$
1510 GOSUB 2240:X=1:Y=6:GOSUB 2320
1520 IF GT=1 THEN GOTO 1570
1530 IF GT=2 THEN GOTO 1620

```

```

1540 IF GT=3 THEN GOTO 1640
1550 IF GT=4 THEN GOTO 1700
1560 IF GT>4 THEN GOTO 1740
1570 ZZ=INT((H1+D7+J8+(J7*2))*WF*WF*GS/500000!)
1580 BS=BS-ZZ
1590 PRINT "BRUIKBAARHEID R.A.F.-BASES IS"
1600 PRINT "VERMINDERD TOT";BS;"%"
1610 VP=VP+ZZ:GOTO 1760
1620 ZZ=INT((H1+D7+J7+J8)*WF*WF*GS/1000000#)
1630 VP=VP+ZZ:GOTO 1760
1640 KF=101-INT(BS*1.4):IF KF<1 THEN KF=1:AZ$=
"DUITSE BOMMENWERPERS":ZY=ZZ:BT=0
1650 GOSUB 2240:GOSUB 1080
1660 ZZ=INT(ZZ/10):BS=BS-ZZ
1670 PRINT "BRUIKBAARHEID R.A.F.-BASES IS"
1680 PRINT "VERMINDERD TOT";BS;"%"
1690 VP=VP+ZZ:GOTO 1760
1700 ZZ=INT((H1+D7+J7+J8)*WF*WF*GS/1000000#)
1710 P=RND(-TIME)
1720 IF INT(100*RND(1)+1)<WF THEN SR=SR-1:HR=HR-3:
X=1
1730 IF X=1 THEN PRINT "PRODUKTIE VERWOEST!!":X=0:
VP=VP+ZZ:GOTO 1760
1740 ZZ=INT((H1+D7+J7)*WF*WF*GS/1000000#)
1750 VP=VP+ZZ
1760 X=1:Y=12:GOSUB 2320
1770 PRINT ZZ;"WINSTPUNTEN ZIJN TOEGEKEND AAN DE ";
LU$
1780 GOSUB 2220:RETURN
1790 H1=H1+GR:D7=D7+GR:J8=J8+GR
1800 J7=J7+GR:B9=B9+GR:B0=B0+GR
1810 SP=SP+SR:HU=HU+HR:DD=DD+1
1820 IF GT<>1 AND GT<>3 THEN BS=BS+2
1830 RETURN
1840 CLS
1850 IF RND(-TIME)>.8 THEN GOTO 1900
1860 PRINT "BOMBARDEMENTEN OP FRANSE HAVENS OM"
1870 PRINT "INVASIESCHEPEN TOT ZINKEN TE BRENGEN"
1880 GOSUB 2240:PRINT "1 WINSTPUNT IS TOEGEKEND
AAN DE R.A.F."
1890 VP=VP-1:GOTO 1950
1900 PRINT "DE EERSTE MINISTER BEVEELT VERGELD-"
1910 PRINT "INGSAANVAL OP BERLIJN!"
1920 PRINT

```

```

1930 PRINT "DE FUHRER IS FURIEUS"
1940 CF=CF+100
1950 GOSUB 2220:IF DD<15 THEN GOTO 40
1960 CLS
1970 PRINT"HET SPEL IS UIT"
1980 IF VP<46 THEN A$="R.A.F.":B$="TACTISCHE"
1990 IF VP<56 THEN A$="R.A.F.":B$="TOTALE"
2000 IF VP>55 AND VP<65 THEN GOTO 2050
2010 IF VP>65 THEN A$=LU$:B$="TACTISCHE"
2020 IF VP>74 THEN A$=LU$:B$="TOTALE"
2030 X=1:Y=10:GOSUB 2320
2040 PRINT "DE ";A$;" BEHAALT EEN ";B$;"
      OVERWINNING":GOTO 2070
2050 PRINT "DE SLAG IS ONBESLIST"
2060 PRINT "GEEN VAN BEIDE BEHAALDE VOORDEEL"
2070 PRINT "DE ";LU$;" SCOORT";VP;"PUNTEN"
2080 END
2090 REM INITIALISATIE
2100 HE$="HE 111":DO$="DO 17"
2110 JU$="JU 88":ST$="ST 87"
2120 B9$="BF 109":B0$="BF 110"
2130 SP$="SPITFIRE":HU$="HURRICANE"
2140 BF$="LUCHTDOELGESCHUT":LU$="LUFTWAFPE"
2150 HU=560:SP=190:BF=100:HR=12
2160 D7=300:H1=400:J8=270:SR=3
2170 J7=150:B9=330:B0=230:GR=2
2180 GS=.55:BS=75:DD=1
2190 SCREEN 0
2200 KEY OFF
2210 RETURN
2220 FOR TD=1 TO 1000:NEXT TD
2230 FOR TD=1 TO 500:NEXT TD
2240 FOR TD=1 TO 400:NEXT TD
2250 FOR TD=1 TO 90:NEXT TD
2260 FOR TD=1 TO 10:NEXT TD
2270 RETURN
2280 LOCATE 5,22:PRINT"GA VERDER NA TOETSDRUK"
2290 A$=INKEY$
2300 IF A$="" THEN GOTO 2290
2310 RETURN
2320 LOCATE X,Y:RETURN
2330 REM GELUIDSROUTINE
2340 PLAY "O6C"
2350 RETURN

```

```
2360 REM TITEL
2370 CLS
2380 COLOR 15,1
2390 X=10:Y=12:GOSUB 2320
2400 PRINT"SLAG OM ENGELAND"
2410 FOR Y=12 TO 1 STEP -2:GOSUB 2320:PRINT"SLAG
      OM ENGELAND"
2420 FOR TD=1 TO 100:NEXT TD:NEXT Y
2430 FOR Y=12 TO 22 STEP 2:GOSUB 2320:PRINT"SLAG
      OM ENGELAND"
2440 FOR TD=1 TO 100:NEXT TD:NEXT Y
2450 P=RND(-TIME)
2460 FOR Z=1 TO 250
2470 X=INT(RND(1)*40)
2480 Y=INT(RND(1)*22)
2490 LOCATE X,Y:PRINT"*"
2500 NEXT Z
2510 FOR TD=1 TO 1000:NEXT TD
2520 RETURN
```

EIGENHEIMERHEERD

In dit simulatieprogramma worden alle problemen die zich bij de exploitatie van een boerderij kunnen voordoen, nagespeeld: het zware werk, het verdienen van genoeg geld om de vaste kosten te kunnen betalen, de strijd met het weer en de vakbonden en het stukgaan van machines.

Elk jaar moet u beslissen of u geld leent van de bank, of u kapitaalgoederen vervangt, of u land en schapen aan- of verkoopt, of u de gewassen bevoeit, of u op vakbondseisen ingaat en welke soort gewas u aanplant. Daarna moet u het weer afwachten, dat de kwaliteit en de omvang van uw oogst bepaalt.

In het begin van het spel krijgt u, de hardwerkende boer, een leeftijd toebedeeld, die ligt tussen 27 en 35 jaar. Het spel gaat door, jaar na jaar, tot u 50 bent en met pensioen gaat.

Na de titelroutine verschijnt een beeldpagina met informatie over uw situatie. Hierin staat informatie over alles wat nodig is om met succes een boerderij draaiende te houden. Deze informatie moet u aandachtig doorlezen. U begint het spel met \$3000 op de bank, een tractor, een vrachtauto, een schuur en een landarbeider om u te helpen.



Eigenheimerheerd bestaat uit 350 bunder land en 100 schapen. U krijgt te horen hoeveel de bank u wil lenen en hoeveel u eventueel jaarlijks moet terugbetalen. Vervolgens heeft u de mogelijkheid geld te lenen van de bank. Heeft u al een lening en wilt die u eerder afbetalen, dan kunt u een negatieve lening nemen en zo het geld vervroegd aflossen.

Nadat u de bankzaken hebt afgehandeld, krijgt u informatie over de kapitaalgoederen die mogelijk stukgegaan zijn en moeten worden vervangen. Daarna krijgt u de mogelijkheid goederen, land of schapen te kopen voor het komende jaar. Als u hier een negatief getal invoert, kunt u, als u wilt, land of schapen verkopen. Verkoop van schapen is een goed middel om aan kontant geld te komen. Probeer de schapen te kopen als de prijs laag is en ze te verkopen bij een hoge prijs! U zult ook merken dat schapen zich vrij snel vermeerderen, dus moet u ervoor zorgen dat u voldoende land heeft om het aantal schapen te kunnen onderhouden.

De volgende stap in het spel is de beslissing of u de gewassen wilt bevoeien. Door bevoeiing heeft u meer kans op een goede oogst. Vervolgens moet u kiezen welk gewas u gaat planten. De soorten gewassen worden aangegeven in volgorde van geldelijke opbrengst.

Dan de kwestie van het weer, bedorven gewas door regen of droogte, of juist een extra hoge opbrengst door een prima weertje. Evenals in de werkelijkheid is er geen peil op te trekken; het enige dat u kunt doen is de weergoden aanroepen voor de juiste regenval.

De vakbond van landarbeiders is het volgende probleem dat u tegenkomt. De vakbond probeert altijd een verbetering voor de leden te bewerkstelligen. De benodigde informatie verschijnt op het scherm en u moet besluiten wat u gaat doen. Het is niet verstandig om te snel in te gaan op de eisen van de vakbond. Zoals u weet worden de lonen dan te hoog. Wacht u te lang, dan kunnen er stakingen uitbreken, die leiden tot grote verliezen in de opbrengst. Conclusie: afwegen is het devies.

Vervolgens steekt de computer zich in het jasje van de boekhouder en berekent hoeveel geld u dit jaar heeft verdiend. Van de verdiensten worden de lonen van de arbeider(s) afgetrokken, evenals de afbetaling op de lening en de algemene onderhoudskosten. Is dat alles gebeurd, dan bent u een jaar verder en krijgt u weer met weersgesteldheden, vakbonden en opbrengsten te maken.

Bereikt u de pensioengerechtigde leeftijd (50 jaar in het boerenvak), of gaat u in de loop van de jaren failliet, dan is het spel afgelopen.

Programmabeschrijving:

de regels 20–130 controleren de uitvoering van het programma. Hierin worden de verschillende benodigde routines aangeroept. Regel 20 stuurt het programma naar de opstartroutine waarin alle beginwaarden voor het spel worden vastgelegd. De variabele CS in regel 3510 is de beginprijs voor een schaap. Regel 3530 dient ervoor om aan de hand van de variabelen X en Y een positie op het scherm te bepalen.

De instructie GOSUB 3800 in regel 30 roept de titelroutine aan, die met behulp van FOR-NEXT lussen de titel op het scherm PRINT. De volgende GOSUB in regel 40 PRINT de gegevens over uw situatie op het scherm.

De routine die de bankzaken behandelt, loopt van regel 360-420 en wordt aangeroepen door GOSUB 360 in regel 50. De regels 480-520 berekenen de rente en de afbetalingen. Het geleende geld wordt dan opgeteld bij het geld dat u op de bank heeft (variabele MB).

De volgende routine gaat na of u eventueel kapitaalgoederen moet vervangen. Goederen gaan stuk in de regels 2760-2870. Deze goederen worden dan vervangen in regel 530-700. De kosten van vervanging worden automatisch afgetrokken van uw bankrekening.

De routine van regel 740-1170 zorgt voor de aan- en verkoopfuncties in het spel. Deze routine zit logisch in elkaar, hetgeen u zelf kunt constateren bij zorgvuldige bestudering.

De volgende routine stelt u in staat het land te bevoeien. Hierna wordt aangegeven hoeveel land u nodig heeft om al uw schapen te kunnen onderhouden en hoeveel land er beschikbaar is voor gewassen. Daarna kiest u het gewas dat u wilt aanplanten. Deze routine maakt gebruik van de uitputtingsgraad (EL), die de oogstopbrengst beïnvloedt en ervoor zorgt dat u het land een jaar braak moet laten liggen, zodat het zich kan herstellen.

De regenval wordt berekend en op het scherm gebracht door de routine in regel 1530-1870. Deze routine (die wordt aangeroepen in regel 100) heeft grote gevolgen voor de jaarlijkse opbrengst. Hoge percentages verrotte of verdroogde gewassen kunnen het resultaat zijn.

GOSUB 1880 in regel 110 is de routine waarin de vakbond van landarbeiders probeert u een loonsverhoging af te dwingen voor haar leden. U kunt akkoord gaan met de eisen of een staking riskeren, die de oogst beïnvloedt. Aan u de keuze.

Het bedrag dat u in een jaar heeft omgezet, wordt berekend in de routine die in regel 120 wordt aangeroepen. Uw netto inkomen wordt berekend in regel 2240. Bij deze berekening worden alle belangrijke factoren ingecalculeerd.

Na terugkeer uit deze routine (RETURN), gaat het programma naar regel 2400, waar het jaar wordt afgerond met verschillende gebeurtenissen. Ten eerste wordt de toename van de kudde schapen berekend. Vervolgens rekent het programma uit of u voldoende land heeft voor de schapen en past het waar nodig de verschillende variabelen aan. Hier worden ook de prijsschommelingen voor de schapen berekend.

In het volgende onderdeel (t/m regel 2910) wordt nagegaan of de boerderij in de loop van het jaar te lijden heeft gehad van een aantal mogelijke rampen.

Regel 2920 verhoogt uw leeftijd met een jaar (FA is de leeftijd van de boer) en gaat dan na of u de pensioengerechtigde leeftijd heeft bereikt. Is dat het geval, dan gaan we naar de eindroutine, waar wordt meegedeeld hoe goed u heeft geboerd. Het zal u opvallen dat u erg veel moet verdienen om de kwalificatie 'uitstekende boer' te verdienen.

Bent u nog steeds geen 50, dan gaat het programma na of u niet failliet bent gegaan en gaat dan terug naar het begin van het hoofdprogramma voor nog een jaar Eigenheimerheerd.

Uiteraard hebben we niet alle geheimen van het programma prijsgegeven; ga dus voorzichtig te werk. Veel plezier met het spel!

```
10 REM EIGENHEIMERHEERD
20 GOSUB 3420
30 GOSUB 3800
40 GOSUB 140
50 GOSUB 360
60 GOSUB 560
70 GOSUB 740
80 GOSUB 1180
90 GOSUB 1290
100 GOSUB 1530
110 GOSUB 1880
120 GOSUB 2180
130 GOTO 2400
140 X=1:Y=3:GOSUB 3530
150 PRINT"UW SITUATIE IS:"
160 PRINT" BANKBALANS $";MB
170 PRINT" UW LEEFTIJD IS";FA
180 PRINT
190 PRINT" U HEBT:"
200 PRINT TR;"TREKKER"
210 PRINT HD;"ZAAIMACHINE"
220 PRINT HV;"OOGSTMACHINE"
230 PRINT TK;"VRACHTWAGEN"
240 PRINT BN;"SCHUUR"
250 PRINT
260 PRINT AC;"BUNDER LAND"
270 PRINT SH;"SCHAPEN"
280 IF IB=0 THEN PRINT" GEEN IRRIGATIE"
290 IF IB=1 THEN PRINT" GEIRRIGEERD LAND"
300 PRINT EL;"% LANDUITPUTTING"
310 PRINT LR;"ARBEIDERS"
320 PRINT " $";PR;"= KOSTEN PER ARBEIDER"
330 PRINT " $";CL;"= PRIJS PER BUNDER LAND"
340 PRINT" $";BL;"= KREDIETLIET BANK"
350 PRINT" $";RP;"= TERUGBETALINGEN":RETURN
```



```

360 GOSUB 3740:GOSUB 3640
370 PRINT CHR$(11);"WILT U VAN DE BANK LENEN
      (J/N) ";
380 INPUT A$
390 GOSUB 3740
400 IF A$<>"J" AND A$<>"N" THEN GOTO 370
410 IF A$="N" THEN RETURN
420 GOSUB 3570
430 PRINT CHR$(11);"HOEVEEL WILT U LENEN ";
440 INPUT AB
450 GOSUB 3740
460 IF AB>BL THEN PRINT"ZOVEEL WIL DE BANK U NIET
      LENEN!":GOSUB 3720:GOSUB 3550
470 IF AB>BL THEN GOSUB 3740:GOTO 430
480 YL=50-FA
490 IF YL>20 THEN YL=20
500 TB=TB+AB:BL=BL-AB
510 RP=INT(TB*2.2/YL)
520 MB=MB+AB
530 GOSUB 3560:GOSUB 3740:GOSUB 140
540 PRINT CHR$(11);"UW TERUGBETALING = $";RP
550 GOSUB 3560:GOSUB 3740:RETURN
560 YY=YY+1:IF YY<2 THEN RETURN
570 IF TR=1 THEN CB=CB+1:GOTO 600
580 PRINT:PRINT"U MOET EEN TRACTOR KOPEN ($5000)":
      GOSUB 3600:GOSUB 3580
590 MB=MB-5000:TR=1
600 IF TK=1 THEN CB=CB+1:GOTO 630
610 PRINT:PRINT"U MOET EEN TRUCK KOPEN ($10000)":
      GOSUB 3600:GOSUB 3580
620 MB=MB-10000:TK=1
630 IF BN=0 THEN GOTO 650
640 IF AC/BN<500 THEN GOTO 680
650 PRINT:PRINT"U MOET EEN SCHUUR BOUWEN ($3000)":
      GOSUB 3600:GOSUB 3580
660 MB=MB-3000:BN=BN+1:CB=CB-1
670 GOTO 740
680 IF DB=1 THEN CB=CB+1:GOTO 710
690 PRINT:PRINT"U MOET EEN NIEUWE PUT SLAAN ($2000)":
      GOSUB 3600:GOSUB 3580
700 MB=MB-2000:CB=CB-1:DB=1
710 IF CB=3 THEN PRINT CHR$(11);"U HOEFT DIT JAAR GEEN
      KAPITAAL-"
720 IF CB=3 THEN PRINT"GOEDEREN TE VERVANGEN":

```

```

GOSUB 3550
730 CB=0:CLS:GOSUB 140:GOSUB 3760:RETURN
740 CLS:PRINT"U HEBT $";MB;"OP DE BANK":PRINT:
PRINT:PRINT
750 PRINT"WAT WILT U KOPEN?":PRINT
760 PRINT"1 = ZAAIMACHINE ($15000)":PRINT
770 PRINT"2 = OOGSTMACHINE ($15000)":PRINT
780 PRINT"3 = LAND ($";CL;"/BUNDER)":PRINT
790 PRINT"4 = SCHAPEN ($";CS;"/STUK)":PRINT
800 PRINT"5 = HANDEL AFSLUITEN":PRINT:PRINT
810 PRINT"NEGATIEF NUMMER IS VERKOPEN"
820 INPUT C$
830 IF C$<"1" OR C$>"5" THEN GOTO 740
840 IF C$="1" THEN GOSUB 3740:GOTO 890
850 IF C$="2" THEN GOSUB 3740:GOTO 950
860 IF C$="3" THEN GOSUB 3740:GOTO 1010
870 IF C$="4" THEN GOSUB 3740:GOTO 1090
880 IF C$="5" THEN GOSUB 3740:GOTO 1170
890 IF HD=1 THEN PRINT CHR$(11);"U HEBT AL EEN
ZAAIMACHINE":GOSUB 3720
900 IF MB<15000 THEN PRINT CHR$(11);"U HEBT
NIET GENOEG GELD"
910 IF HD=1 OR MB<15000 THEN GOSUB 3550:GOSUB
3740:GOTO 740
920 HD=1:MB=MB-15000
930 PRINT CHR$(11);"U HEBT EEN ZAAIMACHINE
GEKOCHT"
940 GOSUB 3550:GOSUB 3740:GOTO 740
950 IF HV=1 THEN PRINT CHR$(11);"U HEBT AL EEN
OOGSTMACHINE!":GOSUB 3720
960 IF MB<15000 THEN PRINT CHR$(11);"U HEBT
NIET GENOEG GELD"
970 IF HV=1 OR MB<15000 THEN GOSUB 3550:GOSUB
3740:GOTO 740
980 HV=1:MB=MB-15000
990 PRINT CHR$(11);"U HEBT EEN OOGSTMACHINE GEKOCHT"
1000 GOSUB 3550:GOSUB 3740:GOTO 740
1010 PRINT CHR$(11);"HOEVEEL BUNDER WILT U KOPEN";
1020 INPUT AS:IF AS<0 AND ABS(AS)>AC THEN GOTO 740
1030 IF AS*CL>MB THEN GOSUB 3740:PRINT CHR$(11);"
U KUNT MAAR";INT(MB/CL);"BUNDER KOPEN"
1040 IF AS*CL>MB THEN GOSUB 3720:GOSUB 3550:GOSUB
3740:GOTO 740
1050 AC=AC+AS:MB=MB-(AS*CL):GOSUB 3740

```

```

1060 IF AS<0 THEN PRINT CHR$(11);"U HEBT";ABS(AS);"
      BUNDER VERKOCHT":GOTO 1080
1070 PRINT CHR$(11);"U HEBT";AS;"BUNDER GEKOCHT"
1080 GOSUB 3550:GOSUB 3740:GOTO 740
1090 PRINT CHR$(11);"HOEVEEL SCHAPEN WILT U KOPEN?":
1100 INPUT SS:IF SS<0 AND ABS(SS)>SH THEN GOTO 740
1110 IF SS*CS>MB THEN GOSUB 3740:PRINT CHR$(11);"U
      KUNT MAAR";INT(MB/CS);"SCHAPEN KOPEN"
1120 IF SS*CS>MB THEN GOSUB 3720:GOSUB 3550:GOSUB
      3740:GOTO 740
1130 SH=SH+SS:MB=MB-(SS*CS):GOSUB 3740
1140 IF SS<0 THEN PRINT CHR$(11);"U HEBT";ABS(SS);
      "SCHAPEN VERKOCHT":GOTO 1160
1150 PRINT CHR$(11);"U HEBT";SS;"SCHAPEN GEKOCHT"
1160 GOSUB 3550:GOSUB 3740:GOTO 740
1170 CLS:RETURN
1180 X=7:Y=6:GOSUB 3530
1190 PRINT"U HEBT";MB;"OP DE BANK"
1200 GOSUB 3640:X=4:Y=10:GOSUB 3530
1210 PRINT"WILT U UW LAND BEVLOEIEN (J/N)"
1220 X=8:Y=12:GOSUB 3530
1230 PRINT"(KOST $2 PER BUNDER)";
1240 INPUT WI$
1250 IF WI$<>"J" AND WI$<>"N" THEN GOTO 1180
1260 IF WI$="N" THEN IB=0:IP=1
1270 IF WI$="J" THEN IB=1:IP=1.3:MB=MB-(AC*2)
1280 GOSUB 3570:CLS:GOSUB 140:RETURN
1290 GOSUB 3760:CLS
1300 SL=INT(SH/5):PL=AC-SL
1310 PRINT"U HEBT";AC;"BUNDER":PRINT
1320 PRINT"UW SCHAPEN HEBBEN";SL;"NODIG"
1330 PRINT"OM TE GRAZEN"
1340 PRINT
1350 PRINT"U HEBT";PL;"BUNDER OM UW GEWAS TE"
1360 PRINT"VERBOUWEN":PRINT
1370 PRINT"DE UITPUTTINGSGRAAD VAN UW LAND"
1380 PRINT"IS:-";EL;"%":PRINT
1390 PRINT"WAT WILT U:-":PRINT
1400 PRINT"1 = GERST"
1410 PRINT"2 = TARWE"
1420 PRINT"3 = ERWTEN"
1430 PRINT"4 = BRAAK LATEN LIGGEN":PRINT
1440 PRINT"WELK GEWAS WILT U VERBOUWEN";
1450 INPUT P$

```

```

1460 IF P$<>"1" AND P$>"4" THEN GOTO 1290
1470 IF EL>99 AND P$<>"3" AND P$<>"4" THEN PRINT:
PRINT"UW LAND IS UITGEPUT"
1480 IF EL>99 AND P$<>"3" AND P$<>"4" THEN GOSUB
3720:GOSUB 3570:GOTO 1290
1490 IF P$="1" THEN GOTO 3270
1500 IF P$="2" THEN GOTO 3310
1510 IF P$="3" THEN GOTO 3350
1520 IF P$="4" THEN GOTO 3390
1530 X=6:Y=5:GOSUB 3530
1540 PRINT"HOE ZIT HET MET DE REGENVAL?":GOSUB
3640
1550 Q=RND(-TIME)
1560 WW=0:FOR XX=1 TO 40
1570 IF RND(1)>.5 THEN WW=WW+1
1580 NEXT XX
1590 FOR VV=0 TO 39
1600 VPOKE 880+VV,219
1610 NEXT VV
1620 FOR VV=0 TO 5
1630 VPOKE 640+VV,219
1640 NEXT VV
1650 FOR VV=1 TO WW
1660 VPOKE 760+VV,249
1670 GOSUB 3580
1680 NEXT VV
1690 PRINT:PRINT:PRINT:PRINT"DIT JAAR REGENDE
HET:";INT(WW*2.54);"CENTIMETER":PRINT:PRINT
1700 GOSUB 3550:GOSUB 3760
1710 COLOR 11
1720 CLS
1730 PRINT"DIT JAAR REGENDE HET";INT(WW*2.54);
"CENTIMETER":PRINT:PRINT
1740 IF WW<10 THEN PRINT"GROTE DROOGTE !!!":PRINT:
PRINT:GOSUB 3720
1750 IF WW<10 THEN DI=INT(RND(1)*50)+1
1760 IF WW<10 THEN PRINT"VERLOREN GEWAS:";100-DI;
"%":GOSUB 3540:RETURN
1770 IF WW<19 THEN PRINT "DROOGTE !!":PRINT:PRINT:
GOSUB 3720
1780 IF WW<19 THEN DI=50+INT(RND(1)*50)
1790 IF WW<19 THEN PRINT"VERLOREN GEWAS:";100-DI;
"%":GOSUB 3540:RETURN
1800 IF WW<23 THEN PRINT"PRIMA WEERTJE !!!":PRINT:

```

```

PRINT:GOSUB 3600
1810 IF WW<23 THEN DI=100:GOSUB 3580:RETURN
1820 IF WW<31 THEN PRINT"WATERSCHADE !!!":PRINT:
PRINT:GOSUB 3720
1830 IF WW<31 THEN DI=50+INT(RND(1)*50)
1840 IF WW<31 THEN PRINT"GEWAS ROT:";100-DI;"%":
GOSUB 3540:RETURN
1850 PRINT"OVERSTROMINGSRAMP!!!":PRINT:PRINT
1860 DI=INT(RND(1)*50)+1
1870 PRINT"GEWAS ROT:";100-DI;"%":GOSUB 3540:
RETURN
1880 CLS:LOCATE 4,4:PRINT"UW ARBEIDERS EISEN
LOONSVERHOGING"
1890 IF UM=1 THEN GOTO 1960
1900 WI=INT(20*RND(1))+1
1910 PRINT TAB(7)"ZE EISEN";WI;"% ERBIJ":PRINT:
PRINT
1920 SC=INT(100*RND(1))+1
1930 PRINT TAB(4)"ER IS EEN";SC;"% KANS DAT ZE"
1940 PRINT TAB(4)"GAAN STAKEN"
1950 IF UM=0 THEN GOTO 2040
1960 GOSUB 3550
1970 PRINT TAB(2)"OMDAT U HUN LAATSTE CLAIM"
1980 PRINT TAB(2)"HEBT AFGEWEZEN HEBBEN ZE HUN
EISEN"
1990 PRINT TAB(2)"VERHOOGD"
2000 WI=INT(WI*1.5)+3
2010 SC=INT(SC*2)+5
2020 PRINT TAB(2)"DE BONDEN VRAGEN NU":PRINT
TAB(2)WI;"% LOONSVERHOGING"
2030 PRINT:PRINT" DE KANS DAT ZIJ GAAN STAKEN
IS";SC;"%"
2040 PRINT:PRINT"WILT U DE VERHOGING BETALEN
(J/N)"
2050 A$=INKEY$:IF A$="" THEN GOTO 2050
2060 IF A$="J" THEN PR=INT(PR*WI/100)+PR:UM=0:
SK=1:WI=0:SC=0:GOTO 2160
2070 IF A$<>"N" THEN GOTO 2040
2080 CLS
2090 PRINT:PRINT:PRINT TAB(1)"DE VAKBONDSLEIDERS
ZIJN NU IN BERAAD":GOSUB 3550
2100 UB=INT(100*RND(1))+1
2110 IF UB>SC THEN PRINT:PRINT TAB(1)"ZE HEBBEN

```

```

--
--
BESLOTEN NIET TE STAKEN"
2120 IF UB>SC THEN SK=1:UM=1:GOTO 2160
2130 PRINT:PRINT TAB(5)"UW ARBEIDERS STAKEN":GOSUB
3560
2140 PRINT:PRINT TAB(1)"UW OOGST BRENGT 30% MINDER
OP"
2150 UM=0:SC=0:WI=0:SK=.7
2160 GOSUB 3760
2170 RETURN
2180 CLS
2190 IF HD=0 THEN DF=1
2200 IF HD=1 THEN DF=1.5
2210 IF HV=0 THEN VI=1
2220 IF HV=1 THEN VI=1.5
2230 PRINT:PRINT:PRINT TAB(2)"NU MAAR ZIEN WAT DE
OPBRENGST IS...":GOSUB 3550
2240 IN=INT(BP*WP*PP*FP*DI/100*DF*VI*BT*SK*PL*IP*
20)
2250 CLS
2260 LOCATE 6,3:PRINT"UW OOGSTOPBRENGST = $";IN
2270 MB=MB+IN:GOSUB 3600:GOSUB 3580
2280 CLS
2290 LOCATE 0,3:PRINT"NIEUWE BANKBALANS=$";MB
2300 PRINT:PRINT"BEDRIJFSKOSTEN: AF $";(PR*LR)
2310 PRINT:PRINT"TERUGBETALINGEN BANK: AF $";RP
2320 MB=MB-(PR*LR)-RP
2330 PRINT:PRINT:PRINT"BLIJFT OVER: $";MB
2340 LR=INT(AC/500)+1
2350 X=INT(3*RND(1))+1
2360 IF X=1 THEN BL=BL+1000
2370 CL=INT(15*RND(1))*INT(CL/100)+CL
2380 GOSUB 3760
2390 RETURN
2400 X=INT(50*RND(1))+1
2410 Y=INT(X*SH/100)
2420 SH=SH+Y
2430 CLS
2440 PRINT"UW KUDDE IS UITGEBREID MET";Y;"SCHAPEN"
2450 PRINT"U HEBT NU";SH;"SCHAPEN"
2460 IF AC*5<SH THEN SH=AC*5:PRINT
2470 IF AC*5=SH THEN PRINT"OP UW";AC;"BUNDER LAND"
2480 IF AC*5=SH THEN PRINT"KUNNEN SLECHTS";SH;"
SCHAPEN GRAZEN"
2490 GOSUB 3550:X=INT(60*RND(1))-20
2500 IF X<0 THEN A$="GEDAALD"

```

```

2510 IF X>0 THEN A$="GESTEGEN"
2520 PRINT
2530 IF X=0 THEN PRINT"DE WAARDE VAN UW SCHAPEN
IS HETZELFDE":GOTO 2560
2540 PRINT"DE WAARDE VAN UW SCHAPEN IS"
2550 PRINT A$;" MET";ABS(X);"% "
2560 GOSUB 3600:GOSUB 3570
2570 CS=INT(CS*X/100)+CS
2580 CLS:X=0:Y=10:GOSUB 3530
2590 X=INT(50*RND(1)+1)-INT(AC/40)
2600 IF X<1 THEN X=1
2610 IF X=1 THEN GOTO 2700
2620 IF X=2 THEN GOTO 2740
2630 IF X=3 THEN GOTO 2770
2640 IF X=4 THEN GOTO 2790
2650 IF X=5 THEN GOTO 2810
2660 IF X=6 THEN GOTO 2830
2670 IF X=7 THEN GOTO 2850
2680 IF X=8 THEN GOTO 2870
2690 GOTO 2910
2700 X=INT(20*RND(1)+1)
2710 Y=INT(X*SH/100)
2720 PRINT"ENGE ZIEKTE DOODDE";Y;"SCHAPEN"
2730 SH=SH-Y:GOTO 2890
2740 X=INT(10*RND(1)+1)
2750 PRINT"DE BLIKSEM TROF";X;"% VAN DE SCHAPEN":
GOSUB 3560:CLS
2760 BT=BT-(X/100):BB=1:GOTO 2890
2770 PRINT"VRACHTWAGEN TOTAL LOSS"
2780 TK=TK-1:GOTO 2890
2790 PRINT"TREKKER MOET WORDEN VERVANGEN"
2800 TR=TR-1:GOTO 2890
2810 PRINT"JAMMER, SCHUUR VERBRAND"
2820 BN=BN-1:GOTO 2890
2830 PRINT"ZAAIMACHINE INGESTORT, PECH"
2840 HD=HD-1:GOTO 2890
2850 PRINT"OOGSTMACHINE IN DE PRAK"
2860 HV=HV-1:GOTO 2890
2870 PRINT"UW IRRIGATIEPUT IS OPGEDROOGD":GOSUB
3570
2880 DB=0
2890 GOSUB 3720
2900 GOTO 2920
2910 PRINT TAB(1)"VROUWE FORTUNA HEEFT U

```



```

    TOEGELACHEN":GOSUB 3640
2920 GOSUB 3560:FA=FA+1:BB=BB-1:IF BB<0 THEN
    BT=1:BB=0:CLS
2930 IF FA=>50 THEN GOSUB 3680:X=8:Y=6:GOSUB
    3530
2940 IF FA=>50 THEN PRINT"U BENT GEPENSIONEERD":
    COLOR 11
2950 IF FA=>50 THEN GOTO 3000
2960 IF MB+BL<0 THEN GOSUB 3720:X=8:Y=6:GOSUB 3530
2970 IF MB+BL<0 THEN PRINT"U BENT BANKROET":J=0:
    PRINT:PRINT
2980 IF MB+BL<0 THEN GOTO 3050
2990 GOTO 40
3000 FS=MB+(AC*CL)+(TR*2000)+(TK*2000)+(BN*1000)+
    (HR*5000)+(HV*5000)+(SH*CS)
3010 GOSUB 3550
3020 PRINT:PRINT:PRINT TAB(6)"DE WAARDE VAN UW
    BEZIT IS $";FS
3030 J=INT(FS/100000!)
3040 GOSUB 3560
3050 IF J=0 THEN PRINT"DAT WORDT DE BIJSTAND
    MIJN VRIEND":FOR TD=1 TO 2000:NEXT TD
3060 IF J=1 THEN PRINT"U BENT EEN ECHTE SAPPELAAR"
3070 IF J=1 THEN PRINT"HET HOUDT NIET OVER"
3080 IF J=1 THEN PRINT"NA AL DIE HARDE JAREN":FOR
    TD=1 TO 3000:NEXT TD:GOTO 3210
3090 IF J=2 THEN PRINT"U BENT EEN GESLAAGDE BOER"
3100 IF J=2 THEN PRINT"UW PENSIOEN ZAL NIET EEN
    VAN"
3110 IF J=2 THEN PRINT"DE LAAGSTE ZIJN":FOR TD=1
    TO 3000:NEXT TD:GOTO 3210
3120 IF J=3 THEN PRINT"U BENT EEN GOEDE BOER"
3130 IF J=3 THEN PRINT"U ZULT HET ER NU VAN
    KUNNEN NEMEN"
3140 IF J=3 THEN PRINT"TIJDENS UW PENSIOEN":FOR
    TD=1 TO 3000:NEXT TD:GOTO 3210
3150 IF J<4 THEN GOTO 3210
3160 FOR X=1 TO 6
3170 PRINT:PRINT TAB(5)"U BENT EEN UITSTEKENDE
    BOER!"
3180 PRINT:GOSUB 3600
3190 NEXT X
3200 FOR TD=1 TO 3000:NEXT TD
3210 CLS

```



```

3220 LOCATE 10,10:PRINT"NOG EEN SPEL (J/N)
3230 A$=INKEY$
3240 IF A$="J" THEN RUN
3250 IF A$="N" THEN LOCATE 10,12:PRINT"TOT
      ZIENS.....":END
3260 GOTO 3230
3270 PRINT:PRINT"U HEBT GERST GEZAAID":GOSUB 3560
3280 BP=RND(1)*5/10+1
3290 WP=1:PP=1:FP=1
3300 EL=EL+15:CLS:RETURN
3310 PRINT:PRINT"U HEBT TARWE GEZAAID":GOSUB 3560
3320 WP=RND(1)*3/10+1
3330 BP=1:PP=1:FP=1
3340 EL=EL+10:CLS:RETURN
3350 PRINT:PRINT"U HEBT ERWTEN GEZAAID":GOSUB
      3560
3360 PP=RND(1)*1/10+1
3370 BP=1:WP=1:FP=1
3380 EL=EL+10:CLS:RETURN
3390 PRINT:PRINT"UW LAND LIGT BRAAK":GOSUB 3560
3400 FP=0:EL=0
3410 CLS:RETURN
3420 REM INITIALISATIE
3430 CLS
3440 KEY OFF
3450 WIDTH 40
3460 MB=3000:TR=1:BN=1:DB=1
3470 TK=1:AC=350:SH=100:BT=1
3480 CL=100:PR=2000:BL=5000
3490 P=RND(-TIME)
3500 FA=INT(RND(1)*8)+27
3510 LR=1:CS=INT(10*RND(1))+8
3520 CLS:RETURN
3530 LOCATE X,Y:RETURN
3540 FOR TD=1 TO 1000:NEXT TD
3550 FOR TD=1 TO 1000:NEXT TD
3560 FOR TD=1 TO 500:NEXT TD
3570 FOR TD=1 TO 400:NEXT TD
3580 FOR TD=1 TO 100:NEXT TD
3590 RETURN
3600 REM GELUIDSRoutine
3610 PLAY"O4C"
3620 GOSUB 3560
3630 RETURN

```

```
3640 REM GELUIDSROUTINE
3650 PLAY"O5C"
3660 GOSUB 3560
3670 RETURN
3680 REM GELUIDSROUTINE
3690 PLAY"O6C"
3700 GOSUB 3560
3710 RETURN
3720 REM GELUIDSROUTINE
3730 RETURN
3740 REM EERSTE REGEL LEEGMAKEN *****
3750 PRINT CHR$(11);SPC(80):RETURN
3760 LOCATE 8,0:PRINT"GA VERDER NA TOETSDRUK"
3770 B$=INKEY$
3780 IF B$="" THEN GOTO 3770
3790 RETURN
3800 REM TITELPAGINA
3810 CLS
3820 FOR TX=1 TO 23
3830 REM
3840 PRINT TAB(TX)"EIGENHEIMERHEERD"
3850 FOR TD=1 TO 50:NEXT TD
3860 REM
3870 NEXT TX
3880 FOR TD=1 TO 1000:NEXT TD
3890 CLS
3900 RETURN
```


DAAG DE COMPUTER UIT

Boter, kaas en eieren in 3-D

Ontcijfer X-code

Othello

Schaakbord NIM

Hersensbreker


BOTER, KAAS EN EIENEN IN 3-D

Dit programma zorgt voor een bijna onvoorstelbaar spel Boter, Kaas en Eieren in drie dimensies. Wanneer u het spel speelt, moet u zich een driedimensionaal bord proberen voor te stellen. Als u dat niet doet, zult u waarschijnlijk fouten maken.

Bij uitvoering van het programma wordt eerst de titel op het scherm gebracht. Daarna wordt het scherm gewist en een speelbord afgebeeld. Dit bord bestaat uit een speelgedeelte aan de rechterkant van het scherm en een sleutel tot het speelbord aan de linkerkant. U moet zich voorstellen dat de borden op elkaar zijn gestapeld: boven, midden en onder. U ziet de borden van boven. De computer zet de kruisjes en u plaatst de nulletjes. In dit spel hebben we de letter O gebruikt in plaats van het cijfer 0.

In de regels 40-60 wordt willekeurig bepaald wie er bij elke spelronde mag beginnen. Als u aan de beurt bent, geeft de computer dat aan. Als u gaat zetten, voert u het nummer in van het gebied (het driedimensionale veld) dat u wilt bezetten. Dit nummer wordt aangegeven op het sleutelspelbord aan de linkerkant van het scherm.

Om te winnen moet u drie nulletjes op een rij zien te krijgen, verticaal, horizontaal of diagonaal. Dat geldt niet alleen voor een plat vlak, zoals bij het gewone Boter, Kaas en Eieren, maar ook voor 3 nullen op een rij 'door de borden heen'.



Dat kunt u bijvoorbeeld bereiken door de gebieden met de getallen 1, 10 en 19 of 5, 14 en 23 te bezetten. Drie op een rij langs een van de diagonale assen betekent ook winst: bijvoorbeeld 1, 14 en 27 of 9, 14 en 19. Let op: een combinatie van 1, 11 en 21 is geen winnend rijtje. Alle diagonale rijen door de kubus heen moeten in elk geval gebied 14 op het middelste bord bezetten om als winnend rijtje te kunnen gelden.

Het programma speelt een oneindig aantal ronden en houdt bij hoeveel keer de speler en hoeveel keer de computer heeft gewonnen. Als u wilt stoppen, druk dan op CTRL-STOP.

In regel 210-380 worden het bord en de sleutel tot het bord afgebeeld. Het werkelijke speelgebied wordt afgebeeld door een karakter af te drukken (door middel van de PRINT-instructie CHR\$(X)), dat -vertaald in ASCII-code- is opgeslagen in elke locatie van de variabele A(X). Deze waarde kan de ASCII-code voor X,O zijn of een punt. De punt wordt gebruikt om een leeg veld aan te geven.

De computer gebruikt de lange routine van de regels 1070-2470 om zijn zet te doen. Hier is zeker niet op de meest effectieve manier geprogrammeerd. Dat is met opzet gedaan. Deze vorm is de meest leesbare en zo zijn foutjes ook makkelijker op te sporen. De regellengte blijft binnen de 80 karakters, zodat de regels leesbaar blijven. Zo wordt vermeden dat u lange zinnen moet opdelen, waardoor rommelige stukjes ontstaan. Tenslotte is het ook makkelijker om alles in te typen. Als u bijvoorbeeld de regels 1090 en 1150 intypt, kunt u de cursorbesturingstoetsen gebruiken om regelnummers en getalswaarden zo te veranderen dat u bijna iedere andere regel in deze routine kunt overnemen zonder elke keer alles te moeten intypen.

```
10 REM BOTER KAAS & EIEREN IN 3-D
20 GOSUB 3020:REM STARTROUTINE
30 GOTO 2860:REM INITIALISATIE
40 GT=RND(-TIME):IF GT>.5 THEN LOCATE 5,6:PRINT"
   U MAG EERST"
50 IF GT>.5 THEN GOSUB 3210
60 IF GT>.5 THEN FOR Z=1 TO 2000:NEXT Z:GOTO 160
70 GOSUB 220:REM PRINT HET BORD
80 GOSUB 400:REM WIN CHECK
90 IF Q<>0 OR M=99 THEN 2710:REM EINDE
100 M=0
110 GOSUB 1070:REM EEN BEWEGING MAKEN
120 IF M=0 OR A(M)<>E THEN M=0:GOSUB 2540
130 PRINT"IK SPEEL OP";M
140 FOR TD=1 TO 2000:NEXT TD
```

```

150 A(M)=B
160 GOSUB 220:REM PRINT HET BORD
170 GOSUB 400:REM WIN CHECK
180 IF Q<>0 OR M=99 THEN 2710:REM EINDE
190 GOSUB 1000:REM BEWEGING VAN DE SPELER
200 GOTO 70
210 REM *****
220 REM PRINT HET BORD
230 CLS
240 GOSUB 3230
250 PRINT TAB(8);"BOVENSTE":PRINT
260 PRINT TAB(8);" 1 2 3 ";CHR$(A(1));" ";
  CHR$(A(2));" ";CHR$(A(3))
270 PRINT TAB(8);" 4 5 6 ";CHR$(A(4));" ";
  CHR$(A(5));" ";CHR$(A(6))
280 PRINT TAB(8);" 7 8 9 ";CHR$(A(7));" ";
  CHR$(A(8));" ";CHR$(A(9))
290 PRINT:PRINT TAB(8);"MIDDELSTE":PRINT
300 PRINT TAB(8);"10 11 12 ";CHR$(A(10));" ";
  CHR$(A(11));" ";CHR$(A(12))
310 PRINT TAB(8);"13 14 15 ";CHR$(A(13));" ";
  CHR$(A(14));" ";CHR$(A(15))
320 PRINT TAB(8);"16 17 18 ";CHR$(A(16));" ";
  CHR$(A(17));" ";CHR$(A(18))
330 PRINT:PRINT TAB(8);"ONDERSTE":PRINT
340 PRINT TAB(8);"19 20 21 ";CHR$(A(19));" ";
  CHR$(A(20));" ";CHR$(A(21))
350 PRINT TAB(8);"22 23 24 ";CHR$(A(22));" ";
  CHR$(A(23));" ";CHR$(A(24))
360 PRINT TAB(8);"25 26 27 ";CHR$(A(25));" ";
  CHR$(A(26));" ";CHR$(A(27))
370 PRINT
380 RETURN
390 REM *****
400 REM WIN CHECK
410 Q=0
420 IF A(1)=E THEN 480
430 IF A(1)=A(2) AND A(2)=A(3) THEN Q=1:RETURN
440 IF A(1)=A(4) AND A(4)=A(7) THEN Q=1:RETURN
450 IF A(1)=A(5) AND A(5)=A(9) THEN Q=1:RETURN
460 IF A(1)=A(10) AND A(10)=A(19) THEN Q=1:
  RETURN

```



```
470 IF A(1)=A(14) AND A(14)=A(27) THEN Q=1:
    RETURN
480 IF A(10)=E THEN 520
490 IF A(10)=A(11) AND A(11)=A(12) THEN Q=10:
    RETURN
500 IF A(10)=A(13) AND A(13)=A(16) THEN Q=10:
    RETURN
510 IF A(10)=A(14) AND A(14)=A(18) THEN Q=10:
    RETURN
520 IF A(19)=E THEN 560
530 IF A(19)=A(20) AND A(20)=A(21) THEN Q=19:
    RETURN
540 IF A(19)=A(22) AND A(22)=A(25) THEN Q=19:
    RETURN
550 IF A(19)=A(23) AND A(23)=A(27) THEN Q=19:
    RETURN
560 IF A(2)=E THEN 590
570 IF A(2)=A(5) AND A(5)=A(8) THEN Q=2:RETURN
580 IF A(2)=A(11) AND A(11)=A(20) THEN Q=2:
    RETURN
590 IF A(11)=E THEN 610
600 IF A(11)=A(14) AND A(14)=A(17) THEN Q=11:
    RETURN
610 IF A(20)=E THEN 630
620 IF A(20)=A(23) AND A(23)=A(26) THEN Q=20:
    RETURN
630 IF A(3)=E THEN 680
640 IF A(3)=A(6) AND A(6)=A(9) THEN Q=3:RETURN
650 IF A(3)=A(12) AND A(12)=A(21) THEN Q=3:RETURN
660 IF A(3)=A(5) AND A(5)=A(7) THEN Q=3:RETURN
670 IF A(3)=A(14) AND A(14)=A(25) THEN Q=3:RETURN
680 IF A(12)=E THEN 710
690 IF A(12)=A(15) AND A(15)=A(18) THEN Q=12:
    RETURN
700 IF A(12)=A(14) AND A(14)=A(16) THEN Q=12:
    RETURN
710 IF A(21)=E THEN 740
720 IF A(21)=A(24) AND A(24)=A(27) THEN Q=21:
    RETURN
730 IF A(21)=A(23) AND A(23)=A(25) THEN Q=21:
    RETURN
740 IF A(4)=E THEN 770
750 IF A(4)=A(5) AND A(5)=A(6) THEN Q=4:RETURN
760 IF A(4)=A(13) AND A(13)=A(22) THEN Q=4:RETURN
```

```

770 IF A(13)=E THEN 790
780 IF A(13)=A(14) AND A(14)=A(15) THEN Q=13:
RETURN
790 IF A(22)=E THEN 810
800 IF A(22)=A(23) AND A(23)=A(24) THEN Q=22:
RETURN
810 IF A(6)=E THEN 830
820 IF A(6)=A(15) AND A(15)=A(24) THEN Q=6:RETURN
830 IF A(5)=E THEN 850
840 IF A(5)=A(14) AND A(14)=A(23) THEN Q=5:RETURN
850 IF A(7)=E THEN 890
860 IF A(7)=A(8) AND A(8)=A(9) THEN Q=7:RETURN
870 IF A(7)=A(16) AND A(16)=A(25) THEN Q=7:RETURN
880 IF A(7)=A(14) AND A(14)=A(21) THEN Q=7:RETURN
890 IF A(17)=E THEN 910
900 IF A(17)=A(16) AND A(17)=A(18) THEN Q=16:
RETURN
910 IF A(25)=E THEN 930
920 IF A(25)=A(26) AND A(26)=A(27) THEN Q=25:
RETURN
930 IF A(8)=E THEN 950
940 IF A(8)=A(17) AND A(17)=A(26) THEN Q=8:RETURN
950 IF A(9)=E THEN 980
960 IF A(9)=A(18) AND A(18)=A(27) THEN Q=9:RETURN
970 IF A(9)=A(14) AND A(14)=A(19) THEN Q=9:RETURN
980 RETURN
990 REM *****
1000 REM ACCEPTEER SPELERZET
1010 PRINT "UW ZET:";:INPUT Y
1020 IF Y<1 OR Y>27 THEN 1010
1030 IF A(Y)<>E THEN 1010
1040 A(Y)=C
1050 RETURN
1060 REM *****
1070 REM EEN ZET DOEN
1080 M=1:IF A(1)<>E THEN 1150
1090 IF A(1)=A(3) AND A(1)<>E THEN M=2:RETURN
1100 IF A(2)=A(3) AND A(2)<>E THEN RETURN
1110 IF A(4)=A(7) AND A(4)<>E THEN RETURN
1120 IF A(5)=A(9) AND A(5)<>E THEN RETURN
1130 IF A(10)=A(19) AND A(10)<>E THEN RETURN
1140 IF A(14)=A(27) AND A(14)<>E THEN RETURN
1150 M=3:IF A(3)<>E THEN 1210
1160 IF A(2)=A(1) AND A(2)<>E THEN RETURN

```

```

1170 IF A(6)=A(9) AND A(6)<>E THEN RETURN
1180 IF A(5)=A(7) AND A(5)<>E THEN RETURN
1190 IF A(12)=A(21) AND A(12)<>E THEN RETURN
1200 IF A(14)=A(25) AND A(14)<>E THEN RETURN
1210 M=4:IF A(4)<>E THEN 1250
1220 IF A(1)=A(7) AND A(1)<>E THEN RETURN
1230 IF A(5)=A(6) AND A(5)<>E THEN RETURN
1240 IF A(13)=A(22) AND A(13)<>E THEN RETURN
1250 M=5:IF A(5)<>E THEN 1310
1260 IF A(1)=A(9) AND A(1)<>E THEN RETURN
1270 IF A(2)=A(8) AND A(2)<>E THEN RETURN
1280 IF A(3)=A(7) AND A(3)<>E THEN RETURN
1290 IF A(4)=A(6) AND A(4)<>E THEN RETURN
1300 IF A(14)=A(23) AND A(14)<>E THEN RETURN
1310 M=6:IF A(6)<>E THEN 1350
1320 IF A(4)=A(5) AND A(4)<>E THEN RETURN
1330 IF A(3)=A(9) AND A(3)<>E THEN RETURN
1340 IF A(15)=A(24) AND A(15)<>E THEN RETURN
1350 M=7:IF A(7)<>E THEN 1400
1360 IF A(1)=A(4) AND A(1)<>E THEN RETURN
1370 IF A(8)=A(9) AND A(8)<>E THEN RETURN
1380 IF A(5)=A(3) AND A(5)<>E THEN RETURN
1390 IF A(16)=A(25) AND A(16)<>E THEN RETURN
1400 M=8:IF A(8)<>E THEN 1440
1410 IF A(7)=A(9) AND A(7)<>E THEN RETURN
1420 IF A(5)=A(2) AND A(5)<>E THEN RETURN
1430 IF A(17)=A(26) AND A(17)<>E THEN RETURN
1440 M=9:IF A(9)<>E THEN 1500
1450 IF A(7)=A(8) AND A(7)<>E THEN RETURN
1460 IF A(6)=A(3) AND A(6)<>E THEN RETURN
1470 IF A(1)=A(5) AND A(1)<>E THEN RETURN
1480 IF A(18)=A(27) AND A(18)<>E THEN RETURN
1490 IF A(14)=A(19) AND A(14)<>E THEN RETURN
1500 M=10:IF A(10)<>E THEN 1550
1510 IF A(11)=A(12) AND A(11)<>E THEN RETURN
1520 IF A(13)=A(16) AND A(13)<>E THEN RETURN
1530 IF A(14)=A(18) AND A(14)<>E THEN RETURN
1540 IF A(1)=A(19) AND A(1)<>E THEN RETURN
1550 M=11:IF A(11)<>E THEN 1590
1560 IF A(10)=A(12) AND A(10)<>E THEN RETURN
1570 IF A(14)=A(17) AND A(14)<>E THEN RETURN
1580 IF A(2)=A(20) AND A(2)<>E THEN RETURN
1590 M=12:IF A(12)<>E THEN 1640
1600 IF A(10)=A(11) AND A(10)<>E THEN RETURN

```

```

1610 IF A(15)=A(18) AND A(15)<>E THEN RETURN
1620 IF A(14)=A(16) AND A(14)<>E THEN RETURN
1630 IF A(3)=A(21) AND A(3)<>E THEN RETURN
1640 M=20:IF A(20)<>E THEN 1680
1650 IF A(19)=A(21) AND A(19)<>E THEN RETURN
1660 IF A(23)=A(26) AND A(23)<>E THEN RETURN
1670 IF A(2)=A(11) AND A(2)<>E THEN RETURN
1680 M=21:IF A(21)<>E THEN 1740
1690 IF A(19)=A(20) AND A(19)<>E THEN RETURN
1700 IF A(24)=A(27) AND A(24)<>E THEN RETURN
1710 IF A(23)=A(25) AND A(23)<>E THEN RETURN
1720 IF A(3)=A(12) AND A(3)<>E THEN RETURN
1730 IF A(14)=A(7) AND A(14)<>E THEN RETURN
1740 M=16:IF A(16)<>E THEN 1790
1750 IF A(10)=A(13) AND A(10)<>E THEN RETURN
1760 IF A(17)=A(18) AND A(17)<>E THEN RETURN
1770 IF A(14)=A(12) AND A(14)<>E THEN RETURN
1780 IF A(7)=A(25) AND A(7)<>E THEN RETURN
1790 M=17:IF A(17)<>E THEN 1830
1800 IF A(14)=A(11) AND A(14)<>E THEN RETURN
1810 IF A(16)=A(18) AND A(16)<>E THEN RETURN
1820 IF A(8)=A(26) AND A(8)<>E THEN RETURN
1830 M=18:IF A(18)<>E THEN 1880
1840 IF A(16)=A(17) AND A(16)<>E THEN RETURN
1850 IF A(15)=A(12) AND A(15)<>E THEN RETURN
1860 IF A(14)=A(10) AND A(14)<>E THEN RETURN
1870 IF A(9)=A(27) AND A(9)<>E THEN RETURN
1880 M=19:IF A(19)<>E THEN 1940
1890 IF A(20)=A(21) AND A(20)<>E THEN RETURN
1900 IF A(22)=A(25) AND A(22)<>E THEN RETURN
1910 IF A(23)=A(27) AND A(23)<>E THEN RETURN
1920 IF A(1)=A(10) AND A(1)<>E THEN RETURN
1930 IF A(14)=A(9) AND A(14)<>E THEN RETURN
1940 M=13:IF A(13)<>E OR RND(-TIME)>.2 THEN 2000
1950 IF A(14)=E AND A(15)=E OR A(16)=E AND A(10)=E
    THEN 2000
1960 IF A(10)=A(16) AND A(10)<>E THEN RETURN
1970 IF A(2)=A(22) AND A(2)<>E THEN RETURN
1980 IF A(14)=A(15) AND A(14)<>E THEN RETURN
1990 IF A(M)=E THEN RETURN
2000 M=14:IF A(14)<>E THEN 2100
2010 IF A(10)=A(18) AND A(10)<>E THEN RETURN
2020 IF A(12)=A(16) AND A(12)<>E THEN RETURN
2030 IF A(13)=A(15) AND A(13)<>E THEN RETURN

```

```

2040 IF A(11)=A(17) AND A(11)<>E THEN RETURN
2050 IF A(5)=A(23) AND A(5)<>E THEN RETURN
2060 IF A(1)=A(27) AND A(1)<>E THEN RETURN
2070 IF A(3)=A(25) AND A(3)<>E THEN RETURN
2080 IF A(7)=A(21) AND A(7)<>E THEN RETURN
2090 IF A(9)=A(19) AND A(9)<>E THEN RETURN
2100 M=15:IF A(15)<>E THEN 2140
2110 IF A(12)=A(18) AND A(12)<>E THEN RETURN
2120 IF A(13)=A(14) AND A(13)<>E THEN RETURN
2130 IF A(6)=A(24) AND A(6)<>E THEN RETURN
2140 M=22:IF A(22)<>E THEN GOTO 2180
2150 IF A(19)=A(25) AND A(19)<>E THEN RETURN
2160 IF A(23)=A(24) AND A(23)<>E THEN RETURN
2170 IF A(4)=A(13) AND A(4)<>E THEN RETURN
2180 M=23:IF A(23)<>E THEN 2240
2190 IF A(22)=A(24) AND A(22)<>E THEN RETURN
2200 IF A(20)=A(26) AND A(20)<>E THEN RETURN
2210 IF A(19)=A(27) AND A(19)<>E THEN RETURN
2220 IF A(15)=A(21) AND A(15)<>E THEN RETURN
2230 IF A(5)=A(14) AND A(5)<>E THEN RETURN
2240 M=24:IF A(24)<>E THEN 2280
2250 IF A(21)=A(27) AND A(21)<>E THEN RETURN
2260 IF A(22)=A(23) AND A(22)<>E THEN RETURN
2270 IF A(6)=A(15) AND A(6)<>E THEN RETURN
2280 M=25:IF A(25)<>E THEN 2330
2290 IF A(19)=A(22) AND A(19)<>E THEN RETURN
2300 IF A(26)=A(27) AND A(26)<>E THEN RETURN
2310 IF A(23)=A(21) AND A(23)<>E THEN RETURN
2320 IF A(7)=A(16) AND A(7)<>E THEN RETURN
2330 M=26:IF A(26)<>E THEN 2370
2340 IF A(23)=A(20) AND A(23)<>E THEN RETURN
2350 IF A(25)=A(27) AND A(25)<>E THEN RETURN
2360 IF A(8)=A(17) AND A(8)<>E THEN RETURN
2370 M=27:IF A(27)<>E THEN 2430
2380 IF A(25)=A(26) AND A(25)<>E THEN RETURN
2390 IF A(24)=A(21) AND A(24)<>E THEN RETURN
2400 IF A(19)=A(23) AND A(19)<>E THEN RETURN
2410 IF A(9)=A(18) AND A(9)<>E THEN RETURN
2420 IF A(1)=A(14) AND A(1)<>E THEN RETURN
2430 M=2:IF A(2)<>E THEN 2490
2440 IF A(1)=E AND A(3)=E OR A(5)=E AND A(8)=E
    THEN 2490
2450 IF A(1)=A(3) AND A(1)<>E THEN RETURN
2460 IF A(11)=A(20) AND A(11)<>E THEN RETURN

```

```

2470 IF A(5)=A(8) AND A(5)<>E THEN RETURN
2480 REM *****
2490 IF RND(-TIME)>.7 THEN 2510
2500 IF A(14)=E THEN M=14:RETURN
2510 IF A(5)=E THEN M=5:RETURN
2520 IF A(23)=E THEN M=23:RETURN
2530 M=0
2540 Z=0
2550 Z=Z+1
2560 M=INT(RND(-TIME)*12)+1
2570 IF A(M)=E THEN RETURN
2580 IF Z<100 THEN 2550
2590 Z=0
2600 M=0
2610 Z=Z+1
2620 M=INT(RND(-TIME)*27)+1
2630 IF A(M)=E THEN RETURN
2640 IF Z<500 THEN 2610
2650 PRINT:PRINT"GELIJK SPEL !!!"
2660 FOR TD=1 TO 2000:NEXT TD
2670 M=99
2680 RETURN
2690 GOTO 2750
2700 REM *****
2710 REM EINDE VAN HET SPEL
2720 PRINT
2730 IF A(Q)=B THEN PRINT"IK HEB GEWONNEN!!":MS=
MS+1
2740 IF A(Q)=C THEN PRINT"U WINT ....":HS=HS+1
2750 PRINT
2760 IF Q=0 THEN GOTO 100
2770 PRINT"MIJN SCORE IS:";MS
2780 PRINT"UW SCORE IS: ";HS
2790 FOR Z=1 TO 27
2800 A(Z)=E
2810 NEXT Z
2820 FOR TD=1 TO 2000:NEXT TD
2830 COLOR 1,13
2840 GOTO 40
2850 REM *****
2860 REM INITIALISATIE
2870 CLS
2880 SCREEN 0
2890 KEY OFF

```

```

2900 DIM A(27),D(12)
2910 B=ASC("X"):REM STUK VAN DE COMPUTER
2920 C=ASC("O"):REM STUK VAN DE SPELER
2930 E=ASC("."):REM ONGEBRUIKTE PLAATS
2940 M=0:Q=0
2950 HS=0:MS=0
2960 FOR T=1 TO 12
2970 READ Q
2980 D(T)=Q
2990 NEXT T
3000 GOTO 2790
3010 DATA 1,3,7,9,19,21,25,27,10,12,16,18
3020 REM BEGINPLAATJE *****
3030 CLS:KEY OFF
3040 COLOR 12,1
3050 FOR Y=1 TO 22
3060 LOCATE 8,Y-1:PRINT"
3070 LOCATE 8,Y:PRINT"BOTER KAAS & EIEREN"
3080 FOR TD=1 TO 100:NEXT TD
3090 NEXT Y
3100 CLS
3110 XX$="          BOTER KAAS & EIEREN"
3120 GX=35
3130 FOR XG=1 TO 35
3140 G2$=MID$(XX$,GX,XG)
3150 LOCATE 0,11:PRINT G2$;
3160 FOR Z=1 TO 75:NEXT Z
3170 GX=GX-1
3180 NEXT XG
3190 FOR Z=1 TO 500:NEXT Z
3200 RETURN
3210 PLAY"O5CEG"
3220 RETURN
3230 PLAY"O5GEC"
3240 RETURN

```

ONTCIJFER X-CODE

Reactie en inzicht zijn de vereisten voor het spel ONTCIJFER X-CODE. De computer kiest uit een reeks van 237 woorden willekeurig een woord, klutst het door elkaar en laat het van rechts naar links over het scherm lopen. Aan u de taak om het bedoelde woord in te typen voordat het door elkaar geklutste woord de muur bereikt heeft.

U kunt het spel op tien verschillende niveaus spelen. Niveau 1 is het gemakkelijkste niveau. Het niveauverschil is gelegen in de afstand van de muur tot de rechter beeldschermrand. Hoe verder de muur naar rechts staat, des te minder tijd heeft u om de juiste oplossing in te typen.

Nadat in regel 80 de verschillende woorden zijn ingelezen (door middel van de READ-instructie), wordt in regel 90 de subroutine aangeroepen die de in het spel gebruikte sprites definieert.

Twee van de sprites zijn happer-monstertjes, die bij de titelpagina worden gebruikt om de titel op te eten. De derde sprite is een explosie, die plaatsvindt wanneer het woord tegen de muur botst.



Het aantal beurten dat u heeft gespeeld, wordt bijgehouden in regel 110 (variabele BE). De variabele OK in regel 990 houdt het aantal keren bij dat u het juiste woord heeft geraden. Deze twee variabelen worden in regel 1090 en 1110 ook benut om de verhouding beurten – goede oplossingen weer te geven.

Het hoofdprogramma van ONTCIJFER X-CODE bevindt zich in de regels 220–280. Daar wordt stapsgewijs het geklutste woord van rechts naar links bewogen. Met behulp van de instructie ON INTERVAL GOSUB springt het programma elke halve seconde naar de intervalroutine in regel 610–700, waarin de computer het toetsenbord uitleest, de getypte letters PRINT en de verstreken tijd rechtsboven in beeld PRINT.

De regels 1300–1630 zijn DATA-regels waarin de verschillende woorden en de DATA voor de sprites staan.

Wanneer u de opbouw van het programma goed doorhebt, zal het een kleine moeite zijn om andere, eventueel ook langere woorden in de DATA-regels te zetten waardoor u het spel naar eigen inzicht kunt aanpassen.

```
10 REM ONTCIJFER X-CODE *****
20 CLS:KEY OFF
30 COLOR 15,4,4
40 SCREEN 1,2
50 Q=RND(-TIME)
60 DIM S$(238)
70 ON INTERVAL=30 GOSUB 610
80 FOR T=0 TO 236:READ S$(T)::NEXT T
90 GOSUB 440
100 GOSUB 1640
110 BE=BE+1
120 GOSUB 310
130 GOSUB 710
140 CLS
150 LOCATE Z,5:PRINT CHR$(206)
160 FOR Y=6 TO 16:LOCATE Z,Y:PRINT" "
170 REM GRAPH SHIFT-|
180 LOCATE Z,17:PRINT CHR$(205)
190 NEXT Y
200 X=23:Y=10
210 INTERVAL ON
220 REM HOOFDPROGRAMMA *****
230 LOCATE X+1,Y:PRINT SPC(7)
```

```

240 LOCATE X,Y:PRINT G$(G1);G$(G2);G$(G3);G$(G4);
    G$(G5)
250 FOR TD=1 TO 500:NEXT TD
260 X=X-1
270 IF X<Z+1 THEN GOTO 1160
280 GOTO 230
290 FOR TD=1 TO 1000:NEXT TD
300 PRINT G$(G1);G$(G2);G$(G3);G$(G4);G$(G5)
310 REM KLUTS HET WOORD *****
320 G1=0:G2=0:G3=0:G4=0:G5=0
330 F$="":B$=""
340 T=INT(RND(1)*237)+0:F$=S$(T)
350 LE=LEN(F$)
360 FOR G=1 TO LE:G$(G)=MID$(F$,G,1):NEXT G
370 G1=INT(RND(1)*LE)+1
380 G2=INT(RND(1)*LE)+1:IF G2=G1 THEN GOTO 380
390 G3=INT(RND(1)*LE)+1:IF G3=G1 OR G3=G2 THEN
    GOTO 390
400 G4=INT(RND(1)*LE)+1:IF G4=G1 OR G4=G2 OR G4=G3
    THEN GOTO 400
410 IF LE<5 THEN G5=0
420 FOR T=1 TO LE:IF G1<>T AND G2<>T AND G3<>T AND
    G4<>T THEN G5=T
430 NEXT T:RETURN
440 REM SPRITE-DEFINIERING *****
450 FOR I=1 TO 32
460 READ B%
470 S0$=S0$+CHR$(B%)
480 NEXT I
490 SPRITE$(0)=S0$
500 FOR I=1 TO 32
510 READ B%
520 S1$=S1$+CHR$(B%)
530 NEXT I
540 SPRITE$(1)=S1$
550 FOR I=1 TO 32
560 READ B%
570 S2$=S2$+CHR$(B%)
580 NEXT I
590 SPRITE$(2)=S2$
600 RETURN
610 REM INTERVALROUTINE *****
620 A$=INKEY$

```

```

630 B$=B$+A$
640 IF A$<>"" THEN BEEP
650 LOCATE 13,0:PRINT B$
660 IF B$=F$ THEN GOTO 980
670 IF LEN(B$)=LEN(F$) THEN GOTO 1230
680 LOCATE 20,0:PRINT"TIJD:";INT((TIME/60-TIME/
3600)
690 LOCATE X,Y
700 RETURN
710 REM WELK NIVEAU *****
720 CLS
730 LOCATE 0,11
740 PRINT"WELK NIVEAU WILT U? 1-10";
750 INPUT N
760 IF N<1 OR N>10 THEN GOTO 720
770 IF N=1 THEN Z=5
780 IF N=2 THEN Z=6
790 IF N=3 THEN Z=7
800 IF N=4 THEN Z=8
810 IF N=5 THEN Z=9
820 IF N=6 THEN Z=10
830 IF N=7 THEN Z=11
840 IF N=8 THEN Z=12
850 IF N=9 THEN Z=13
860 IF N=10 THEN Z=14
870 TIME=0
880 RETURN
890 REM NOG EEN SPEL *****
900 SCREEN 1,2
910 CLS
920 INTERVAL OFF
930 LOCATE 6,10:PRINT"NOG EEN SPEL (J/N)"
940 Z$=INKEY$
950 IF Z$="J" THEN GOTO 110
960 IF Z$="N" THEN CLS:END
970 GOTO 940
980 REM GOED GERADEN *****
990 OK=OK+1
1000 LOCATE 0,23
1010 FOR A=1 TO 23:PRINT:NEXT A
1020 LOCATE 3,12
1030 PRINT"U HEBT HET GOED GERADEN"
1040 LOCATE 3,13
1050 PRINT"HET WAS INDERDAAD ";B$

```

```

1060 LOCATE 3,14:PRINT"U HEBT HET IN";INT((TIME/
      60-TIME/3600)-1;"SEC."
1070 LOCATE 3,15:PRINT"GERADEN"
1080 LOCATE 3,16
1090 PRINT"DIT WAS DE";OK;"E KEER"
1100 LOCATE 3,17
1110 PRINT"VAN DE";BE;"BEURTEN DAT U HET"
1120 LOCATE 3,18
1130 PRINT"GOED GERADEN HEEFT"
1140 FOR TD=1 TO 3000:NEXT TD
1150 GOTO 890
1160 REM TIJDSOVERSCHRIJDING *****
1170 PUT SPRITE 0,((Z+1)*8,75),2,0
1180 LOCATE 3,19:PRINT"U HEBT DE TIJDSLEMIET"
1190 LOCATE 3,20:PRINT"OVERSCHREDEN, HET WOORD"
1200 LOCATE 3,21:PRINT"WAS: ";F$
1210 FOR TD=1 TO 2000:NEXT TD
1220 GOTO 890
1230 REM FOUT
1240 LOCATE 3,19
1250 PRINT"U HEBT ";B$;" GERADEN, DAT"
1260 LOCATE 3,20
1270 PRINT"WAS FOUT, HET WAS ";F$
1280 FOR TD=1 TO 2000:NEXT TD
1290 GOTO 890
1300 REM WOORDEN *****
1310 DATA AMPER,DACHT,AARD,DRAAD,AMBER,AREND,
      ADEL,AARDE,ANGST,APEN,AJUIN,ABDIJ
1320 DATA ABEEL,DADEL,AKTE,ARKEN,ACTIE,BOORD,
      BOOM,BORST,BLOOT,BERG,BREUK,BOREN
1330 DATA BEKER,BAKER,BORD,BALEN,BENEN,BODEM,
      BROM,BARST,BREEK,BERM,BEREN,BUREN
1340 DATA DORST,DRANK,DUIM,DOVEN,DUREN,DRINK,
      DURF,DRUIF,DENKT,DREK,DROOM,DUURT
1350 DATA EERST,ERGER,CENT,ENGER,ERVEN,ERKER,
      EGEL,ENKEL,EINDE,ELKE,EVERS,EIGEN
1360 DATA FRAAI,FRUIT,FOEI,FRANK,FLAUW,FLUIM,
      FOUT,FORSE,FLUIT,FORT,FRONS,FRIES
1370 DATA GRAAG,GERST,GORT,GROUW,GROEI,GUNST,
      GEEL,GAREN,GEEFT,GIER,GEURT,GELDT
1380 DATA HAALT,HOORT,HERT,HANGT,HAAKT,HALEN,
      HONK,HALVE,HEKEL,HIER,HAREM,HATEN
1390 DATA IMKER,IMMER,INKT,IJLDE,IJDER,INDIE,
      INDO,IJVER,IJDEL,INCA,IJZER,IJZIG

```

```

1400 DATA KASSA,KARIG,KOLK,KOERD,KOORD,KEURT,
      KERK,KEVER,KERST,KURK,KORTE,KORST
1410 DATA JUICH,JOKTE,JONK,JOKER,JOLIG,JOUWE,
      JADE,JARIG,JODEL,JURK,JAKJE,JUIST
1420 DATA LEVER,LUIDE,LOOF,LOSTE,LURKT,LIEVE,
      LINK,LEKKE,LUCHT,LORD,LACHT,LOENS
1430 DATA MAART,MACHT,MEID,MAGER,MINDE,MAAND,
      MELK,MOTEN,MEREN,MOND,MEEST,MOEST
1440 DATA NEGER,NAGEL,NAAR,NOORD,NOEST,NEDER,
      NOCH,NOGAL,NAAKT,NODE,NADER,NETEL
1450 DATA ORDER,OPPER,OORD,OLIJK,OUDER,OUWEL,
      OVER,ONDER,OTTER,ORDE,OKSEL,ORGEL
1460 DATA PAARD,POVER,PARK,PORDE,POKER,PRIMA,
      PLAN,PLENS,PUIST,PLEK,PEKEL,PLUUT
1470 DATA REEDS,RECHT,RAGE,RATEL,ROVER,REGEL,
      RAND,RAKEL,ROUTE,ROOK,RAKET,RUSTE
1480 DATA SLAAP,SLOEP,SLOP,SLEUF,SCHAP,SCHEP,
      STIK,STERK,SARDE,SMET,SONDE,SORES
1490 DATA TEGEL,TERGT,TELG,TAART,TUNER,TOREN,
      TAND,TEDER,TEPEL,TALK,TOEFT,TAFEL
1500 DATA UIVER,UITEN,UIEN,UNIEK,UILEN,URKER,
      VIER,VERSE,VOREN
1510 REM SPRITE-DATA *****
1520 DATA 0,97,97,1,1,3,7,127
1530 DATA 127,7,3,1,1,97,97,0
1540 DATA 0,134,134,128,128,192,224,254
1550 DATA 254,224,192,128,128,134,134,0
1560 DATA 31,63,127,127,127,127,127,127
1570 DATA 127,127,127,127,127,127,63,31
1580 DATA 248,252,158,159,254,248,224,128
1590 DATA 128,224,248,254,159,158,252,248
1600 DATA 31,63,126,126,127,127,127,127
1610 DATA 127,127,127,127,126,126,63,31
1620 DATA 192,224,144,152,252,254,255,144
1630 DATA 144,255,254,252,152,144,224,192
1640 REM TITELPAGINA *****
1650 CLS
1660 COLOR 15,1,1
1670 R=0
1680 LOCATE 6,11
1690 PRINT"NOIJCTRFE EOC-XD"
1700 FOR TD=1 TO 1000:NEXT TD
1710 LOCATE 6,11
1720 PRINT"ONIJCTFRE E-COXD"

```

```
1730 FOR TD=1 TO 1000:NEXT TD
1740 LOCATE 6,11
1750 PRINT"ONTIJCFRE X-COED"
1760 FOR TD=1 TO 1000:NEXT TD
1770 LOCATE 6,11
1780 PRINT"ONTCIJFER X-CODE"
1790 FOR Q=0 TO 255 STEP 2:IF F=1 THEN PUT
    SPRITE 1,(Q,82),9,2:F=0 ELSE PUT SPRITE 2,
    (Q,82),9,1:F=1
1800 IF Q/8=INT(Q/8) THEN LOCATE R,11:PRINT
    CHR$(32):R=R+1
1810 FOR TD=1 TO 50:NEXT TD
1820 NEXT Q
1830 PUT SPRITE 1,(100,100),0,1
1840 PUT SPRITE 2,(100,100),0,1
1850 RETURN
```

OTHELLO


U kunt nu de computer uitdagen om een spelletje OTHELLO te spelen. De ge-computeriseerde versie van dit populaire bordspel laat u vele uren zwaar hersenwerk verrichten in uw pogingen om de machtige MSX te verslaan. U zult zien dat de computer dit spel erg goed speelt en dat het erg moeilijk is om te winnen.

OTHELLO wordt gespeeld op een bord van 8 bij 8. De spelers plaatsen om de beurt een stuk op het bord. De stukken worden zo op het bord gezet dat één of meer stukken van uw tegenspeler gevangen zitten tussen twee van uw stukken. De gevangen stukken worden dan van u. Aan het eind van het spel is degene die de meeste stukken heeft winnaar.

Ieder speler begint met twee stukken op het bord. Uw speelstukken zijn de dichte rondjes en de stukken van de computer zijn de open rondjes (karakters 248 en 249).

Aan het begin van het spel wordt u gevraagd om de eerste zet te doen. Nadat u daarop heeft gereageerd, wordt het scherm gewist en het bord afgebeeld.

Rechts van het bord bevindt zich een kader, waarin wordt aangegeven dat er over een zet wordt nagedacht.



De computer geeft aan wanneer het uw beurt is. Voer eerst het verticale nummer van het veld in en daarna het horizontale nummer, zodat de computer weet op welk veld u het stuk wilt zetten. Als u geen zet kunt doen, moet u een O invoeren.

Het bord wordt gePRINT door de regels 580-780. Deze routine brengt het vierkante net op het scherm, gaat na op welke velden stukken staan en van welke speler de stukken zijn. Ook wordt nagegaan of beide spelers nog meedoen. Naast het speelbord wordt aangegeven hoeveel stukken iedere speler heeft en wie aan de beurt is.

```
10 REM OTHELLO
20 GOSUB 1310
30 DIM A(10,10)
40 CL$=CHR$(12):P$=""
50 REM P$->24 SPATIES
60 GOTO 1110
70 REM GELUIDSRoutine
80 PLAY "O4CEG"
90 RETURN
100 REM INVOER
110 LOCATE 18,21:RETURN
120 ON FLAG GOTO 210,370,380,480
130 REM DENKEN
140 PLAY "T12008C"
150 RETURN
160 PRINT CL$
170 GOSUB 1000:GOSUB 570
180 IF A$="1" THEN GOTO 420
190 LOCATE 23,20:PRINT"MIJN ZET   ":GOSUB 1220
200 FLAG=U1:GOTO 120
210 S=O:T=X:H=0:FOR A=2 TO 9:FOR B=2 TO 9:
    IF A(A,B)<>46 THEN GOTO 370
220 Q=0:FOR C=-1 TO 1:FOR D=-1 TO 1:K=0:F=A:G=B
230 IF A(F+C,G+D)<>S THEN 250
240 K=K+U1:F=F+C:G=G+D:GOTO 230
250 IF A(F+C,G+D)<>T THEN 270
260 Q=Q+K
270 NEXT D:NEXT C
280 IF A=A2 OR A=9 THEN Q=Q*A2
290 IF B=A2 OR B=9 THEN Q=Q*A2
```



```

300 IF A=3 OR A=8 THEN Q=Q/A2
310 IF B=3 OR B=8 THEN Q=Q/A2
320 IF (A=A2 OR A=9) AND (B=3 OR B=8) THEN Q=Q/A2
330 IF (A=3 OR A=8) AND (Z=2 OR Z=9) THEN Q=Q/A2
340 IF Q<H OR Q=0 OR (INT(RND(1))*U1)+1>P3 AND
    Q=H) THEN GOTO 370
350 H=Q:M=A:N=B
360 FLAG=2:GOTO 120
370 FLAG=3:GOTO 120
380 NEXT B:GOSUB 130:NEXT A:IF H=0 AND R=0 THEN
    860
390 IF H=0 THEN GOTO 410
400 GOSUB 790
410 GOSUB 570:GOSUB 1260
420 S=X:T=O
430 LOCATE 23,20:PRINT"UW BEURT";
440 C$=""
450 INPUT B$
460 GOSUB 130
470 FLAG=4:GOTO 120
480 IF B$="" THEN 450
490 IF B$="-" THEN C$="":GOTO 450
500 IF B$="O" THEN R=0:GOTO 560
510 IF ASC(B$)<48 OR ASC(B$)>57 THEN GOTO 450
520 GOSUB 100:C$=C$+B$:IF LEN(C$)=2 THEN R=VAL(
    C$):GOTO 540
530 GOTO 450
540 M=INT(R/10)+1:N=R-10*INT(R/10)+1
550 GOSUB 790
560 GOSUB 570:GOTO 190
570 C=0:H=0
580 LOCATE 16,0:PRINT"OTHELLO":PRINT:PRINT
590 PRINT"    1 2 3 4 5 6 7 8"
600 PRINT
610 LOCATE 0,4:FOR B=2 TO 9:PRINT B-1;
620 FOR D=2 TO 9:IF A(B,D)=X THEN C=C+1:PRINT
    CHR$(32);CHR$(248);:GOTO 640
630 IF A(B,D)=O THEN PRINT CHR$(32);CHR$(249);:
    H=H+1
640 IF A(B,D)<>X AND A(B,D)<>O THEN PRINT " .";
650 NEXT D
660 PRINT B-1:PRINT:NEXT B
670 PRINT "    1 2 3 4 5 6 7 8"
680 PRINT

```

```

690 IF C+H=64 THEN 860
700 IF C=0 THEN 860
710 IF H=0 THEN 860
720 LOCATE 23,13:PRINT"IK HEB: "CHR$(248);
730 IF C>10 THEN PRINT " ";C:PRINT
740 IF C<10 THEN PRINT " ";C:PRINT
750 LOCATE 23,15:PRINT"U HEBT: ";CHR$(249);
760 IF H>10 THEN PRINT " ";H:PRINT
770 IF H<10 THEN PRINT " ";H:PRINT
780 PRINT:RETURN
790 FOR C=-1 TO 1:FOR D=-U1 TO U1:F=M:G=N
800 IF A(F+C,G+D)<>S THEN 820
810 F=F+C:G=G+D:GOTO 800
820 IF A(F+C,G+D)<>T THEN 850
830 A(F,G)=T:IF M=F AND N=G THEN 850
840 F=F-C:G=G-D:GOTO 830
850 NEXT D:NEXT C:RETURN
860 REM WIE WINT ER?
870 FOR TD=1 TO 1000:NEXT TD
880 LOCATE 25,23
890 IF C>H THEN PRINT"IK WIN ";C;" ";H
900 IF C<H THEN PRINT"U WINT";H;" ";C
910 IF H=C THEN PRINT"GELIJK"
920 FOR TD=1 TO 2000:NEXT TD
930 PRINT CL$
940 LOCATE 10,10
950 PRINT:PRINT"      NOG EEN SPEL (J/N)?"
960 A$=INKEY$
970 IF A$="J" THEN RUN
980 IF A$="N" THEN PRINT CL$:END
990 GOTO 960
1000 REM
1010 LOCATE 25,2:PRINT"DENK!"
1020 LOCATE 25,3:PRINT"      "
1030 REM GRAPH-R, GRAPH--(3X), GRAPH-Y
1040 FOR I=0 TO 3
1050 LOCATE 25,4+I:PRINT"      "
1060 REM GRAPH SHIFT-|, SPATIE(3X), GRAPH SHIFT-|
1070 NEXT I
1080 LOCATE 25,7:PRINT"      "
1090 REM GRAPH-V, GRAPH--(3X), GRAPH-N
1100 RETURN
1110 PRINT CL$
1120 X=248:O=249

```

```

1130 U1=1:P2=.3:A2=2
1140 FOR B=1 TO 10:FOR C=1 TO 10:IF B>U1 AND C<>U1
    AND B<> 10 AND C<>10 THEN A(B,C)=ASC(".")
1150 NEXT C:NEXT B
1160 A(5,5)=X:A(6,6)=X:A(5,6)=O:A(6,5)=O:P=0
1170 PRINT CL$:LOCATE 4,10:PRINT"WILT U BEGINNEN? ";
1180 PRINT"JA=1   NEE=2"
1190 A$=INKEY$
1200 IF A$="1" OR A$="2" THEN GOTO 160
1210 GOTO 1190
1220 LOCATE 26,4:PRINT CHR$(205);CHR$(32);
    CHR$(205)
1230 VPOKE 227,254
1240 LOCATE 26,6:PRINT"... "
1250 RETURN
1260 LOCATE 26,4:PRINT"   "
1270 LOCATE 26,5:PRINT"   "
1280 LOCATE 26,6:PRINT"   "
1290 LOCATE 2,18
1300 RETURN
1310 REM TITELPAGINA
1320 CLS
1330 WIDTH 40
1340 KEY OFF
1350 Y=11:X=39:X1=12
1360 LOCATE X,Y
1370 FOR F=1 TO 17
1380 READ B
1390 FOR Z=39 TO X1 STEP -1
1400 LOCATE Z,Y:PRINT CHR$(B);
1410 FOR TD=1 TO 25:NEXT TD
1420 LOCATE Z,Y:PRINT CHR$(32);
1430 NEXT Z
1440 LOCATE Z,Y:PRINT CHR$(B)
1450 IF X1<29 THEN X1=X1+1
1460 NEXT F
1470 FOR TD=1 TO 2000:NEXT TD
1480 RETURN
1490 DATA 79,84,72,69,76,76,79
1500 DATA 32,79,80,32,85,87
1510 DATA 32,77,83,88

```


SCHAAKBORD NIM

Dit programma biedt een interessante variant op het NIM-spel. Men begint met een willekeurig aantal speelstukken (gewoonlijk lucifers) en twee spelers halen om de beurt een aantal stukken weg. De speler die het laatste stuk moet weghalen, heeft verloren.

De computer neemt een willekeurig aantal stukken, zet die op een schaakbord en bepaalt hoeveel stukken er per beurt mogen worden weggehaald. Dat wordt gedaan in regel 80.

De routine in regel 500–570 plaatst het aantal speelstukken in de A-array. Vervolgens drukken de regels 110–150 de benodigde informatie af.

Het bord en de speelstukken worden afgebeeld door middel van de routine in regel 170–260. U kunt de vorm van de speelstukken veranderen door de waarde in regel 420 te veranderen.



```

10 REM SCHAAKBORD NIM *****
20 GOSUB 650
30 CLS
40 CLEAR
50 S=0:C=0
60 Q=RND(-TIME)
70 Z=INT(11*RND(1))+21
80 F=INT(3*RND(1))+3
90 DIM A(32)
100 GOTO 500
110 IF S=0 THEN PRINT TAB(7)"STUKKEN OP HET
    BORD:";Z
120 PRINT
130 IF S=0 THEN PRINT TAB(4)"HET MAXIMUM AANTAL
    DAT U KUNT":PRINT
140 IF S=0 THEN PRINT TAB(11)"NEMEN IS:";F
150 IF C>0 AND S=0 THEN LOCATE 7,16:PRINT"U NEEMT
    ER";C;"EN IK NEEM ER";D
160 LOCATE 0,6
170 FOR A=0 TO 3
180 PRINT
190 FOR B=0 TO 3
200 PRINT TAB(15)CHR$(A(29+A-B*8));CHR$(32);
210 NEXT B
220 PRINT
230 FOR B=0 TO 3
240 PRINT TAB(15)CHR$(A(25+A-B*8));CHR$(32);
250 NEXT B
260 NEXT A
270 IF S=1 THEN LOCATE 16,16:PRINT"U WINT":GOTO 580
280 IF S=2 THEN LOCATE 16,16:PRINT"IK WIN":GOTO 580
290 PRINT
300 LOCATE 7,19
310 PRINT"HOEVEEL WILT U ER NEMEN";
320 INPUT C
330 IF C<1 THEN GOTO 320
340 IF C>F THEN GOTO 320
350 IF C>Z THEN GOTO 320
360 Z=Z-C
370 IF Z>0 THEN GOTO 400
380 S=2
390 IF S=2 THEN GOTO 500
400 FOR TD=1 TO 100:NEXT TD
410 D=Z-1-INT((Z-1)/(F+1))*(F+1)

```

```

420 IF D=0 AND NOT Z=1 THEN D=INT(F*RND(1))+1
430 IF NOT D<Z THEN 420
440 IF Z<F+2 AND INT(5*RND(1))+1=5 THEN D=D+INT(
    2*RND(1))+1-INT(2*RND(1))+1
450 IF D>F THEN GOTO 400
460 IF D>Z THEN GOTO 400
470 IF D<0 THEN D=1
480 Z=Z-D
490 IF Z=0 THEN S=1
500 FOR A=1 TO Z
510 A(A)=249
520 NEXT A
530 FOR A=Z+1 TO 32
540 A(A)=248
550 NEXT A
560 CLS
570 GOTO 110
580 FOR TD=1 TO 2000:NEXT TD
590 CLS
600 LOCATE 8,10:PRINT"NOG EEN SPEL (J/N)?"
610 A$=INKEY$
620 IF A$="J" THEN GOTO 30
630 IF A$="N" THEN END
640 GOTO 610
650 REM TITEL
660 KEY OFF
670 CLS
680 SCREEN 0
690 WIDTH40
700 COLOR 15
710 FLAG=0
720 P=RND(-TIME)
730 FOR Q=1 TO 20
740 X=INT(RND(1)*30)
750 Y=INT(RND(1)*20)+1
760 LOCATE X,Y
770 IF FLAG=1 THEN PRINT"NIM":FLAG=0:GOTO 790
780 IF FLAG=0 THEN PRINT"SCHAAKBORD":FLAG=1
790 FOR TD=1 TO 100:NEXT TD
800 NEXT Q
810 FOR TD=1 TO 2000:NEXT TD
820 RETURN

```


HERSENBREKER

Als laatste spel in dit boek een knaller: een computerversie van het spel Mastermind. Ditmaal niet met kleuren maar met verschillende symbolen.

De computer kiest een combinatie uit zes verschillende symbolen, die onderaan het beeldscherm staan aangegeven. U krijgt tien beurten om de juiste oplossing te bedenken. Daarvoor hebt u een flinke dosis inzicht in de mogelijkheden nodig en moet u de aanwijzingen die de computer geeft op de juiste manier benutten.

Het programma roept in regel 20 de subroutine aan waarin het gewenste SCREEN en de juiste kleur worden ingesteld en schakelt de balk met functietoetsen uit (KEY OFF). Daarna worden de verschillende karakters die in het spel worden gebruikt in regel 1700-1770 opnieuw gedefinieerd. Na RETURN wordt in regel 40-170 de titelpagina verzorgd.

Het hoofdprogramma van het spel start in regel 180. De PRINT-regels 250-300 PRINTen de verschillende karakters waaruit u een keuze kunt maken onder in beeld. De regels 310-380 dienen om een willekeurige reeks karakters in het geheugen van de computer op te slaan. In deze regels wordt er ook voor gezorgd dat eenzelfde karakter niet meerdere keren door de computer kan worden gekozen.



Regel 420 vraagt om uw invoer van een karakterreeks. Daarna wordt in de regels 490–600 gekeken of de door u gekozen karakters overeenkomen met de karakters in het geheugen van de computer en zo ja, of ze ook op de juiste plaats staan. De variabelen B en W houden bij hoeveel karakters juist zijn en hoeveel er op de juiste plaats staan. Aan de hand van deze beide variabelen wordt in de regels 920–1040 een rondje gePRINT wanneer het karakter juist is én op de juiste plaats staat. Een inverse rondje geeft aan wanneer het karakter juist is, maar niet op de juiste plaats staat.

De variabele PO houdt bij in hoeveel beurten u de juiste combinatie raadt. Afhankelijk van de waarde van PO geeft de computer, wanneer u de juiste oplossing raadt, in de regels 1550–1580 het bijbehorende commentaar.

```
10 REM HERSENBREKER
20 GOSUB 1630
30 CLS
40 REM TITELPAGINA
50 FOR X=1 TO 12
60 READ B
70 VPOKE 6144+201+X,B
80 FOR TD=1 TO 175:NEXT TD
90 NEXT X
100 FOR TD=1 TO 450:NEXT TD
110 FOR X=1 TO 14
120 READ B
130 VPOKE 6144+712+X,B
140 FOR TD=1 TO 10:NEXT TD
150 NEXT X
160 A$=INKEY$
170 IF A$="" THEN 160
180 REM HOOFDPROGRAMMA
190 CLS
200 LOCATE 4,10
210 PRINT"KIES NU VIER FIGUREN"
220 FOR TD=1 TO 500:NEXT TD
230 CLS
240 LOCATE 2,22
250 PRINT"1=";CHR$(100);CHR$(32);
260 PRINT"2=";CHR$(101);CHR$(32);
270 PRINT"3=";CHR$(102);CHR$(32);
280 PRINT"4=";CHR$(103);CHR$(32);
290 PRINT"5=";CHR$(104);CHR$(32);
```



```

300 PRINT"6=";CHR$(105)
310 A$="123456"
320 A1$=MID$(A$,INT(RND(1)*6)+1,1):A1=VAL(A1$)
330 A2$=MID$(A$,INT(RND(1)*6)+1,1):A2=VAL(A2$)
340 IF A2=A1 THEN 330
350 A3$=MID$(A$,INT(RND(1)*6)+1,1):A3=VAL(A3$)
360 IF A3=A1 OR A3=A2 THEN 350
370 A4$=MID$(A$,INT(RND(1)*6)+1,1):A4=VAL(A4$)
380 IF A4=A1 OR A4=A2 OR A4=A3 THEN 370
390 LOCATE 0,19
400 X$="":PRINT"KIES DE JUISTE FIGUREN ";
410 LOCATE 22,19
420 INPUT X$
430 IF LEN(X$)<>4 THEN GOTO 1350
440 FOR E=1 TO 4:X1$=MID$(X$,E,1):V=VAL(X1$)
450 IF V<1 OR V>6 THEN GOTO 1350
460 NEXT E
470 PO=PO+1:B=0:W=0
480 AA$=A1$+A2$+A3$+A4$
490 FOR J=1 TO 4
500 X1$=MID$(X$,J,1):G(J)=VAL(X1$)
510 X2$=MID$(AA$,J,1):C(J)=VAL(X2$)
520 IF G(J)=C(J) THEN B=B+1:G(J)=0:C(J)=0
530 X1$="":X2$="":NEXT J
540 FOR J=1 TO 4:IF C(J)=0 THEN 600
550 H=0:FOR K=1 TO 4
560 IF C(J)=0 THEN 590
570 IF C(J)<>G(K) THEN 590
580 H=1:G(K)=0:C(J)=0
590 NEXT K:W=W+H
600 NEXT J
610 ON PO GOTO 620,630,640,650,660,670,680,690,
700,710
620 LOCATE 6,1:PRINT"1 ";:GOTO 720
630 LOCATE 6,2:PRINT"2 ";:GOTO 720
640 LOCATE 6,3:PRINT"3 ";:GOTO 720
650 LOCATE 6,4:PRINT"4 ";:GOTO 720
660 LOCATE 6,5:PRINT"5 ";:GOTO 720
670 LOCATE 6,6:PRINT"6 ";:GOTO 720
680 LOCATE 6,7:PRINT"7 ";:GOTO 720
690 LOCATE 6,8:PRINT"8 ";:GOTO 720
700 LOCATE 6,9:PRINT"9 ";:GOTO 720
710 LOCATE 6,10:PRINT"10 ";
720 X1$=MID$(X$,1,1):X1=VAL(X1$)

```

```

730 X2$=MID$(X$,2,1):X2=VAL(X2$)
740 X3$=MID$(X$,3,1):X3=VAL(X3$)
750 X4$=MID$(X$,4,1):X4=VAL(X4$)
760 P=0:T=0
770 P=P+1:ON X1 GOTO 810,820,830,840,850,860
780 P=P+1:ON X2 GOTO 810,820,830,840,850,860
790 P=P+1:ON X3 GOTO 810,820,830,840,850,860
800 P=P+1:ON X4 GOTO 810,820,830,840,850,860
810 PRINT CHR$(100);" ";GOTO 870
820 PRINT CHR$(101);" ";GOTO 870
830 PRINT CHR$(102);" ";GOTO 870
840 PRINT CHR$(103);" ";GOTO 870
850 PRINT CHR$(104);" ";GOTO 870
860 PRINT CHR$(105);" ";
870 FOR TD=1 TO 50:NEXT TD
880 ON P GOTO 780,790,800
890 ON T GOTO 1100,1110,1120,1130
900 PRINT CHR$(30);:IF B=0 THEN 980
910 GOSUB 1290
920 ON B GOTO 930,940,950,960
930 PRINT CHR$(106);:GOTO 980
940 PRINT CHR$(106);CHR$(106);:GOTO 980
950 PRINT CHR$(106);CHR$(106);CHR$(106);:GOTO 980
960 PRINT CHR$(106);CHR$(106);CHR$(106);CHR$(106)
970 FOR TD=1 TO 500:NEXT TD:GOTO 1380
980 IF W=0 THEN 1050
990 GOSUB 1320
1000 ON W GOTO 1010,1020,1030,1040
1010 PRINT CHR$(107);:GOTO 1050
1020 PRINT CHR$(107);CHR$(107);:GOTO 1050
1030 PRINT CHR$(107);CHR$(107);CHR$(107);:GOTO
1050
1040 PRINT CHR$(107);CHR$(107);CHR$(107);
CHR$(107);
1050 LOCATE 0,19:PRINT SPC(29)
1060 LOCATE 9,15
1070 IF PO<10 THEN 390
1080 LOCATE 2,14:PRINT SPC(7)
1090 T=T+1:ON A1 GOTO 810,820,830,840,850,860
1100 T=T+1:ON A2 GOTO 810,820,830,840,850,860
1110 T=T+1:ON A3 GOTO 810,820,830,840,850,860
1120 T=T+1:ON A4 GOTO 810,820,830,840,850,860
1130 LOCATE 1,16:PRINT"DIT ZIJN DE JUISTE FIGUREN"
1140 LOCATE 9,20:PRINT"DRUK EEN TOETS"

```

```

1150 A$=INKEY$
1160 IF A$="" THEN GOTO 1150
1170 CLS
1180 LOCATE 5,10:PRINT"HET IS FOUT!!"
1190 LOCATE 5,12
1200 PRINT"10 BEURTEN IS TEVEEL"
1210 FOR TD=1 TO 2000:NEXT TD:CLS
1220 LOCATE 5,10:PRINT"NOG EEN KEER J/N"
1230 A$=INKEY$
1240 IF A$="J" THEN RUN
1250 IF A$<>"N" THEN GOTO 1230
1260 CLS
1270 LOCATE 8,10
1280 PRINT"BANGERD!":END
1290 REM GELUID
1300 PLAY"A"
1310 RETURN
1320 REM SOUND
1330 PLAY"C"
1340 RETURN
1350 REM ONJUISTE INVOER
1360 LOCATE 5,20:PRINT"ONJUISTE INVOER"
1370 FOR TD=1 TO 250:NEXT TD:GOSUB 1610:GOTO 390
1380 CLS:LOCATE 7,10:PRINT"JIJ WINT!!!"
1390 FOR Y=15 TO 1 STEP-1
1400 SOUND 0,100
1410 SOUND 1,Y
1420 SOUND 6,0
1430 SOUND 7,8
1440 SOUND 8,16
1450 SOUND 11,5
1460 SOUND 12,1
1470 SOUND 13,8
1480 FOR TD=1 TO 50:NEXT TD
1490 NEXT Y
1500 FOR Y=1 TO 13:SOUND Y,0:NEXT Y
1510 LOCATE 8,22:PRINT"DRUK EEN TOETS"
1520 A$=INKEY$
1530 IF A$="" THEN 1520
1540 CLS:LOCATE 7,10
1550 IF PO=1 THEN PRINT"      WAT EEN GELUK":GOTO
1590
1560 IF PO=2 OR PO=3 THEN PRINT"      NIET GEK":
GOTO 1590

```

```

1570 IF PO=4 OR PO=5 OR PO=6 THEN PRINT"
AARDIG":GOTO 1590
1580 IF PO=7 OR PO=8 THEN PRINT"    ZO ZO":
GOTO 1590
1590 FOR TD=1 TO 400:NEXT TD
1600 GOTO 1220
1610 LOCATE 0,20:PRINT SPC(30)
1620 RETURN
1630 REM INITIALISATIE
1640 SCREEN 1
1650 KEY OFF
1660 Q=RND(-TIME)
1670 COLOR 15,1,1
1680 LOCATE 9,12
1690 PRINT"EVEN GEDULD"
1700 REM KARAKTERS HERDEFINIEREN
1710 FOR N=100 TO 107
1720 FOR M=0 TO 7
1730 READ A
1740 VPOKE N*8+M,A
1750 NEXT M
1760 NEXT N
1770 RETURN
1780 REM DATA VOOR KARAKTERS
1790 DATA 255,255,255,255,255,255,255,255
1800 DATA 1,3,7,15,31,63,127,255
1810 DATA 16,56,124,254,254,16,56,0
1820 DATA 195,231,126,60,60,126,231,195
1830 DATA 0,0,24,60,126,255,255,0
1840 DATA 0,60,60,60,60,60,60,0
1850 DATA 0,60,126,255,255,126,60,0
1860 DATA 255,195,129,0,0,129,195,255
1870 REM DATA VOOR TITELPAGINA
1880 DATA 72,69,82,83,69,78,66,82
1890 DATA 69,75,69,82,68,82,85,75,32,69
1900 DATA 69,78,32,84,79,69,84,83

```



ZELF PROGRAMMA'S SCHRIJVEN

Om zelf een boeiend spel te schrijven, moet u beginnen met een duidelijk omlijnd idee. Dat lijkt vanzelfsprekend, maar als u niet begint met een scherp omschreven doel zult u ontdekken dat het erg moeilijk is om een spelletje te maken dat niet alleen een uitdaging vormt maar ook leuk is om te spelen.

Er vallen grofweg drie soorten spelletjes te onderscheiden:

- bordspelletjes, zoals schaken, solitaire en dammen
- arcadespelletjes, zoals Pacman en Space Invaders
- kansspelletjes, zoals kaartspelen en dobbelen.

Door spelletjes te bekijken en boeken te lezen kunt u ideeën opdoen voor het zelf schrijven van programma's. Voordat u gaat programmeren is het, zoals we al eerder opgemerkt hebben, absoluut noodzakelijk dat u een vastomlijnd idee hebt over wat u moet doen om tot een eindproduct te komen. Maak op papier een ontwerp van alle stadia van het spelletje, van alle functies die de computer moet uitvoeren en van de handelingen die de speler moet verrichten. Moet de computer een denkspel spelen, zoals schaak, of moet hij een spel spelen waarbij toeval een rol speelt, zoals roulette?



Als u een simulatiespel (zoals EIGENHEIMERHEERD) aan het schrijven bent, moet u van tevoren bedenken welke factoren een rol kunnen spelen en hoe die elkaar beïnvloeden. Het is verstandig om in zo'n soort spel een lijstje te maken van alle namen van variabelen die u wilt gebruiken en aan te geven waar deze namen voor staan. Hiermee vermijdt u het probleem dat u variabelen met dezelfde naam gebruikt in verschillende routines.

Als u eenmaal het programmaplan hebt gemaakt, bedenk dan van tevoren welke programmeringstechniek u nodig heeft. Als het een arcade-achtig spelletje is, moet u beslissen of er sprites nodig zijn of dat programmeerbare karakters misschien geschikter zijn.

Maak dan, rekening houdend met al deze factoren, een ruw schets van het programma en noteer tegelijkertijd alle routines die nodig zijn om het goed lopend en makkelijk leesbaar te maken. De grotere programma's in dit boek, zoals EIGENHEIMERHEERD, HET ROMEINSE RIJK en SLAG OM ENGELAND zijn geschreven met een controle-programma, dat in de eerste regels staat. Dit programma roept vervolgens de verschillende subroutines aan.

Deze manier van programmeren, die meestal 'module-programmering' wordt genoemd, is erg effectief. Programma's blijven zo netjes en overzichtelijk met alle verwante functies bij elkaar in één routine. Lezen en fouten opsporen worden erdoor vergemakkelijkt. Bij HET ROMEINSE RIJK bijvoorbeeld weet u, als de informatieverschaffing niet goed werkt, dat het probleem moet worden gezocht in regel 2980-3120, de routine die het senaatsverslag op het scherm brengt.

Routines die maar één keer worden gebruikt, zoals de opstart routines (waarin de variabelen, de sprites en dergelijke opgezet worden) en de titelroutine staan meestal aan het eind van het programma. Probeer de routines te rangschikken naar belangrijkheid. Als u een routine vaak gebruikt, zet deze dan aan het begin van het programma, dat bevordert de snelheid ervan.

In de arcade-achtige en de bewegende grafische spelletjes is snelheid een belangrijk element. Als een spel te snel is, kan altijd een vertragingsslus worden ingebouwd, maar het is erg moeilijk om een langzaam spel te versnellen. In spelletjes zoals AFDALING IN CHAOS en SCHIET hebben we een kernprogramma gebruikt dat alleen bestaat uit de basiselementen die nodig zijn om het programma te laten lopen. Alle andere routines zitten buiten deze lus en worden pas opgeroepen wanneer ze nodig zijn. Dit kernprogramma moet zo beknopt mogelijk worden gehouden voor een snelle en effectieve afwikkeling van het spel.

Als u een programma maakt waarbij invoer van een speler nodig is, zorg er dan van tevoren voor dat u voldoende routines hebt om fouten op te sporen, zodat er geen ongeoorloofde opdrachten worden geaccepteerd door de computer. Vergeet ook niet ervoor te zorgen dat de speler tijd genoeg krijgt om de informatie op het beeldscherm te lezen!

Als u het spelletje alleen heeft ontworpen, bent u de grootste (en enige) deskundige in de wereld op het terrein van uw spelletje. U kunt niet verwachten dat iemand die uw spelletje gaat spelen altijd uw regels zal opvolgen of intelligent zal reageren. Houd daarom in het ontwerp altijd rekening met de meest onwaarschijnlijke mogelijkheden.

Een van de eerste dingen die we doen als we een nieuw spelletje voorgelegd krijgen ter publicatie, is het spelletje spelen, maar dan op de slechtst mogelijk manier. In een avonturenprogramma proberen we bijvoorbeeld door muren heen te lopen, niet bestaande voorwerpen op te pakken en andere onmogelijke dingen te doen. In bewegende grafische spelletjes proberen we van het beeldscherm te verdwijnen, schieten we op eigen voorwerpen en doen in het algemeen alles wat we kunnen om het spelletje mis te laten gaan. Als het een bordspelletje is, zetten we de stukken in de verkeerde richting en doen zoveel mogelijk ongeoorloofde en domme zetten om te kijken of het programma hiertegen opgewassen is.

Wil het spel een topper worden, dan moet het boeiend zijn en een uitdaging vormen voor de speler, zodat hij of zij zich niet verveelt. Veel geluid en kleur en goede routines, die op mislukkingen en succes reageren, kunnen bijdragen tot het succes van het spel.

Tenslotte moet uw spelletje grondig worden getest. Zorg ervoor dat alle routines, zelfs de routines die nauwelijks worden gebruikt, te allen tijde en onder alle omstandigheden goed functioneren.

Als u een spelletje heeft geschreven waarvan u denkt dat het interessant, spannend en uitdagend is, waarom legt u het dan niet ter publicatie voor? Zo kunnen ook anderen ervan genieten. Wij hebben altijd zeer veel belangstelling voor nieuwe programma's.

STAPSGEWIJS EEN PROGRAMMA ONTWIKKELEN

Tot slot van dit boek willen we proberen om het voorgaande hoofdstuk te verduidelijken door in etappes een programma te ontwikkelen.

Het programma dat u nog rest, is er een van het eenvoudige arcade-type. Wij geven u het basisprincipe, maar het is aan u om er een spannend en uitdagend spel van te maken. De bedoeling is dat u de sprite naar willekeurig op het scherm verschijnende hartjes stuurt om deze op te eten. In deze versie is er geen sprake van winst of verlies; het is een demonstratie van de mogelijkheden.

TITELPAGINA

We hebben de verschillende onderdelen in de volgorde van het uiteindelijke programma uitgewerkt. Ten eerste de titelpagina. Om het programma zo gestructureerd mogelijk te houden hebben we de titelroutine aan het eind van het programma gezet.

Door middel van de GOSUB-instructie in regel 20 wordt de titelroutine aangeroepen. De regels 5010-5030 geven de gewenste kleur en het gewenste SCREEN aan en laten de balk met functietoetsen uit beeld verdwijnen (KEY OFF). Met behulp van de READ-instructie wordt de titel in regel 5050 uit de DATA ingelezen en door middel van PRINT CHR\$(B) in regel 5070 gePRINT.

```
10 REM MAAK ZELF EEN SPEL
20 GOSUB 5000:REM TITELPAGINA
.
.
.
5000 REM TITELPAGINA
5010 COLOR 1,14,14
5020 SCREEN 1
5030 KEY OFF
5040 FOR Q=1 TO 17
5050 READ B
5060 LOCATE 5+Q,12
5070 PRINT CHR$(B)
5080 FOR TD=1 TO 100:NEXT TD
5090 NEXT Q
5100 FOR TD=1 TO 2000:NEXT TD
5110 RETURN
.
.
.
10000 REM DATA VOOR TITELPAGINA
10010 DATA 77,83,88,32,80,82,79,71,82
10020 DATA 65,77,77,65,32,77,73,88
```

DEFINITIE VAN EEN SPRITE

Omdat we dit spel als voorbeeldspel zien, hebben we ervoor gekozen om éen sprite én een opnieuw gedefinieerd karakter te gebruiken. Eerst behandelen we de sprite.

Evenals de titelroutine laten we ook de spriteroutine door middel van een GOSUB aan het begin van het programma aanroepen. De sprite is bij de MSX niets anders dan een string. We voegen dan ook de verschillende DATA (door middel van een READ-instructie) toe aan S0\$. Als alle DATA zijn gelezen, noemen we de sprite S0\$.

```
.  
. .  
30 GOSUB 4000:REM DEFINIEER SPRITE  
. .  
4000 REM SPRITE DEFINIERING  
4010 FOR I=1 TO 32  
4020 READ B%  
4030 S0$=S0$+CHR$(B%)  
4040 NEXT I  
4050 SPRITE$(0)=S0$  
4060 RETURN  
. .  
10030 REM DATA VOOR SPRITE  
10040 DATA 127,255,199,199,215,215,215,255  
10050 DATA 254,252,255,255,106,96,49,31  
10060 DATA 254,255,227,227,235,235,235,255  
10070 DATA 127,63,255,255,86,6,92,248
```

EEN KARAKTER HERDEFINIËREN

Nadat de sprite is opgezet, herdefiniëren we in de volgende subroutine karakter 100 (d) tot een hartje. Door de waarden in regel 3010 te veranderen kunt u zoveel karakters herdefiniëren als u nodig heeft. Het meest eenvoudige is om daarvoor karakters te gebruiken die u verder in het programma (uiteraard) niet nodig heeft, bijvoorbeeld de karakters na karakter 100 (d, e, f en volgende kleine letters).

```
.  
. .  
40 GOSUB 3000:REM KARAKTER HERDEFINIEREN  
. .  
3000 REM HERDEFINIEREN VAN KARAKTERS  
3010 FOR N=100 TO 100  
3020 FOR M=0 TO 7  
3030 READ A  
3040 VPOKE N*8+M,A  
3050 NEXT M  
3060 NEXT N  
3070 RETURN  
. .  
10080 REM DATA VOOR NIEUW KARAKTER  
10090 DATA 68,238,254,124,124,56,56,16
```

INITIALISATIE

Voordat het spel kan beginnen, moeten er nog een aantal zaken worden geregeld. Gebruikt u de cursorbesturingstoetsen of een joystick, welke kleur krijgt de sprite, wat zijn de beginwaarden voor de sprite, en andere vragen. Al deze vragen komen aan de orde in het blok INITIALISATIE. Ook zetten we hier een ON INTERVAL-routine op, die elke seconde wordt aangeroepen, om met behulp van de variabele T een tijdslimiet te creëren waarbinnen het hartje opgegeven moet zijn. De regels 800–830 PRINTen een spatie (karakter 32) op de plaats waar eerst het hartje stond.

```
.
.
.
100 REM INITIALISATIE
110 ON INTERVAL=60 GOSUB 900
120 X=10:Y=10
130 Q=RND(-TIME)
140 CLS
150 LOCATE 2,10
160 PRINT"WAT GEBRUIKT U"
170 LOCATE 4,12
180 PRINT"1 JOYSTICK (POORT1)"
190 LOCATE 4,13
200 PRINT"2 JOYSTICK (POORT2)"
210 LOCATE 4,14
220 PRINT"3 CURSORBESTURINGSTOETSEN"
230 LOCATE 2,15
240 INPUT B
250 LOCATE 2,17
260 PRINT"WELKE KLEUR MOET DE SPRITE"
270 LOCATE 2,18
280 PRINT"KRIJGEN";
290 INPUT K
300 CLS
310 IF B=1 THEN R=1
320 IF B=2 THEN R=2
330 IF B=3 THEN R=0
.
.
.
800 REM HARTJE GEPAKT?
810 LOCATE M,N
820 PRINT CHR$(32)
830 RETURN
900 REM INTERVALROUTINE
910 T=T+1
920 RETURN
```

BEWEGING EN CONTROLE

Als laatste onderdeel hebben we gekozen voor beweging en controle. De gemaakte sprite moet over het beeld kunnen worden bewogen. Daarvoor maken we gebruik van de instructie STICK(R), waarbij de variabele R tijdens de initialisatie een waarde heeft gekregen. Gebruikt u een joystick in poort 1 of 2 dan krijgt variabele R respectievelijk de waarde 1 en 2. Gebruikt u de cursorbesturingstoetsen dan krijgt R de waarde 0. STICK(R) leest de invoer en geeft aan de variabele D een waarde (tussen 1 en 8), waarbij elke waarde een richting vertegenwoordigt. Zo betekent waarde 1 OMHOOG en waarde 6 LINKSONDER. Door gebruik te maken van de variabelen X en Y kunnen we in regel 440 door middel van PUT SPRITE(X,Y),K(leur) de gemaakte sprite naar een bepaalde schermpositie brengen.

Verder wordt in deze routine gecontroleerd of er een nieuw hartje moet worden gePRINT. Dit wordt bekeken aan de hand van de variabele T, die elke seconde met 1 wordt verhoogd (zie voorgaande routine). Regel 510 kijkt of de sprite boven het hartje staat. Zo ja, dan wordt de subroutine in regel 800-830 aangeroepen, waar op de plaats van het hartje een spatie wordt gePRINT.

Tenslotte wordt in regel 520 het programma weer naar regel 350 gestuurd, waar opnieuw de besturing uitgelezen wordt.

```
.  
. .  
340 REM BEWEGING EN CONTROLE  
350 INTERVAL ON  
360 D=STICK(R)  
370 IF D=1 THEN Y=Y-1  
380 IF D=2 THEN X=X+1:Y=Y-1  
390 IF D=3 THEN X=X+1  
400 IF D=4 THEN X=X+1:Y=Y+1  
410 IF D=5 THEN Y=Y+1  
420 IF D=6 THEN X=X-1:Y=Y+1  
430 IF D=7 THEN X=X-1  
440 IF D=8 THEN X=X-1:Y=Y-1  
450 PUT SPRITE 0,(X,Y),K  
460 IF T<10 THEN GOTO 520  
470 M=INT(RND(1)*30)+1  
480 N=INT(RND(1)*21)+1  
490 LOCATE M,N  
500 PRINT CHR$(100)  
510 T=0  
520 IF INT(X/8)-1=M AND INT(Y/8)+1=N THEN GOSUB  
800  
530 GOTO 360
```

N.B. U kunt dit programma het beste in één keer intypen. Schrijft u de delen afzonderlijk weg, dan moet u ze later MERGE n tot 1 programma.

Hebt u alle afzonderlijke onderdelen ingetypt en geeft u de RENUM-instructie, dan ziet het programma er zo uit:

```
10 REM MAAK ZELF EEN SPEL
20 GOSUB 710:REM TITELPAGINA
30 GOSUB 640:REM DEFINIEER SPRITE
40 GOSUB 560:REM KARAKTER HERDEFINIEREN
50 REM INITIALISATIE
60 ON INTERVAL=60 GOSUB 530
70 X=10:Y=10
80 Q=RND(-TIME)
90 CLS
100 LOCATE 2,10
110 PRINT"WAT GEBRUIKT U"
120 LOCATE 4,12
130 PRINT"1 JOYSTICK (POORT1)"
140 LOCATE 4,13
150 PRINT"2 JOYSTICK (POORT2)"
160 LOCATE 4,14
170 PRINT"3 CURSORBESTURINGSTOETSEN"
180 LOCATE 2,15
190 INPUT B
200 LOCATE 2,17
210 PRINT"WELKE KLEUR MOET DE SPRITE"
220 LOCATE 2,18
230 PRINT"KRIJGEN";
240 INPUT K
250 CLS
260 IF B=1 THEN R=1
270 IF B=2 THEN R=2
280 IF B=3 THEN R=0
290 REM BEWEGING EN CONTROLE
300 INTERVAL ON
310 D=STICK(R)
320 IF D=1 THEN Y=Y-1
330 IF D=2 THEN X=X+1:Y=Y-1
340 IF D=3 THEN X=X+1
350 IF D=4 THEN X=X+1:Y=Y+1
360 IF D=5 THEN Y=Y+1
370 IF D=6 THEN X=X-1:Y=Y+1
380 IF D=7 THEN X=X-1
390 IF D=8 THEN X=X-1:Y=Y-1
```



```

400 PUT SPRITE 0,(X,Y),K
410 IF T<10 THEN GOTO 470
420 M=INT(RND(1)*30)+1
430 N=INT(RND(1)*21)+1
440 LOCATE M,N
450 PRINT CHR$(100)
460 T=0
470 IF INT(X/8)-1=M AND INT(Y/8)+1=N THEN GOSUB
    490
480 GOTO 310
490 REM HARTJE GEPAKT?
500 LOCATE M,N
510 PRINT CHR$(32)
520 RETURN
530 REM INTERVALROUTINE
540 T=T+1
550 RETURN
560 REM HERDEFINIEREN VAN KARAKTERS
570 FOR N=100 TO 100
580 FOR M=0 TO 7
590 READ A
600 VPOKE N*8+M,A
610 NEXT M
620 NEXT N
630 RETURN
640 REM SPRITE DEFINIERING
650 FOR I=1 TO 32
660 READ B%
670 S0$=S0$+CHR$(B%)
680 NEXT I
690 SPRITE$(0)=S0$
700 RETURN
710 REM TITELPAGINA
720 COLOR 1,14,14
730 SCREEN 1
740 KEY OFF
750 FOR Q=1 TO 17
760 READ B
770 LOCATE 5+Q,12
780 PRINT CHR$(B)
790 FOR TD=1 TO 100:NEXT TD
800 NEXT Q
810 FOR TD=1 TO 2000:NEXT TD
820 RETURN

```

830 REM DATA VOOR TITELPAGINA
840 DATA 77,83,88,32,80,82,79,71,82
850 DATA 65,77,77,65,32,77,73,88
860 REM DATA VOOR SPRITE
870 DATA 127,255,199,199,215,215,215,255
880 DATA 254,252,255,255,106,96,49,31
890 DATA 254,255,227,227,235,235,235,255
900 DATA 127,63,255,255,86,6,92,248
910 REM DATA VOOR NIEUW KARAKTER
920 DATA 68,238,254,124,124,56,56,16

Programmeren in BASIC is voor vele aspirant-programmeurs vaak moeilijker dan het hoeft te zijn. Met **MSX-PROGRAMMA MIX** heeft u de beschikking over complete listings en een uitgebreide toelichting op de opbouw van de programma's, zodat u spelenderwijs komt waar u wezen wilt. En dat lukt des te beter omdat die programma's een goed en efficiënt gebruik maken van de mogelijkheden van uw MSX-computer.

Heeft u niet zozeer de behoefte eigen programma's te leren schrijven, dan biedt **MSX-PROGRAMMA MIX** een gevarieerde hoeveelheid listings, klaar om ingetypt en gebruikt te worden.

ISBN 90 6789 060 X



ADDISON-WESLEY

